

BASIC COMPUTER

David H. Ahl

Spill

Bind 2

BORGEN



**BASIC Computerspil
bind 2**

BASIC Computerspil, 2

Oversat fra amerikansk af Christian Borup.

Originalens titel:

BASIC Computer Games

Copyright © 1978 by Creative Computing Press

Danske udgave © Borgens Forlag 1986

Omslag: Jesper Bach Jensen

Tegninger: George Beker

Published by Borgens Forlag, Valbygaardsvej 33,

DK-2500 Copenhagen Valby

Trykt hos Narayana Press, Gylling

ISBN 87-418-7685-7

1. udgave. 1. oplag. 1986

David Ahl

BASIC **computerspil**

Bind 2

Redigeret af David H. Ahl

Programændringer: Steve North

**Dansk oversættelse, bearbejdning og programmering:
Christian Borup**

Borgen

Forord til den danske udgave

I den danske udgave af Ahl's »BASIC Computer Games« er der anvendt to forskellige programformater. Det ene er beregnet for en skærm med en linielængde på 40 tegn, mens det andet format er beregnet til en skærm med en linielængde på 80. Visse af bogens programmer er beregnet til kørsel på lineskriver, og derfor kan kørslerne af disse programmer virke forvirrende ved første øjekast.

Der er overalt anvendt så »skrabet« en BASIC som muligt, og visse af programmerne er grundet forskelle mellem engelsk og dansk grammatik revideret kraftigt. Det samme gælder for visse af programmernes udskriftsrutiner.

I den danske oversættelse er der, i bogens appendiks, tilføjet programmet BASIC-TEST, der giver mulighed for at se en fuld prøvekørsel af et program, der indeholder alle de BASIC-instruktioner som resten af bogens programmer indeholder. Det anbefales derfor, at man starter med at indtaste og prøvekøre BASIC-TEST, idet man derved hurtigt får et overblik over, hvor eventuelle afvigelser fra bogens BASIC kan optræde.

Christian Borup
Oktober 1985

Indhold · bind 1

Indledning · 6	
Om denne bog · 6	
BASIC · 7	
Oversættelse til andre former for BASIC · 9	
Til slut · 10	
Forord til den danske udgave · 11	
Spillene:	
AKTIER · 12	
AMERIKANSK FODBOLD · 15	
AMERIKANSK FODBOLD, VERSION 2 · 19	
AWARI · 22	
BAGELS · 24	
BAGLÆNS · 26	
BANNER · 28	
BASKETBALL · 30	
BILLE · 33	
BOGSTAV · 36	
BOKSNING · 37	
BOMBARDEMENT · 39	
BONDESPIL · 41	
BORGERKRIG · 44	
BOWLING · 49	
BUNKE · 52	
BYTTEPENGE · 53	
CIFFER · 54	
CRAPS · 56	
DAM · 57	
DAM FOR EN · 59	
DRONNINGESKAK · 61	
DYRERIGET · 63	
ENOGTYVE · 65	
FLIP-FLOP · 68	
FRASEGENERATOR · 69	
GODDAV · 70	
GOLF · 72	
Appendix · 75	
Fortegnelse over spillene opdelt i kategorier · 77	

Indhold · bind 2

Om denne bog · 6
GOMOKO · 7
GÆT · 8
HAMMURABI · 9
HAPS · 11
HASARD · 13
HESTEVÆDDELØB · 14
HOP · 16
HURKLE · 17
HÆNGT MAND · 18
HØJT-LAVT · 21
HØVDING · 22
ISHOCKEY · 23
KALENDER · 27
KAMP · 29
KANIN · 31
KAST · 32
KAST BOMBERNE · 33
KEMIKER · 35
KONGE · 36

KRIG · 39
KRYDS OG BOLLE · 40
KRYDS OG BOLLE, VERSION 2 · 41
KRYDS OG BOLLE I 3-D · 42
KUBUS · 45
LABYRINT · 47
LIGE VINDER · 49
LIGE VINDER, VERSION 2 · 50
LITTERATUR QUIZ · 52
LIV · 53
LIV FOR TO · 56
LOVE · 58
MASTER MIND · 59
MATEMATISK TERNING · 62
MÅL · 63
MÅNELANDING · 65
MÅNELANDING. L.E.M. · 67
MÅNELANDING, RAKET · 70
NAVN · 72
Appendix · 73
Fortegnelse over spillene opdelt i kategorier · 75

Indhold · bind 3

Om denne bog · 6
NICOMACHUS · 7
NIM · 8
OMRING · 10
ORBIT · 11
ORD · 13
PELSHANDLER · 14
PILESPIG · 16
PIZZA · 18
PLOT I 3-D · 20
POESI · 21
POKER · 22
ROULETTE · 26
RUDERE · 29
RUSSISK ROULETTE · 30
SAKS-STEN-PAPIR · 31
SINUSKURVE · 32
SKYTTE · 33
SLAGSKIB · 34
SLALOM · 37
SMUGORM · 39
SOLITAIRE · 41
SPLAT · 44
STJERNER · 46
SUPER STAR TREK INSTRUKTIONER · 47
SUPER STAR TREK X · 50
SYNONYM · 57
SÆNKE SLAGSKIBE · 58
TAL · 61
TERNING · 62
TOG · 63
TYREFÆGTNING · 64
TYVEKNÆGT · 66
TÆNDSTIKKER · 68
TÅRNET I HANOI · 69
UBÅDSJAGT · 71
UGEDAG · 72
Appendix · 74
Fortegnelse over spillene opdelt i kategorier · 76

Om denne bog

BASIC computerspil: Nærværende mikrodatamat-udgave er en kraftig bearbejdet version af min første bog: »101 BASIC Computer Games«, der i 1973 blev udgivet af Digital Equipment Corporation. Programmerne i den første bog var skrevet i seks forskellige versioner af BASIC. Programmerne i nærværende bog er alle skrevet i Microsoft BASIC. I den første bog fandtes flere programmer, der enten ikke kunne omskrives til Microsoft BASIC, eller som ikke var besværet værd. Disse programmer er fjernet og erstattet af nye. Jeg siger tak til alle, som med tålmodighed og forstørrelsesglas tydede de næsten ulæselige programmer i den første bog og fik dem til at virke.

David H. Ahl
September 1978

Om forfatteren:

David H. Ahl har været interesseret i datamater i mange år. I 1956 lærte han at programmere i ALGOL og CAL (Cornell Assembly Language) på en Burroughs B-200. Siden da har han prøvet kræfter med så forskellige uhyrer som: Bendix G-15 og G-20, IBM 650, 704, 7090, 1401, 1130, 360/40 og 360/50, CDC 3200 og 3600, GE 115, 235 og 635, HP 2000, NCR Century 50 og 100, DEC PDP-8, 10, 11, 12 og 15 samt så at sige alle eksisterende mikrodatamater.

Han skrev sit første computerspil allerede ugen efter, at han havde lært at programmere. Siden da har han arbejdet med både alvorlige spil (som f.eks. Carnegie-Mellon Management Game og energisimulering i USA) og underholdningsspil.

Hans liv har altid været knyttet til datamater inden for områder som marketing, produktprognose, pædagogisk forskning, salg af datamater til skoler og universiteter, og han har fungeret som lærer og forelæser. Som udgiver af det amerikanske tidsskrift »Creative Computing« har han fortsat med at udvikle nye områder, hvor datamater kan anvendes.

Gæt

I programmet Gæt vælger datamaten et tilfældigt heltal mellem 0 og en øvre grænse, som du selv fastsætter. Du skal selv senere forsøge at gætte, hvilket tal der er tale om ved hjælp af den vejledning, som datamaten giver.

Du bør være i stand til at gætte tallet på et forsøg mindre, end det antal cifre der skal til for at vise tallet i binær form. Dette bør give dig et fingerpeg om en god søgemetode.

Gæt er oversat fra et oprindeligt program, der var skrevet i FOCAL, der optrådte i bogen: »Computers in the Classroom« af Walt Koetke fra Lexington High School, Massachusetts.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
GÆT  
-----
```

```
I DETTE SPIL SKAL DU GÆTTE TAL.
```

```
JEG TÆNKER PÅ ET TAL MELLEM 1 OG  
EN ØVRE GRÆNSE SOM DU HAR VALGT.
```

```
DEREFTER ER DET DIN OPGAVE AT GÆTTE,  
HVILKET TAL JEG TÆNKER PÅ.
```

```
HVILKEN ØVRE GRÆNSE ?100
```

```
JEG TÆNKER PÅ ET TAL MELLEM 1 OG 100
```

```
DIT GÆT ?50
```

```
FOR LAVT. PRØV MED ET STØRRE TAL.
```

```
DIT GÆT ?75
```

```
FOR HØJT. PRØV MED ET MINDRE TAL.
```

```
DIT GÆT ?62
```

```
FOR LAVT. PRØV MED ET STØRRE TAL.
```

```
DIT GÆT ?70
```

```
FOR LAVT. PRØV MED ET STØRRE TAL.
```

```
DIT GÆT ?73
```

```
FOR HØJT. PRØV MED ET MINDRE TAL.
```

```
DIT GÆT ?72
```

```
DET VAR RIGTIGT!
```


```
DU KLAREDE DET PÅ 6 FORSØG!!
```

```
VIRKELIG GODT.
```

```
JEG TÆNKER PÅ ET TAL MELLEM 1 OG 100
```

```
DIT GÆT ?0
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "          GÆT"  
130 PRINT "-----"  
140 PRINT  
150 PRINT "I DETTE SPIL SKAL DU GÆTTE TAL."  
160 PRINT  
170 PRINT "JEG TÆNKER PÅ ET TAL MELLEM 1 OG"  
180 PRINT "EN ØVRE GRÆNSE SOM DU HAR VALGT."  
190 PRINT  
200 PRINT "DEREFTER ER DET DIN OPGAVE AT GÆTTE,"  
210 PRINT "HVILKET TAL JEG TÆNKER PÅ."  
220 PRINT  
230 PRINT "HVILKEN ØVRE GRÆNSE ";  
240 INPUT L  
250 M = INT (L * RND (1) + 1)  
260 PRINT  
270 L1 = INT ( LOG (L) / LOG (2) ) + 1  
280 PRINT "JEG TÆNKER PÅ ET TAL MELLEM 1 OG";L  
290 G = 1  
300 PRINT  
310 PRINT "DIT GÆT ";  
320 INPUT N  
330 PRINT  
340 IF N > 0 THEN 380  
350 PRINT  
360 PRINT  
370 GOTO 590  
380 IF N = M THEN 450  
390 G = G + 1  
400 IF N > M THEN 430  
410 PRINT "FOR LAVT. PRØV MED ET STØRRE TAL."  
420 GOTO 300  
430 PRINT "FOR HØJT. PRØV MED ET MINDRE TAL."  
440 GOTO 300  
450 PRINT "DET VAR RIGTIGT!"  
460 PRINT  
470 PRINT "DU KLAREDE DET PÅ";G;"FORSØG!!"  
480 PRINT  
490 IF G < L1 THEN 540  
500 IF G = L1 THEN 550  
510 PRINT "IFØLGE STATISTIKKEN BURDE DU"  
520 PRINT "HAVE GÆTTET DET PÅ";L1;"FORSØG."  
530 GOTO 560  
540 PRINT "VIRKELIG ";  
550 PRINT "GODT."  
560 PRINT  
570 PRINT  
580 GOTO 280  
590 END
```



Hammurabi

I dette spil hjælper du herskeren i Babylon – Hammurabi – med at styre byen. Når spillet starter, har byen 1.000 hektar marker, 100 beboere og 3.000 tons korn i forrådet.

Du kan handle (købe og sælge) agerjord med nabobyerne, og du betaler med korn. Priserne varierer fra 17 til 26 tons pr. hektar. Du skal også anvende kornet som føde til dine undersåtter og som såsæd til næste års afgrøder. Du vil hurtigt finde ud af, at et vist antal mennesker kun kan overkomme at passe en vis mængde mark, og at befolkningen sulter, hvis den ikke får nok at spise. Du skal også udsættes for ulykker som pest, rotter, der spiser det oplagrede korn, samt høstudbytter der varierer.

Du vil også mærke, at det ikke er let at styre de ressourcer, du har til rådighed i dette spil, også selv om det kun er over 10 år. Også her vil en befolkningsekspllosion gøre sig gældende.

Programmet blev oprindeligt skrevet i FOCAL på DEC, og forfatteren er ukendt. David Ahl oversatte det til BASIC og bestemte, at spillet skulle strække sig over en 10-årig periode. Hvis du vil ændre noget i programmet, vil det ikke være så vanskeligt, idet det indeholder mange REM-sætninger.

Som en bemærkning for dem der kan lide sladder, kan det nævnes, at navnet Hammurabi – i David Ahls version af programmet – et enkelt sted skrives med kun et 'm'. Denne fejl har nu spredt sig vidt omkring, og der findes en hel generation af edb-studerende, som virkelig tror, at navnet Hammurabi kun indeholder et 'm'.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

HAMMURABI

I DETTE SPIL FÅR DU LEJLIGHED TIL AT
HJÆLPE HAMMURABI MED AT HERSKE OVER
DET GAMLE BABYLON.

DU ER ANSAT FOR EN 10 ÅRS PERIODE.

HERSKER HAMMURABI, JEG BEDER OM
TILLADELSE TIL AT RAPPORTERE FØLGENDE:

I ÅR 1 DØDE 0 MENNESKER AF SULT,
OG DER KOM 5 MENNESKER TIL BYEN.
BEFOLKNINGSTALLET ER NU 100
BYEN EJER 1000 HEKTAR.
DU HAR HØSTET 3 TON PR. HEKTAR.
ROTTERNE HAR SPIST 200 TON.
NU HAR DU 2800 TON I SILOERNE.

LAND KOSTER 19 TON PR. HEKTAR.

HVOR MANGE HEKTAR VIL DU KØBE ?10

HVOR MANGE TON KORN
SKAL FOLKET LEVE AF ?2000

HVOR MANGE HEKTAR SKAL TILSÅS ?1000

MEN DU HAR KUN 100 MENNESKER TIL
AT PASSE MARKERNE. PRØV IGEN.

HVOR MANGE HEKTAR SKAL TILSÅS ?700

HERSKER HAMMURABI, JEG BEDER OM
TILLADELSE TIL AT RAPPORTERE FØLGENDE:

I ÅR 2 DØDE 0 MENNESKER AF SULT,
OG DER KOM 10 MENNESKER TIL BYEN.
EN FRYGTELIG PEST HAR HUSERET.....
HALVDELEN AF BEFOLKNINGEN ER OMKOMMET.
BEFOLKNINGSTALLET ER NU 55
BYEN EJER 1010 HEKTAR.
DU HAR HØSTET 5 TON PR. HEKTAR.
ROTTERNE HAR SPIST 65 TON.
NU HAR DU 3695 TON I SILOERNE.

LAND KOSTER 17 TON PR. HEKTAR.

HVOR MANGE HEKTAR VIL DU KØBE ?10

HVOR MANGE TON KORN
SKAL FOLKET LEVE AF ?1000

HVOR MANGE HEKTAR SKAL TILSÅS ?900

MEN DU HAR KUN 55 MENNESKER TIL
AT PASSE MARKERNE. PRØV IGEN.

HVOR MANGE HEKTAR SKAL TILSÅS ?400

HERSKER HAMMURABI, JEG BEDER OM
TILLADELSE TIL AT RAPPORTERE FØLGENDE:

I ÅR 3 DØDE 5 MENNESKER AF SULT,
OG DER KOM 22 MENNESKER TIL BYEN.
BEFOLKNINGSTALLET ER NU 72
BYEN EJER 1020 HEKTAR.
DU HAR HØSTET 3 TON PR. HEKTAR.
ROTTERNE HAR SPIST 0 TON.
NU HAR DU 3525 TON I SILOERNE.

LAND KOSTER 18 TON PR. HEKTAR.

HVOR MANGE HEKTAR VIL DU KØBE ?100

HVOR MANGE TON KORN
SKAL FOLKET LEVE AF ?400

HVOR MANGE HEKTAR SKAL TILSÅS ?200

DU HAR SULTET 52 IHJEL I ÅR!!

GRUNDET DENNE UHØRT ELENDEGE FORVALT-
NING, ER DU IKKE BLOT BLEVET ANKLAGET
OG FYRET, MEN DU ER OGSÅ BLEVET
UDRÅBT TIL FOLKEFJENDE!!!!

FARVEL SÅ LÆNGE.....

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          HAMMURABI"
130 PRINT "-----"
140 PRINT
150 PRINT "I DETTE SPIL FÅR DU LEJLIGHED TIL AT"
160 PRINT "HJÆLPE HAMMURABI MED AT HERSKE OVER"
170 PRINT "DET GAMLE BABYLON."
180 PRINT
190 PRINT "DU ER ANSAT FOR EN 10 ÅRS PERIODE."
200 PRINT
210 D1 = 0
220 P1 = 0
230 Z = 0
240 P = 95
250 S = 2800
260 H = 3000
270 E = H - S
280 Y = 3
290 A = H / Y
300 I = 5
310 Q = 1
320 D = 0
330 Z = Z + 1
340 PRINT "HERSKER HAMMURABI, JEG BEDER OM"
350 PRINT "TILLADELSE TIL AT RAPPORTERE FØLGENDE:"
360 PRINT
370 PRINT "I ÅR";Z;"DØDE";D;"MENNESKER AF SULT,"
380 PRINT "OG DER KOM";I;"MENNESKER TIL BYEN."
390 P = P + I
400 IF Q > 0 THEN 440
410 P = INT (P / 2)
420 PRINT "EN FRYGTELIG PEST HAR HUSERET....."
430 PRINT "HALVDELEN AF BEFOLKNINGEN ER OMKOMMET."
440 PRINT "BEFOLKNINGSTALLET ER NU";P
450 PRINT "BYEN EJER";A;"HEKTAR."
460 PRINT "DU HAR HØSTET";Y;"TON PR. HEKTAR."
470 PRINT "ROTTERNE HAR SPIST";E;"TON."
480 PRINT "NU HAR DU";S;"TON I SILOERNE."
490 PRINT
500 IF Z = 11 THEN 1580
510 C = INT (10 * RND (1))
520 Y = C + 17
530 PRINT "LAND KOSTER";Y;"TON PR. HEKTAR."
540 PRINT
550 PRINT "HVOR MANGE HEKTAR VIL DU KØBE ";
560 INPUT Q
570 PRINT
580 IF Q < 0 THEN 1520
590 IF Y * Q < = S THEN 620
600 GOSUB 1420
610 GOTO 550
620 IF Q = 0 THEN 670
630 A = A + Q
640 S = S - Y * Q
650 C = 0
660 GOTO 770
670 PRINT "HVOR MANGE HEKTAR VIL DU SÆLGE ";
680 INPUT Q
690 PRINT
700 IF Q < 0 THEN 1520
710 IF Q < (A) THEN 740
720 GOSUB 1460
730 GOTO 670
740 A = A - Q
750 S = S + Y * Q
760 C = 0
770 PRINT "HVOR MANGE TON KORN"
780 PRINT "SKAL FOLKET LEVE AF ";
790 INPUT Q
800 PRINT
810 IF Q < 0 THEN 1520
820 REM Prøves der på at bruge mere
830 REM korn end der er i siloerne?
840 IF Q < = S THEN 870
850 GOSUB 1420
860 GOTO 770
870 S = S - Q
880 C = 1
890 PRINT "HVOR MANGE HEKTAR SKAL TILSÅS ";
900 INPUT D
910 PRINT
920 IF D = 0 THEN 1090
930 IF D < 0 THEN 1520
940 REM Tilsås mere end der ejes?
950 IF D < = (A) THEN 990

```

```

960 GOSUB 1460
970 GOTO 890
980 REM Er der nok korn til saning?
990 IF INT (D / 2) < = S THEN 1030
1000 GOSUB 1420
1010 GOTO 890
1020 REM Er der nok mennesker til at passe markerne?
1030 IF D < 10 * P THEN 1080
1040 PRINT "MEN DU HAR KUN";P;"MENNESKER TIL"
1050 PRINT "AT PASSE MARKERNE.          PRØV IGEN."
1060 PRINT
1070 GOTO 890
1080 S = S - INT (D / 2)
1090 GOSUB 1500
1100 REM En god host!
1110 Y = C
1120 H = D * Y
1130 E = 0
1140 GOSUB 1500
1150 IF INT (C / 2) < > C / 2 THEN 1180
1160 REM Rotterne gar amok!
1170 E = INT (S / C)
1180 S = S - E + H
1190 GOSUB 1500
1200 REM Lad os fa nogle born
1210 I = INT (C * (20 * A + S) / P / 100 + 1)
1220 REM Hvor mange mennesker er matte?
1230 C = INT (Q / 20)
1240 REM Skrak! 15% chance for pest!
1250 Q = INT (10 * (2 * RND (1) - .3))
1260 IF P < C THEN 320
1270 REM Tilstrekkelig sult?
1280 D = P - C
1290 IF D > .45 * P THEN 1340
1300 P1 = ((Z - 1) * P1 + D * 100 / P) / Z
1310 P = C
1320 D1 = D1 + D
1330 GOTO 330
1340 PRINT
1350 PRINT "DU HAR SULTET";D;"IHJEL I AR!!"
1360 PRINT
1370 PRINT "GRUNDET DENNE UHØRT ELENDIGE FORVALT-"
1380 PRINT "NING, ER DU IKKE BLOT BLEVET ANKLAGET"
1390 PRINT "OG FYRET, MEN DU ER OGSA BLEVET"
1400 PRINT "UDRABT TIL FOLKEFJENDE!!!!!"
1410 GOTO 1880
1420 PRINT "HØJTÆREDE HAMMURABI, OMBESTEM DIG, DU"
1430 PRINT "HAR KUN";S;"TON KORN.....IGEN!"
1440 PRINT
1450 RETURN

1460 PRINT "ØRVÆRDIGE HAMMURABI, OVERVEJ DINE ORD"
1470 PRINT "DU EJER KUN";A;"HEKTAR."
1480 PRINT
1490 RETURN
1500 C = INT ( RND (1) * 5) + 1
1510 RETURN
1520 PRINT
1530 PRINT "DYREBARE HAMMURABI, JEG KAN IKKE GØRE"
1540 PRINT "DET DU BEDER MIG OM. DU M SKAFFE DIG"
1550 PRINT "EN ANDEN RDGIVER!!!!!"
1560 PRINT
1570 GOTO 1880
1580 PRINT "I DIN 10 RS EMBEDSPERIODE, ER";P1;"%"
1590 PRINT "AF BEFOLKNINGEN SULTET IHJEL OM RET."
1600 PRINT "DVS. AT";D1;"MENNESKER ER DØDE."
1610 L = A / P
1620 PRINT
1630 PRINT "DU BEGYNDTE MED 10 HEKTAR PR. PERSON,"
1640 PRINT "OG ENDTE MED";" "L" ";HEKTAR PR."
1650 PRINT "PERSON."
1660 PRINT
1670 IF P1 > 33 THEN 1370
1680 IF L < 7 THEN 1370
1690 IF P1 > 10 THEN 1780
1700 IF L < 9 THEN 1780
1710 IF P1 > 3 THEN 1830
1720 IF L < 10 THEN 1830
1730 PRINT "FANTASTISK GODT KLARET!!!!!! SELV IKKE"
1740 PRINT "CHARLEMAGNE, DISRAELI OG JEFFERSON"
1750 PRINT "VILLE I FÆLLESSKAB HAVE KUNNET KLARE"
1760 PRINT "DET BEDRE."
1770 GOTO 1880
1780 PRINT "DIN HARDHENDEDE FREMFÆRD SMAGER AF NERO"
1790 PRINT "OG IVAN DEN GRUSOMME. BEFOLKNINGEN (DEM"
1800 PRINT "DER ER TILBAGE) SYNES DU ER EN LED"
1810 PRINT "DIKTATOR, OG DE HADER DIG EN HEL DEL!"
1820 GOTO 1880
1830 PRINT "DU KUNNE HAVE KLARET DIG BEDRE, MEN I"
1840 PRINT "VIRKELIGHEDEN VAR DET IKKE S SKIDT."
1850 PRINT INT (P * .8 * RND (1));"MENNESKER
ØNSKER"
1860 PRINT "DIG SNIGMYRDET, MEN VI HAR JO ALLE VORE"
1870 PRINT "SM TRIVIELLE PROBLEMER."
1880 PRINT
1890 PRINT "FARVEL S LNGE....."
1900 PRINT
1910 END

```

Haps

Dette program er skrevet med udgangspunkt i et matematisk spil, som oprindeligt blev beskrevet af Martin Gardner i januar-nummeret af »Scientific America« fra 1973. Spillerne bestemmer størrelsen på en kage, som symboliseres af et net, der højst kan måle 9 firkantede kage-stykker på hver led. Det øverste venstre stykke er giftigt. Spillerne skiftes til at tage en bid af kagen, og der begyndes i nederste højre hjørne. Når du skal spise et kagestykke, så indtaster du rækken og kolonnen for et af de stykker, der er tilbage af kagen. Alle stykker til højre for og neden for dette stykke (inklusive stykket selv) forsvinder.

Et valgfrit antal spillere kan deltage. Datamaten fungerer kun som dommer og spiller ikke selv med. Det er svært at udarbejde gode strategier for to spillere. At udarbejde sådanne for tre eller flere spillere er en virkelig udfordring.

Datamatversionen af spillet er skrevet af Peter Sessions fra People's Computer Company.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

HAPS

DETTE ER SPILLET 'HAPS' (FRA SCIENTIFIC AMERICAN, JAN 1973).

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

HAPS SPILLES AF 1 ELLER FLERE SPILLERE (KUN MENNESKELIGE).

SÅDAN SER SPILLEBRÆTTET UD (5 GANGE 7):

```
 1 2 3 4 5 6 7 8 9
1 G * * * * *
2 * * * * *
3 * * * * *
4 * * * * *
5 * * * * *
```

BRÆTTET FORESTILLER EN STOR KAGE, DER ER R RÆKKER HØJ, OG K KOLONNER BRED. DU INDTASTER SELV R OG K VED SPILLET'S BEGYNDELSE. I KAGENS ØVERSTE VENSTRE HJØRNE ER DER ET GIFTIG KVADRAT (G). DEN DER SPISER DET GIFTIGE KVADRAT HAR TABT.

NRR DU SKAL TAGE EN BID, SKAL DU BLOT TASTE RÆKKEN OG KOLONNEN FOR EN AF BIDDERNE I KAGEN, OG SÅ VIL ALLE STYKKER UNDER OG TIL HØJRE FOR DET INDTASTEDE STYKKE OG INKLUSIVE DETTE BLIVE SPIST. - HAPS!

DET ER IKKE TILLADT AT SPISE STYKKER, DER ER UDEFØR KAGEN, ELLER SOM ALLE-REDE ER SPIST!

SÅ GRR VI IGANG....

HVOR MANGE ØNSKER AT SPILLE ?2

HVOR MANGE RÆKKER ?5

HVOR MANGE KOLONNER ?5

```
 1 2 3 4 5 6 7 8 9
1 G * * * * *
2 * * * * *
3 * * * * *
4 * * * * *
5 * * * * *
```

SPILLER 1
HVOR VIL DU HAPSE (RÆKKE,KOLONNE) ?3,3

```
 1 2 3 4 5 6 7 8 9
1 G * * * * *
2 * * * * *
3 * *
4 * *
5 * *
```

SPILLER 2
HVOR VIL DU HAPSE (RÆKKE,KOLONNE) ?4,1

```
 1 2 3 4 5 6 7 8 9
1 G * * * * *
2 * * * * *
3 * *
4
5
```

SPILLER 1
HVOR VIL DU HAPSE (RÆKKE,KOLONNE) ?2,3

```
 1 2 3 4 5 6 7 8 9
1 G * * * * *
2 * *
3 * *
4
5
```

SPILLER 2
HVOR VIL DU HAPSE (RÆKKE,KOLONNE) ?3,2

```
 1 2 3 4 5 6 7 8 9
1 G * * * * *
2 * *
3 *
4
5
```

SPILLER 1
HVOR VIL DU HAPSE (RÆKKE,KOLONNE) ?3,1

```
 1 2 3 4 5 6 7 8 9
1 G * * * * *
2 * *
3
4
5
```

SPILLER 2
HVOR VIL DU HAPSE (RÆKKE,KOLONNE) ?1,4

```
 1 2 3 4 5 6 7 8 9
1 G * *
2 * *
3
4
5
```

SPILLER 1
HVOR VIL DU HAPSE (RÆKKE,KOLONNE) ?2,2

```
 1 2 3 4 5 6 7 8 9
1 G * *
2 *
3
4
5
```

SPILLER 2
HVOR VIL DU HAPSE (RÆKKE,KOLONNE) ?1,3

```
 1 2 3 4 5 6 7 8 9
1 G *
2 *
3
4
5
```

SPILLER 1
HVOR VIL DU HAPSE (RÆKKE,KOLONNE) ?2,1

```
 1 2 3 4 5 6 7 8 9
1 G *
2
3
4
5
```

```

SPILLER 2
HVOR VIL DU HAPSE (RÆKKE,KOLONNE) ?1,2

  1 2 3 4 5 6 7 8 9
1 G
2
3
4
5

SPILLER 1
HVOR VIL DU HAPSE (RÆKKE,KOLONNE) ?1,1

SPILLER 1 TABER!

VIL DU SPILLE IGEN ('J' ELLER 'N') ?N

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                HAPS"
130 PRINT "-----"
140 PRINT
150 DIM A(10,10)
160 REM 'Haps', Copyright
170 REM People's Computer Company
180 PRINT
190 PRINT "DETTE ER SPILLET 'HAPS' (FRA SCIENTIFIC"
200 PRINT "AMERICAN, JAN 1973)."

```

```

620 PRINT "HVOR MANGE ØNSKER AT SPILLE ";
630 INPUT P
640 PRINT
650 I1 = 0
660 PRINT "HVOR MANGE RÆKKER ";
670 INPUT R
680 PRINT
690 IF R < = 9 THEN 720
700 PRINT "ALT FOR MANGE RÆKKER (MAKSIMALT 9)..."

```

Hasard

Dette spil er en simulation af kortspillet Hasard (Acey Ducey). I dette spil giver datamaten to kort med billedsiden opad. Du kan derefter vælge enten at gøre en indsats eller ikke, alt efter om du tror, at det næste kort, datamaten giver, vil have en værdi, der ligger mellem de første to.

Du starter med 100 kroner (variabel Q er sat til 100). Dette kan du selv ændre i programlinje 260, hvis du ønsker at starte med mere eller mindre end 100 kroner. Spillet fortsætter, indtil du taber alle dine penge, eller du selv standser programmet.

Det oprindelige program blev skrevet af Bill Palmby, Prairie View, Illinois.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
-----
                HASARD
-----
KORTSPILLET 'HASARD' (ACEY DUCEY) SPIL-
LES PÅ FØLGENDE MÅDE:
KORTGIVEREN (DATAMATEN) GIVER TO KORT
MED BILLEDSIDEN OPAD. DU SKAL SÅ VÆLGE
OM DU VIL VÆDDE ELLER IKKE, AFHÆNGIG AF
OM DU TROR AT VÆRDIEN PÅ DET NÆSTE KORT
SOM DATAMATEN GIVER LIGGER MELLEM DE TO
ALLEREDE GIVNE ELLER IKKE.
HVIS DU IKKE ØNSKER AT VÆDDE, SKAL DU
BLOT VÆLGE EN INDSATS PÅ '0'.

DU HAR NU 100 KRONER

HER ER DE NÆSTE TO KORT:

KNÆGT
KONGE

HVOR MANGE PENGE VÆDDE DU ?10

7

JEG BEKLAGER, MEN DU HAR TABT

DU HAR NU 90 KRONER

HER ER DE NÆSTE TO KORT:

HVOR MANGE PENGE VÆDDE DU ?70

ES

JEG BEKLAGER, MEN DU HAR TABT

KÆRE VEN,
JEG MÅ DESVÆRRE INFORMERE DIG OM,
AT DU ER GRET FALLIT!

VIL DU SPILLE IGEN ('J' ELLER 'N') ?
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                HASARD"
130 PRINT "-----"
140 PRINT
150 PRINT "KORTSPILLET 'HASARD' (ACEY DUCEY) SPIL-"
160 PRINT "LES PÅ FØLGENDE MÅDE:"
170 PRINT "KORTGIVEREN (DATAMATEN) GIVER TO KORT"
180 PRINT "MED BILLEDSIDEN OPAD. DU SKAL SÅ VÆLGE"
190 PRINT "OM DU VIL VÆDDE ELLER IKKE, AFHÆNGIG AF"
200 PRINT "OM DU TROR AT VÆRDIEN PÅ DET NÆSTE KORT"
210 PRINT "SOM DATAMATEN GIVER LIGGER MELLEM DE TO"
220 PRINT "ALLEREDE GIVNE ELLER IKKE."
230 PRINT "HVIS DU IKKE ØNSKER AT VÆDDE, SKAL DU"
240 PRINT "BLOT VÆLGE EN INDSATS PÅ '0'."
250 N = 100
260 Q = 100
270 PRINT
280 PRINT "DU HAR NU";Q;"KRONER"
290 PRINT
300 GOTO 350
310 Q = Q + M
320 GOTO 270
330 Q = Q - M
340 GOTO 270
350 PRINT "HER ER DE NÆSTE TO KORT:"
360 PRINT
370 A = INT (14 * RND (1)) + 2
380 IF A < 2 THEN 370
```

```
390 IF A > 14 THEN 370
400 B = INT (14 * RND (1)) + 2
410 IF B < 2 THEN 400
420 IF B > 14 THEN 400
430 IF A > = B THEN 370
440 IF A < 11 THEN 490
450 IF A = 11 THEN 510
460 IF A = 12 THEN 530
470 IF A = 13 THEN 550
480 IF A = 14 THEN 570
490 PRINT A
500 GOTO 580
510 PRINT "KNÆGT"
520 GOTO 580
530 PRINT "DAME"
540 GOTO 580
550 PRINT "KONGE"
560 GOTO 580
570 PRINT "ES"
580 IF B < 11 THEN 630
590 IF B = 11 THEN 650
600 IF B = 12 THEN 670
610 IF B = 13 THEN 690
620 IF B = 14 THEN 710
630 PRINT B
640 GOTO 720
650 PRINT "KNÆGT"
660 GOTO 720
670 PRINT "DAME"
680 GOTO 720
690 PRINT "KONGE"
700 GOTO 720
710 PRINT "ES"
720 PRINT
730 PRINT "HVOR MANGE PENGE VÆDDE DU ";
740 INPUT M
750 PRINT
760 IF M < 0 THEN 730
770 IF M < > 0 THEN 810
780 PRINT "SIKKEN EN BANGEBUKS!!.....SKAM DIG!!"
790 PRINT
800 GOTO 1220
810 IF M < = Q THEN 870
820 PRINT "-----"
830 PRINT "DESVÆRRE ER DIN INDSATS ALT FOR STOR!!!"
840 PRINT "DU HAR KUN";Q;"KRONER PÅ DIG."
850 PRINT "-----"
860 GOTO 720
870 C = INT (14 * RND (1)) + 2
880 IF C < 2 THEN 870
890 IF C > 14 THEN 870
900 IF C < 11 THEN 950
910 IF C = 11 THEN 970
920 IF C = 12 THEN 990
930 IF C = 13 THEN 1010
940 IF C = 14 THEN 1030
950 PRINT C
960 GOTO 1040
970 PRINT "KNÆGT"
980 GOTO 1040
990 PRINT "DAME"
1000 GOTO 1040
1010 PRINT "KONGE"
1020 GOTO 1040
1030 PRINT "ES"
1040 PRINT
1050 IF C > (A) THEN 1070
1060 GOTO 1100
1070 IF C > = B THEN 1100
1080 PRINT "TILLYKKE!!! DU VANDT!"
1090 GOTO 310
1100 PRINT "JEG BEKLAGER, MEN DU HAR TABT"
1110 IF M < Q THEN 330
1120 PRINT
1130 PRINT
1140 PRINT "KÆRE VEN,"
1150 PRINT "JEG MÅ DESVÆRRE INFORMERE DIG OM,"
1160 PRINT "AT DU ER GRET FALLIT!"
1170 PRINT
1180 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1190 INPUT AS
1200 PRINT
1210 IF LEFT$(AS,1) = "J" THEN 100
1220 PRINT
1230 PRINT "TAK FOR DENNE GANG....."
1240 PRINT "JEG HÅBER DU HAR MORET DIG....."
1250 PRINT "FARVEL."
1260 END
```

Hestevæddeløb

Dette program simulerer et hestevæddeløb over 1.600 meter for treårige fuldblodsheste. Højest 10 personer kan deltage i spillet og gøre indsats i løbet. Den maksimale indsats er 10.000 kroner. I dette program kan man kun satse på vinderen. Indsatsen foretages ved at indtaste hestens nummer, et komma, og indsatsen. Datamaten udskriver derefter hestens placering syv gange, mens løbet er i gang og ved målstregen. Udbetaling af gevinster vises til slut.

Programmet er skrevet af Laurie Chevalier, som på det tidspunkt studerede ved South Portland High School.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

 HESTEVÆDDELØB

VELKOMMEN TIL HÆSTESTADS GALOPBANE

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

OP TIL 10 PERSONER KAN SPILLE AD GANGEN. DER VIL BLIVE UDSKREVET EN LISTE OVER HESTENES ODDS.

DU KAN SÆTTE ET HVILKET SOM HELST BELØB UNDER 100.000 KRONER PÅ EN AF HESTENE.

NÅR LØBET STARTER, VISES EN HEST VED HJÆLP AF DENS NUMMER. MÅLET ER PLACERET NEDERST PÅ SKÆRMEN.

HVOR MANGE VIL SPILLE MED ?2

INDTAST SPILLERNES NAVN NEDENFOR:

SPILLER 1 ?HENRIK

SPILLER 2 ?METTE

1 2 3 4 5 6 7 8
 XXXXXXXX START XXXXXXXX

5
 8 7 2 4 1
 6 3

HEST	NUMMER	ODDS
MONA MIA	1	4 : 1
MISTY JOE	2	13 : 1
MOR KAREN	3	6 : 1
NAUTILUS	4	19 : 1
JOLLY	5	9 : 1
MOLLEFYR	6	6 : 1
MAYBE NOT	7	19 : 1
MIDNAT	8	5 : 1

HVAD ER DIN INDSATS , ('HESTNUMMER, BELØB')

HENRIK ?8,3000
 METTE ?6,8000

XXXXXXXXX START XXXXXXXX

5
 4 7
 2
 1
 8
 3
 6

XXXXXXXXX SLUT XXXXXXXX

XXXXXXXXX SLUT XXXXXXXX

XXXXXXXXX START XXXXXXXX

XXXXXXXXX START XXXXXXXX

5 1
 7
 4
 8
 2
 6 3

7 1
 5

2
 8 4

3
 6

XXXXXXXXX SLUT XXXXXXXX

XXXXXXXXX SLUT XXXXXXXX

XXXXXXXXX START XXXXXXXX

XXXXXXXXX START XXXXXXXX

1
 5 7
 4 2
 8
 3
 6

7 5 1

4
 8 2

XXXXXXXXX SLUT XXXXXXXX

3
 XXXXXXXX SLUT XXXXXXXX

**** LØBSRESULTATERNE ****

PLACERING	HEST	ODDS
1	6	6 : 1
2	3	6 : 1
3	2	13 : 1
4	8	5 : 1
5	4	19 : 1
6	1	4 : 1
7	5	9 : 1
8	7	19 : 1

METTE VINDER 48000 KR.

ØNSKER DU AT SPILLE I NÆSTE LØB (TAST 'J' ELLER 'N') ?N


```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          HESTEVÆDELØB"
130 PRINT "-----"
140 PRINT
150 DIM S(8)
160 PRINT "VELKOMMEN TIL HÆSTESTADS GALOPBANE"
170 PRINT
180 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
190 INPUT X$
200 PRINT
210 IF LEFT$(X$,1) = "N" THEN 330
220 PRINT "OP TIL 10 PERSONER KAN SPILLE AD"
230 PRINT "GANGEN. DER VIL BLIVE UDSKREVET EN"
240 PRINT "LISTE OVER HESTENES ODDS."
250 PRINT
260 PRINT "DU KAN SÆTTE ET HVILKET SOM HELST BELØB"
270 PRINT "UNDER 100.000 KRONER PÅ EN AF HESTENE."
280 PRINT
290 PRINT "NÅR LØBET STARTER, VISES EN HEST VED"
300 PRINT "HJÆLP AF DENS NUMMER. MÅLET ER PLACERET"
310 PRINT "NEDERST PÅ SKÆRMEN."
320 PRINT
330 PRINT "HVOR MANGE VIL SPILLE MED ";
340 INPUT C
350 PRINT
360 PRINT "INDTAST SPILLERNE NAVN NEDENFOR:"
370 PRINT
380 FOR A = 1 TO C
390 PRINT "SPILLER";A;
400 INPUT W$(A)
410 NEXT A
420 PRINT
430 PRINT "HEST"; TAB( 17);"NUMMER"; TAB( 27);"ODDS"
440 PRINT "-----"
450 FOR I = 1 TO 8
460 S(I) = 0
470 NEXT I
480 R = 0
490 FOR A = 1 TO 8
500 D(A) = INT (10 * RND (1) + 1)
510 NEXT A
520 FOR A = 1 TO 8
530 R = R + D(A)
540 NEXT A
550 V$(1) = "MONA MIA"
560 V$(2) = "MISTY JOE"
570 V$(3) = "MOR KAREN"
580 V$(4) = "NAUTILUS"
590 V$(5) = "JOLLY"
600 V$(6) = "MOLLEFYR"
610 V$(7) = "MAYBE NOT"
620 V$(8) = "MIDNAT"
630 FOR N = 1 TO 8
640 PRINT V$(N); TAB( 18);N; TAB( 25);
650 IF INT (R / D(N)) < 10 THEN PRINT " ";
660 PRINT INT (R / D(N));": 1"
670 NEXT N
680 PRINT "-----"
690 PRINT
700 PRINT "HVAD ER DIN INDSATS ,"
710 PRINT "('HESTNUMMER, BELØB)' "
720 PRINT
730 FOR J = 1 TO C
740 PRINT W$(J);" ";
750 INPUT Q(J),P(J)
760 IF P(J) < 1 THEN 780
770 IF P(J) < 100000 THEN 800
780 PRINT "DET KAN DU IKKE SATSE...."
790 GOTO 740
800 NEXT J
810 PRINT
820 PRINT " 1 2 3 4 5 6 7 8 "
830 PRINT "XXXXXXXXX START XXXXXXXXX"
840 FOR I = 1 TO 8
850 M = I
860 M(I) = M
870 Y(M(I)) = INT (100 * RND (1) + 1)
880 IF Y(M(I)) < 10 THEN 970
890 S = INT (R / D(I) + .5)
900 IF Y(M(I)) < S + 17 THEN 990
910 IF Y(M(I)) < S + 37 THEN 1010
920 IF Y(M(I)) < S + 57 THEN 1030
930 IF Y(M(I)) < S + 77 THEN 1050
940 IF Y(M(I)) < S + 92 THEN 1070
950 Y(M(I)) = 7
960 GOTO 1080
970 Y(M(I)) = 1
980 GOTO 1080
990 Y(M(I)) = 2
1000 GOTO 1080
1010 Y(M(I)) = 3
1020 GOTO 1080
1030 Y(M(I)) = 4
1040 GOTO 1080
1050 Y(M(I)) = 5
1060 GOTO 1080
1070 Y(M(I)) = 6
1080 NEXT I
1090 M = I
1100 FOR I = 1 TO 8
1110 S(M(I)) = S(M(I)) + Y(M(I))
1120 NEXT I
1130 I = 1
1140 FOR L = 1 TO 8
1150 FOR I = 1 TO 8 - L
1160 IF S(M(I)) < S(M(I + 1)) THEN 1200
1170 H = M(I)
1180 M(I) = M(I + 1)
1190 M(I + 1) = H
1200 NEXT I
1210 NEXT L
1220 T = S(M(8))
1230 FOR I = 1 TO 8
1240 B = S(M(I)) - S(M(I - 1))
1250 IF B = 0 THEN 1300
1260 FOR A = 1 TO B
1270 PRINT
1280 IF S(M(I)) > 27 THEN 1350
1290 NEXT A
1300 PRINT M(I);
1310 NEXT I
1320 FOR A = 1 TO 28 - T
1330 PRINT
1340 NEXT A
1350 PRINT "XXXXXXXXXX SLUT XXXXXXXXXX"
1360 PRINT
1370 PRINT
1380 PRINT "-----"
1390 PRINT
1400 PRINT
1410 IF T < 28 THEN 830
1420 PRINT "**** LØBSRESULTATERNE ****"
1430 Z9 = 1
1440 PRINT
1450 PRINT "PLACERING"; TAB( 12);"HEST"; TAB(
23);"ODDS"
1460 FOR I = 8 TO 1 STEP - 1
1470 F = M(I)
1480 PRINT Z9; TAB( 12);F; TAB( 21);
1490 IF INT (R / D(F)) < 10 THEN PRINT " ";
1500 PRINT INT (R / D(F));":1"
1510 Z9 = Z9 + 1
1520 NEXT I
1530 FOR J = 1 TO C
1540 IF Q(J) < > M(8) THEN 1590
1550 N = Q(J)
1560 PRINT
1570 PRINT W$(J);" VINDER";
1580 PRINT INT ( INT (R / D(N)) * P(J) + .5);"KR."
1590 NEXT J
1600 PRINT
1610 PRINT "ØNSKER DU AT SPILLE I NÆSTE LØB"
1620 PRINT "(TAST 'J' ELLER 'N') ";
1630 INPUT O$
1640 PRINT
1650 IF LEFT$(O$,1) = "J" THEN 420
1660 END

```

Hop

Dette program plottes en hoppende bold. De fleste udskriftsenheder til datamater udskriver i papirets længderetning (oppefra og ned), men dette program plottes vandret (fra venstre mod højre).

Du angiver selv boldens starthastighed og dens elasticitetskoefficient (en superbold har en elasticitetskoefficient på omkring 0.85, almindelige bolde har en meget lavere værdi). Du angiver også selv tidsintervallet, som skal bruges til at »fange bolden i flugten«. Det er cirka det samme, som sker, når du kaster bolden i et mørkt rum og derefter i hurtig rækkefølge fotograferer den med blitz for at se dens flugt (stroboskopisk lys).

Programmet er oprindeligt skrevet af Val Skalabrin, mens han var ved DEC.

```

MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY
-----
                                HOP
-----

I DETTE SIMULATIONSPROGRAM SKAL DU VÆLGE STARHASTIGHEDEN
FOR EN BOLD SOM KASTES LIGE OP I LUFTEN.

DU SKAL OGSÅ VÆLGE BOLDENS ELASTICITETS-KOEFFICIENT.

ELASTICITETS-KOEFFICIENTEN ER ET TAL MELLEML 0 OG 1.

PROGRAMMET TEGNER EN KURVE OVER BOLDENS HOP, OG FOR AT
DETTE KAN GØRES, SKAL DU VÆLGE DET TIDSINTERVAL SOM DET
STROBOSKOPISE LYS SKAL BRUGE FOR AT KUNNE OPTEGNE
BOLDENS BEVÆGELSER (PRØV MED .1 TIL AT BEGYNDE MED).

TIDSINTERVAL (SEKUNDER) ?.1

STARHASTIGHED (METER/SEKUND) ?25

ELASTICITETS-KOEFFICIENT (0-1) ?.9

METER
10  0
   0 0
9   0 0
   0
8   0      00
   0      00 0
7   0      0
   0      0 0      00
6  0      0 0      0 0      00
   0      0 0      0 0      000
4   0      0      0      0      0 0
   0      0      0      0      0 0
3   0      0      0      0      0 0
0
2   0      0 0      0 0      0
   0      0      0 0      0 0
1
0 0      0      00      0      0
-----
0   1      2      3      4      5      6
                                SEKUNDER

```

```

100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY      "
110 PRINT "-----"
120 PRINT "                                HOP"
130 PRINT "-----"
140 PRINT
150 PRINT
160 DIM T(20)
170 PRINT "I DETTE SIMULATIONSPROGRAM SKAL DU VÆLGE STARHASTIGHEDEN"
180 PRINT "FOR EN BOLD SOM KASTES LIGE OP I LUFTEN."
190 PRINT
200 PRINT "DU SKAL OGSÅ VÆLGE BOLDENS ELASTICITETS-KOEFFICIENT."
210 PRINT
220 PRINT "ELASTICITETS-KOEFFICIENTEN ER ET TAL MELLEML 0 OG 1."
230 PRINT
240 PRINT "PROGRAMMET TEGNER EN KURVE OVER BOLDENS HOP, OG FOR AT"
250 PRINT "DETTE KAN GØRES, SKAL DU VÆLGE DET TIDSINTERVAL SOM DET"
260 PRINT "STROBOSKOPISE LYS SKAL BRUGE FOR AT KUNNE OPTEGNE"
270 PRINT "BOLDENS BEVÆGELSER (PRØV MED .1 TIL AT BEGYNDE MED)."
280 PRINT
290 PRINT "      TIDSINTERVAL (SEKUNDER) ";
300 INPUT S2
310 PRINT
320 PRINT " STARHASTIGHED (METER/SEKUND) ";
330 INPUT V
340 PRINT
350 PRINT "ELASTICITETS-KOEFFICIENT (0-1) ";
360 INPUT C
370 PRINT
380 PRINT "METER"
390 PRINT
400 S1 = INT (70 / (V / (16 * S2)))
410 FOR I = 1 TO S1
420 T(I) = V * C ^ (I - 1) / 16
430 NEXT I
440 FOR H = INT (- 16 * (V / 32) ^ 2 + V ^ 2 / 32 + .5) TO 0 STEP - .5
450 IF INT (H) < > H THEN GOTO 470
460 PRINT H;
470 L = 0
480 FOR I = 1 TO S1
490 FOR T = 0 TO T(I) STEP S2
500 L = L + S2
510 IF ABS (H - (.5 * (- 32) * T ^ 2 + V * C ^ (I - 1) * T)) > .25 THEN
GOTO 530
520 PRINT TAB( L / S2);"0";
530 NEXT T
540 T = T(I + 1) / 2
550 IF H > - 16 * T ^ 2 + V * C ^ (I - 1) * T THEN GOTO 570
560 NEXT I
570 PRINT
580 NEXT H
590 PRINT TAB( 1);
600 FOR I = 1 TO INT (L + 1) / S2 + 1
610 PRINT "-";
620 NEXT I
630 PRINT
640 PRINT " 0";
650 FOR I = 1 TO INT (L + .9995)
660 PRINT TAB( INT (I / S2));I;
670 NEXT I
680 PRINT
690 PRINT TAB( INT (L + 1) / (2 * S2) - 2);"SEKUNDER"
700 PRINT
710 PRINT
720 PRINT
730 PRINT
740 GOTO 250
750 END


```

Hurkle

Hurkle? Hvad er Hurkle. Hurkle er et lykkeligt dyr, som bor i en anden galakse, på en planet der hedder Lirht, som har tre måner. Hurkle er Gwikkernes foretrukne kæledyr. Gwikkerne er den talrigeste dyreart på planeten Lirht, og... nå, hvis du vil vide mere, så må du læse historien »The Hurkle is a Happy Beast«, i bogen »A Way Home«, skrevet af Theodore Sturgeon.

I dette program gemmer en genert Hurkle sig i et net, der måler 10 gange 10 felter. Nettets startpunkt (0,0) er i det sydvestlige hjørne. Dit gæt på, hvor Hurkle befinder sig, skal bestå af to heltal, adskilt af et komma. Efter hvert gæt vil datamaten fortælle dig, i hvilken retning du skal gå for at lede efter Hurkle. Du har fem gæt til at finde ham. Dette kan du også forandre i programlinje 150, dog vil fire gæt for det meste være nok.

Dette program er skrevet af Bob Albrecht fra People's Computer Company.



```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
-----
HURKLE
-----
```

'HURKLE' GEMMER SIG I ET NET, DER MÅLER 10 GANGE 10 FELTER. HJEM-STEDET ER PÅ NETTET PUNKTET '0,0', OG ETHVERT FELT PÅ NETTET ANGIVES VED TO TAL, ADSKILDT AF ET KOMMA.

PRØV AT GÆTTE HVOR 'HURKLE' GEMMER SIG. DU HAR 5 FORSØG TIL RÅDIGHED.

EFTER HVERT FORSØG VIL JEG FORTÆLLE DIG I HVILKEN RETNING DU SKAL GÅ FOR AT FINDE 'HURKLE'.

DIT GÆT NUMMER 1 ?10,10
GÅ MOD SYDVEST

DIT GÆT NUMMER 2 ?4,4
GÅ MOD SYDØST

DIT GÆT NUMMER 3 ?6,2
GÅ MOD ØST

DIT GÆT NUMMER 4 ?7,2

DU FANDT HAM PÅ 4 GÆT.

LAD OS SPILLE IGEN.....

NU GEMMER 'HURKLE' SIG IGEN.

DIT GÆT NUMMER 1 ?0,0
GÅ MOD NORDØST

DIT GÆT NUMMER 2 ?5,5
GÅ MOD VEST

DIT GÆT NUMMER 3 ?5,1
GÅ MOD NORDVEST

DIT GÆT NUMMER 4 ?2,4
GÅ MOD NORD

DIT GÆT NUMMER 5 ?2,6
GÅ MOD SYD

DESVÆRRE, DET VAR DINE 5 FORSØG.

'HURKLE' GEMMER SIG I FELT 2, 5

LAD OS SPILLE IGEN.....

NU GEMMER 'HURKLE' SIG IGEN.

DIT GÆT NUMMER 1 ?

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          HURKLE"
130 PRINT "-----"
140 N = 5
150 G = 10
160 PRINT
170 PRINT "'HURKLE' GEMMER SIG I ET NET, DER MÅLER"
180 PRINT G;"GANGE";G;"FELTER. HJEM-"
190 PRINT "STEDET ER PÅ NETTET PUNKTET '0,0', OG"
200 PRINT "ETHVERT FELT PÅ NETTET ANGIVES VED TO"
210 PRINT "TAL, ADSKILDT AF ET KOMMA."
220 PRINT
230 PRINT "PRØV AT GÆTTE HVOR 'HURKLE' GEMMER SIG."
240 PRINT "DU HAR";N;"FORSØG TIL RÅDIGHED."
250 PRINT
260 PRINT "EFTER HVERT FORSØG VIL JEG FORTÆLLE DIG"
270 PRINT "I HVILKEN RETNING DU SKAL GÅ FOR AT"
280 PRINT "FINDE 'HURKLE'."
290 PRINT
300 A9 = INT (G * RND (1))
310 B = INT (G * RND (1))
320 FOR K = 1 TO N
330 PRINT "DIT GÆT NUMMER";K;
340 INPUT X,Y
350 IF X > G OR X < 0 THEN 330
360 IF Y > G OR Y < 0 THEN 330
370 IF ABS (X - A9) + ABS (Y - B) = 0 THEN 520
380 REM   Skriv oplysninger
390 GOSUB 550
400 PRINT
410 NEXT K
420 PRINT
430 PRINT "DESVÆRRE, DET VAR DINE";N;"FORSØG."
440 PRINT
450 PRINT "'HURKLE' GEMMER SIG I FELT ";A9;"",B
460 PRINT
470 PRINT "LAD OS SPILLE IGEN....."
480 PRINT
490 PRINT "NU GEMMER 'HURKLE' SIG IGEN."
500 PRINT
510 GOTO 300
520 PRINT
530 PRINT "DU FANDT HAM PÅ";K;"GÆT."
540 GOTO 460
550 PRINT "GÅ MOD ";
560 IF Y = B THEN 610
570 IF Y < B THEN 600
580 PRINT "SYD";
590 GOTO 610
600 PRINT "NORD";
610 IF X = A9 THEN 660
620 IF X < A9 THEN 650
630 PRINT "VEST";
640 GOTO 660
650 PRINT "ØST";
660 PRINT
670 RETURN
680 END
```

Hængt mand

Dette er en simulation af spillet Hangman, hvor man skal gætte ord. Datamaten vælger et ord og fortæller dig, hvor mange bogstaver ordet består af. Derefter begynder du at gætte, hvilke bogstaver ordet består af. Hvis du gætter rigtigt, så fortæller datamaten dig, hvor bogstavet findes i ordet. Hvis du derimod gætter forkert, så begynder den at hænge dig. Du har følgende ti gæt, før du bliver helt hængt:

Hoved
Krop
Højre og venstre arm
Højre og venstre ben
Højre og venstre hånd
Højre og venstre fod

Du kan selv indføre flere ord i programmet i de 9 DATA-sætninger, der afslutter programmet, men så skal du også huske at forandre tallet (N), der styrer indlæsningen af ordene, dette gøres i programlinie 170.

David Ahl modificerede programmet til dets nuværende form fra et oprindeligt program, der var skrevet af Kenneth Aupperle fra Melville, New York.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

HÆNGT MAND

HER ER DE BOGSTAVER DU HAR BRUGT:

GÆT ET BOGSTAV ?E

---E

GÆT HVILKET ORD DET ER ?MURE

FORKERT ORD. GÆT ET BOGSTAV IGEN.

HER ER DE BOGSTAVER DU HAR BRUGT:

E

---E

GÆT ET BOGSTAV ?D

DESVÆRRE, DET BOGSTAV ER IKKE I ORDET.

ALLERFØRST TEGNER VI ET HOVED.

XXXXXXX
X X
X ---
X (. .)
X ---
X
X
X
X
X
X
X
X

HER ER DE BOGSTAVER DU HAR BRUGT:

E,D

---E

GÆT ET BOGSTAV ?G

DESVÆRRE, DET BOGSTAV ER IKKE I ORDET.

NU TEGNER VI EN KROP.

XXXXXXX
X X
X ---
X (. .)
X ---
X X
X X
X X
X X
X
X
X

HER ER DE BOGSTAVER DU HAR BRUGT:

E,D,G

---E

GÆT ET BOGSTAV ?H

DESVÆRRE, DET BOGSTAV ER IKKE I ORDET.

SÅ TEGNER VI EN ARM.

XXXXXXX
X X
X ---
X (. .)
X) ---
X) X
X)X
X X
X X
X
X
X

HER ER DE BOGSTAVER DU HAR BRUGT:

E,D,G,H

---E

GÆT ET BOGSTAV ?V

V--E

GÆT HVILKET ORD DET ER ?VEJE

FORKERT ORD. GÆT ET BOGSTAV IGEN.

HER ER DE BOGSTAVER DU HAR BRUGT:

E,D,G,H,V

V--E

GÆT ET BOGSTAV ?I

DESVÆRRE, DET BOGSTAV ER IKKE I ORDET.

DENNE GANG TEGNER VI DEN ANDEN ARM.

XXXXXXX
X X
X ---
X) (. .) (
X) --- (
X) X (
X)X(
X X
X X
X X
X
X
X

HER ER DE BOGSTAVER DU HAR BRUGT:

E,D,G,H,V,I

V--E

GÆT ET BOGSTAV ?U

DESVÆRRE, DET BOGSTAV ER IKKE I ORDET.

LAD OS NU TEGNE DET HØJRE BEN.

XXXXXXX
X X
X ---
X) (. .) (
X) --- (
X) X (
X)X(
X X
X X
X X
X (
X (
X

HER ER DE BOGSTAVER DU HAR BRUGT:

E,D,G,H,V,I,U

V--E

GÆT ET BOGSTAV ?Æ

DESVÆRRE, DET BOGSTAV ER IKKE I ORDET.

DENNE GANG ER DET VENSTRE BEN.

XXXXXXX
X X
X ---
X) (. .) (
X) --- (
X) X (
X)X(
X X
X X
X X
X (
X (
X

HER ER DE BOGSTAVER DU HAR BRUGT:

E,D,G,H,V,I,U,Æ

V--E

GÆT ET BOGSTAV ?A

```

VA-E
GÆT HVILKET ORD DET ER ?VARE

FORKERT ORD. GÆT ET BOGSTAV IGEN.

HER ER DE BOGSTAVER DU HAR BRUGT:

E,D,G,H,V,I,U,Æ,A

VA-E

GÆT ET BOGSTAV ?S

DU FANDT ORDET!

ØNSKER DU ET NYT ORD ?N

DET HAR VÆRET MORSOMT! FARVEL DU...

TAK FOR DENNEGANG.

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          HÆNGT MAND"
130 PRINT "-----"
140 PRINT
150 DIM P$(12,12),L$(20),D$(20),N$(26),U(50)
160 C = 1
170 N = 50
180 FOR I = 1 TO 20
190 D$(I) = "-"
200 NEXT I
210 M = 0
220 FOR I = 1 TO 26
230 N$(I) = ""
240 NEXT I
250 FOR I = 1 TO 12
260 FOR J = 1 TO 12
270 P$(I,J) = " "
280 NEXT J
290 NEXT I
300 FOR I = 1 TO 12
310 P$(I,1) = "X"
320 NEXT I
330 FOR I = 1 TO 7
340 P$(1,I) = "X"
350 NEXT I
360 P$(2,7) = "X"
370 IF C < N THEN 400
380 PRINT "DU KLAREDE ALLE ORDENE!!"
390 GOTO 2170
400 Q = INT (N * RND (1)) + 1
410 IF U(Q) = 1 THEN 400
420 U(Q) = 1
430 C = C + 1
440 RESTORE
450 T1 = 0
460 FOR I = 1 TO Q
470 READ A$
480 NEXT I
490 L = LEN (A$)
500 FOR I = 1 TO LEN (A$)
510 L$(I) = MID$(A$,I,1)
520 NEXT I
530 PRINT "HER ER DE BOGSTAVER DU HAR BRUGT:"
540 PRINT
550 FOR I = 1 TO 26
560 PRINT N$(I);
570 IF N$(I + 1) = "" THEN 600
580 PRINT ", ";
590 NEXT I
600 PRINT
610 PRINT
620 FOR I = 1 TO L
630 PRINT D$(I);
640 NEXT I
650 PRINT
660 PRINT
670 PRINT "GÆT ET BOGSTAV ";
680 INPUT G$
690 R = 0
700 IF LEN (G$) < > 1 THEN 660
710 FOR I = 1 TO 26
720 IF N$(I) = "" THEN 770

```

```

730 IF G$ = N$(I) THEN 2040
740 NEXT I
750 PRINT "PROGRAM-FEJL.  KØR PROGRAMET IGEN."
760 GOTO 2170
770 N$(I) = G$
780 T1 = T1 + 1
790 FOR I = 1 TO L
800 IF L$(I) = G$ THEN 840
810 NEXT I
820 IF R = 0 THEN 870
830 GOTO 890
840 D$(I) = G$
850 R = R + 1
860 GOTO 810
870 M = M + 1
880 GOTO 1200
890 FOR I = 1 TO L
900 IF D$(I) = "-" THEN 930
910 NEXT I
920 GOTO 1160
930 PRINT
940 FOR I = 1 TO L
950 PRINT D$(I);
960 NEXT I
970 PRINT
980 PRINT
990 PRINT "GÆT HVILKET ORD DET ER ";
1000 INPUT B$
1010 PRINT
1020 IF LEN (B$) < > L THEN 990
1030 IF A$ = B$ THEN 1070
1040 PRINT "FORKERT ORD. GÆT ET BOGSTAV IGEN."
1050 PRINT
1060 GOTO 530
1070 PRINT "RIGTIGT! PR KUN";T1;"GÆT."
1080 PRINT
1090 PRINT "ØNSKER DU ET NYT ORD ";
1100 INPUT W$
1110 PRINT
1120 IF LEFT$(W$,1) = "J" THEN 180
1130 PRINT
1140 PRINT "DET HAR VÆRET MORSOMT! FARVEL DU..."
1150 GOTO 2170
1160 PRINT
1170 PRINT "DU FANDT ORDET!"
1180 PRINT
1190 GOTO 1090
1200 PRINT
1210 PRINT "DESVÆRRE, DET BOGSTAV ER IKKE I ORDET."
1220 PRINT
1230 ON M GOTO 1240,1260,1280,1300,1320,1340,1360,
1380,1400,1420
1240 PRINT "ALLERFØRST TEGNER VI ET HOVED."
1250 GOTO 1430
1260 PRINT "NU TEGNER VI EN KROP."
1270 GOTO 1430
1280 PRINT "SÅ TEGNER VI EN ARM."
1290 GOTO 1430
1300 PRINT "DENNE GANG TEGNER VI DEN ANDEN ARM."
1310 GOTO 1430
1320 PRINT "LAD OS NU TEGNE DET HØJRE BEN."
1330 GOTO 1430
1340 PRINT "DENNE GANG ER DET VENSTRE BEN."
1350 GOTO 1430
1360 PRINT "NU RÆKKER VI HÅNDEN OP."
1370 GOTO 1430
1380 PRINT "OG NU RÆKKER VI DEN ANDEN HÅND OP."
1390 GOTO 1430
1400 PRINT "NU TEGNER VI EN FOD."
1410 GOTO 1430
1420 PRINT "HER ER DEN ANDEN FOD --- DU ER HÆNGT!!!"
1430 PRINT
1440 ON M GOTO 1450,1560,1600,1640,1690,1720,1750,
1770,1790,1820
1450 P$(3,6) = "-"
1460 P$(3,7) = "-"
1470 P$(3,8) = "-"
1480 P$(4,5) = "("
1490 P$(4,6) = "."
1500 P$(4,8) = "."
1510 P$(4,9) = ")"
1520 P$(5,6) = "-"
1530 P$(5,7) = "-"
1540 P$(5,8) = "-"
1550 GOTO 1840
1560 FOR I = 6 TO 9
1570 P$(I,7) = "X"
1580 NEXT I

```

```


1590 GOTO 1840
1600 FOR I = 4 TO 7
1610 P$(I,I - 1) = ")"
1620 NEXT I
1630 GOTO 1840
1640 P$(4,11) = "("
1650 P$(5,10) = "("
1660 P$(6,9) = "("
1670 P$(7,8) = "("
1680 GOTO 1840
1690 P$(10,6) = "("
1700 P$(11,5) = "("
1710 GOTO 1840
1720 P$(10,8) = ")"
1730 P$(11,9) = ")"
1740 GOTO 1840
1750 P$(3,11) = ")"
1760 GOTO 1840
1770 P$(3,3) = "("
1780 GOTO 1840
1790 P$(12,10) = ")"
1800 P$(12,11) = "-"
1810 GOTO 1840
1820 P$(12,3) = "-"
1830 P$(12,4) = "("
1840 FOR I = 1 TO 12
1850 FOR J = 1 TO 12
1860 PRINT P$(I,J);
1870 NEXT J
1880 PRINT
1890 NEXT I
1900 PRINT
1910 PRINT
1920 IF M < > 10 THEN 530

```

```

1930 PRINT "DESVÆRRE, DU HAR TABT....."
1940 PRINT
1950 PRINT "ORDET VAR: ";A$
1960 PRINT
1970 PRINT "DET KUNNE DU IKKE GÆTTE."
1980 PRINT
1990 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
2000 INPUT Y$
2010 PRINT
2020 IF LEFT$(Y$,1) = "J" THEN 180
2030 GOTO 2170
2040 PRINT
2050 PRINT "DET BOGSTAV HAR DU GÆTTET FØR!"
2060 PRINT
2070 GOTO 530
2080 DATA "GUL","VEJ","SUR","DUM","VIN","FLY","SKY"
2090 DATA "GRIM","MINE","RUND","HEST","HUND",
"GLAS","VASE"
2100 DATA "PIZZA","METER","CYKEL","PUMPE","FISKE",
"SALAT","VIDEN"
2110 DATA "BUDGET","MORSOM","HASARD","BIOTOP",
"RØDVIN","GUITAR"
2120 DATA "HØFEBER","FYRSKIB","KÆNGURU","TREKANT",
"DATAMAT","BOGREOL"
2130 DATA "DISKETTE","EKSEMPEL","FRIKTION",
"FODSVAMP","MAPPEDYR"
2140 DATA "HØJTTALER","INFLUENZA","ATOMBOMBE",
"SOVETRYNE","METERVARE"
2150 DATA "SILDESALAT","SKIZOFRENI","RESTAURANT",
"CYKELLYGTE"
2160 DATA "PLANTESKOLE","STJERNEKRIG","BÄNDOPTAGER"
2170 PRINT
2180 PRINT "TAK FOR DENNEGANG."
2190 END

```



Højt-lavt

Dette program er en variant af programmet GÆT, men i stedet for bare at gætte på et tal mellem 1 og 100, så vinder du i dette spil penge, hvis du gætter rigtigt. Ifølge forfatteren er instruktionerne følgende:

1. Der findes en sum penge (mellem 1 og 100 kroner) i spillets jackpot.
2. Du har seks forsøg, hvori du skal prøve at gætte jackpot-beløbet.
3. Efter hvert gæt fortæller datamaten, om dit gæt var for højt eller for lavt.
4. Hvis du ikke har gættet det rigtige beløb på seks gæt, så udskriver datamaten beløbet.
5. Hvis du gætter beløbet, så noteres du for beløbet.
6. Efter hver omgang har du mulighed for at gætte igen eller for at afslutte programmet. Hvis du vælger at spille igen, så vil der komme et nyt beløb i jackpotten.
7. Hvis du vinder mere end en gang, vil dine gevinster blive lagt sammen.

Forfatteren til programmet er Dean Altman fra Fort Worth, Texas.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
HØJT-LAVT  
-----
```

```
VELKOMMEN TIL SPILLET 'HØJT-LAVT'.
```

```
I DETTE SPIL HAR DU 6 FORSØG, HVOR DU  
SKAL PRØVE AT GÆTTE DET ANTAL PENGE,  
SOM JACKPOTTEN INDEHOLDER.
```

```
JACKPOTTEN INDEHOLDER FRA 1 TIL 100 KR.
```

```
HVIS DU GÆTTER BELØBET FÅR DU ALLE  
PENGENE I JACKPOTTEN, OG RETTEN TIL ET  
NYT FORSØG. MEN HVIS DU IKKE KAN GÆTTE  
BELØBET, STANDSER SPILLET.
```

```
HVAD GÆTTER DU PÅ ?50  
DIT GÆT ER FOR LAVT.
```

```
HVAD GÆTTER DU PÅ ?75  
DIT GÆT ER FOR HØJT.
```

```
HVAD GÆTTER DU PÅ ?65  
DIT GÆT ER FOR LAVT.
```

```
HVAD GÆTTER DU PÅ ?70  
DIT GÆT ER FOR LAVT.
```

```
HVAD GÆTTER DU PÅ ?73
```

```
TILLYKKE! DU RAMTE RIGTIGT !!!!
```

```
DU VINDER 73 KRONER.
```

```
DU HAR INDTIL NU IALT VUNDET 73 KR.
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
```

```
HVAD GÆTTER DU PÅ ?50  
DIT GÆT ER FOR HØJT.
```

```
HVAD GÆTTER DU PÅ ?25  
DIT GÆT ER FOR LAVT.
```

```
HVAD GÆTTER DU PÅ ?30  
DIT GÆT ER FOR HØJT.
```

```
HVAD GÆTTER DU PÅ ?27
```

```
TILLYKKE! DU RAMTE RIGTIGT !!!!
```

```
DU VINDER 27 KRONER.
```

```
DU HAR INDTIL NU IALT VUNDET 100 KR.
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?N
```

```
FARVEL SÅ LÆNGE, JEG HÅBER DU HAR  
MORET DIG I MIT SELSKAB!!
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "          HØJT-LAVT"  
130 PRINT "-----"  
140 PRINT  
150 PRINT "VELKOMMEN TIL SPILLET 'HØJT-LAVT'. "  
160 PRINT  
170 PRINT "I DETTE SPIL HAR DU 6 FORSØG, HVOR DU"  
180 PRINT "SKAL PRØVE AT GÆTTE DET ANTAL PENGE,"  
190 PRINT "SOM JACKPOTTEN INDEHOLDER."  
200 PRINT  
210 PRINT "JACKPOTTEN INDEHOLDER FRA 1 TIL 100 KR."  
220 PRINT  
230 PRINT "HVIS DU GÆTTER BELØBET FÅR DU ALLE"  
240 PRINT "PENGENE I JACKPOTTEN, OG RETTEN TIL ET"  
250 PRINT "NYT FORSØG. MEN HVIS DU IKKE KAN GÆTTE"  
260 PRINT "BELØBET, STANDSER SPILLET."  
270 PRINT  
280 R = 0  
290 B = 0  
300 Y = INT (100 * RND (1))  
310 PRINT "HVAD GÆTTER DU PÅ ";  
320 INPUT A  
330 B = B + 1  
340 IF A = Y THEN 460  
350 IF A > Y THEN 380  
360 PRINT "DIT GÆT ER FOR LAVT."  
370 GOTO 390  
380 PRINT "DIT GÆT ER FOR HØJT."  
390 PRINT  
400 IF B < 6 THEN 310  
410 PRINT "DU KIKSEDE.....ÆRGELIGT!...."  
420 PRINT  
430 PRINT "JACKPOTTEN VAR PÅ";Y;"KRONER."  
440 R = 0  
450 GOTO 530  
460 PRINT  
470 PRINT "TILLYKKE! DU RAMTE RIGTIGT !!!!"  
480 PRINT  
490 PRINT "DU VINDER";Y;"KRONER."  
500 R = R + Y  
510 PRINT  
520 PRINT "DU HAR INDTIL NU IALT VUNDET";R;"KR."  
530 PRINT  
540 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";  
550 INPUT A$  
560 PRINT  
570 IF LEFT$(A$,1) = "J" THEN 290  
580 PRINT "FARVEL SÅ LÆNGE, JEG HÅBER DU HAR"  
590 PRINT "MORET DIG I MIT SELSKAB!!!"  
600 END
```

Høvding

Programmets forfatter, John Graham, siger, at Høvding er beregnet til at give folk (mest børn) øvelse i de fire regnearter (addition, subtraktion, multiplikation og division).

Det forsøger at gøre dette ved hjælp af underholdning, så hvis brugeren af programmet regner forkert, viser programmet, hvordan vedkommende skulle have gjort det.

Høvding er skrevet af John Graham fra Upper Brookville, New York.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
-----
                HØVDING
-----
```

```
MIT NAVN ER HØVDING ZIFFER KLOGG,
DEN STORE MATEMATIK-GUD.
```

```
DU HAR KALDT PÅ MIG....ER DU PARAT TIL
AT UDSTR PRØVELSERNE ?J
```

```
TAG ET TAL OG LÆG 3 TIL,
DIVIDER RESULTATET MED 5,
GANG MED 8,
DIVIDER MED 5 OG LÆG 5 TIL,
OG TRÆK TILSLUT 1 FRA.
```

```
HVILKET RESULTAT FIK DU ?23
```

```
JEG TØR VÆDDE PÅ, AT DET TAL DU
BEGYNDTE MED VAR: 56.375
```

```
HAR JEG RET I DET ?J
```

```
FARVEL.....
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                HØVDING"
130 PRINT "-----"
140 PRINT
150 PRINT "MIT NAVN ER HØVDING ZIFFER KLOGG,"
160 PRINT "DEN STORE MATEMATIK-GUD."
170 PRINT
180 PRINT "DU HAR KALDT PÅ MIG....ER DU PARAT TIL"
190 PRINT "AT UDSTR PRØVELSERNE ";
200 INPUT A$
210 PRINT
220 IF LEFT$(A$,1) = "J" THEN 260
230 PRINT "HOLD MUND, BLEGFJÆS MED UVIIS TUNGE."
240 PRINT "DER ER INGEN VEJ TILBAGE...."
250 PRINT
260 PRINT "TAG ET TAL OG LÆG 3 TIL,"
270 PRINT "DIVIDER RESULTATET MED 5,"
280 PRINT "GANG MED 8,"
290 PRINT "DIVIDER MED 5 OG LÆG 5 TIL,"
300 PRINT "OG TRÆK TILSLUT 1 FRA."
310 PRINT
320 PRINT "HVILKET RESULTAT FIK DU ";
330 INPUT B
340 PRINT
350 LET C = (B + 1 - 5) * 5 / 8 * 5 - 3
360 PRINT "JEG TØR VÆDDE PÅ, AT DET TAL DU"
370 PRINT "BEGYNDTE MED VAR:";C
380 PRINT
390 PRINT "HAR JEG RET I DET ";
400 INPUT D$
410 PRINT
420 IF LEFT$(D$,1) = "J" THEN 850
430 PRINT "HVAD VAR DET SÅ FOR ET TAL ";
440 INPUT K
450 F = K + 3
460 G = F / 5
470 H = G * 8
480 I = H / 5 + 5
490 J = I - 1
500 PRINT
510 PRINT "NÅEE, SÅ DU TROR DU ER SÅ KLOG...HVA'?"
520 PRINT "SÅ SE LIGE HER....."
530 PRINT
540 PRINT K;"PLUS 3 ER LIG MED";F
550 PRINT "HVILKET DIVIDERET MED 5 GIVER";G
560 PRINT "SOM GANGET MED 8 ER LIG";H;" ". HVIS"
570 PRINT "VI NU DELER MED 5 OG LÆGGER 5 TIL, SÅ"
580 PRINT "FÅR VI";I;" ". OG NÅR VI SÅ TRÆKKER"
590 PRINT "1 FRA, BLIVER RESULTATET";J;" ".
600 PRINT
610 PRINT "GIVER DU MIG RET I DET ";
620 INPUT Z$
630 PRINT
640 IF LEFT$(Z$,1) = "J" THEN 850
650 PRINT "NU HAR DU GJORT MIG MEGET, MEGET GAL!!"
660 PRINT "GID DU MÅ BLIVE RAMT AF LYNET....."
670 PRINT
680 FOR X = 30 TO 22 STEP - 1
690 PRINT TAB(X);"X X"
700 NEXT X
710 PRINT TAB(21);"X XXX"
720 PRINT TAB(20);"X  X"
730 PRINT TAB(19);"XXX X"
740 FOR Y = 20 TO 13 STEP - 1
750 PRINT TAB(Y);"X X"
760 NEXT Y
770 PRINT TAB(12);"XX"
780 PRINT TAB(11);"X"
790 PRINT TAB(10);"*"
800 PRINT
810 PRINT "-----"
820 PRINT
830 PRINT "NU TROR DU VIST MERE PÅ HVAD JEG SIGER!?"
840 GOTO 860
850 PRINT "FARVEL....."
860 END
```


Ishockey

Dette spil simulerer et ishockeyspil. Datamaskinen fungerer som rådgiver og dommer i spillet, der spilles mellem to menneskelige modstandere. Selvfølgelig kan én person spille begge sider.

Programmet beder om holdenes navne, spillernes navne og endda også om dommerens navn. Fire forskellige former for skud er tilladt, og de kan rettes mod fire forskellige områder på banen. Du bliver også spurgt om afleveringer. Spillet minder meget om virkeligheden med megen aktivitet, tacklinger, blokeringer, afleveringer, situationer med 4 spillere mod 2, osv. Desværre er der ingen mulighed for at blive udvist!

Den oprindelige forfatter til spillet er Robert Puopolo. Spillet er modificeret af Steve North fra Creative Computing.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

ISHOCKEY

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

SPØRGSMÅL SVAR

AFLEVERING INDTAST DET ANTAL AFLEVERINGER DU ØNSKER AT LAVERE, FRA 0 TIL 3.

SKUD INDTAST DET TAL DER SVARER TIL DET SKUD DU VIL BRUGE.
1 = SLAGSKUD
2 = HÅNDEDSSKUD
3 = BAGHÅNDEDSSKUD
4 = SMÅLDSKUD

OMRÅDE INDTAST DET TAL, DER SVARER TIL DET OMRÅDE DU RETTER DIT SKUD IMOD. INDTAST:
1 = ØVERSTE VENSTRE HJØRNE
2 = ØVERSTE HØJRE HJØRNE
3 = NEDERSTE VENSTRE HJØRNE
4 = NEDERSTE HØJRE HJØRNE

<< TAST 'RETURN' FOR AT FORTSÆTTE >> ?

NÅR SPILLET STARTER VIL DU BLIVE SPURGT OM SPILLERNE NAVNE. DE INDTASTES I FØLGENDE RÆKKEFØLGE: VENSTRE FORWARD, CENTERFORWARD, HØJRE FORWARD, VENSTRE BACK, HØJRE BACK, MÅLMAND.

DE ANDRE SPØRGSMÅL I SPILLET HAR FORKLARENDE TEKST.

HVAD HEDDER DE TO HOLD ?SMØRGOD, FEDUM

INDTAST DET ANTAL MINUTTER, SOM KAMPEN SKAL VARE ?3

HOLDKAPTJAJEN FOR SMØRGOD HOLDET BEDES INDTASTE SPILLERNE NAVNE NU:

SPILLER NUMMER 1 ?KNUD KNUBS
SPILLER NUMMER 2 ?FREDE MÅLSTRUP
SPILLER NUMMER 3 ?OTTO BOLD
SPILLER NUMMER 4 ?PETER PUCK
SPILLER NUMMER 5 ?SØREN SKUD
SPILLER NUMMER 6 ?MARTIN MÅL

HOLDKAPTJAJEN FOR FEDUM HOLDET BEDES GØRE DET SAMME:

SPILLER NUMMER 1 ?ERIK ETNUL
SPILLER NUMMER 2 ?FINN FINTE
SPILLER NUMMER 3 ?KARSTEN KUL
SPILLER NUMMER 4 ?RUNE HULRUM
SPILLER NUMMER 5 ?MOGENS MEDE
SPILLER NUMMER 6 ?BO BULEMÅL

INDTAST DOMMERENS NAVN ?IRWIN INJUST

SMØRGOD STILLER OP:

KNUD KNUBS
FREDE MÅLSTRUP
OTTO BOLD
PETER PUCK
SØREN SKUD
MARTIN MÅL

FEDUM STILLER OP:

ERIK ETNUL
FINN FINTE
KARSTEN KUL
RUNE HULRUM
MOGENS MEDE
BO BULEMÅL

VI ER KLAR TIL AFTENENS KAMP!
IRWIN INJUST VIL SMIDE PUCKEN MELLEML
FREDE MÅLSTRUP OG FINN FINTE.
SMØRGOD HAR KONTROL OVER PUCKEN.

HVOR MANGE AFLEVERINGER ?1

OTTO BOLD FINDER SØREN SKUD MED EN PERFEKT AFLEVERING.
SØREN SKUD BRYDER IGENNEM!

SKUD ?1

SØREN SKUD FYRER ET HELT FANTASTISK SLAGSKUD AF!!!!

SKUD MOD HVILKET OMRÅDE ?2

BO BULEMÅL FÅR PUCKEN LIGE I ANSIGTET!!
HAN ER SKADET.
BACKEN MOGENS MEDE DÆKKER FORAN MÅLET.

OG VI ER PARAT TIL FACE-OFF.
FEDUM HAR PUCKEN.

HVOR MANGE AFLEVERINGER ?2

NU ER DER 3 MAND MOD 2.
KUN PETER PUCK OG SØREN SKUD ER TILBAGE.
KARSTEN KUL AFLEVERER TIL ERIK ETNUL.
ERIK ETNUL DROPPER TIL MOGENS MEDE.

SKUD ?1

MOGENS MEDE FYRER ET HELT FANTASTISK SLAGSKUD AF!!!!

SKUD MOD HVILKET OMRÅDE ?4

RESULTAT FEDUM

STILLINGEN: FEDUM: 1
SMØRGOD: 0
MÅL SCORET AF: MOGENS MEDE
ASSISTERET AF: ERIK ETNUL
OG: KARSTEN KUL

OG VI ER PARAT TIL FACE-OFF.
SMØRGOD HAR KONTROL OVER PUCKEN.

HVOR MANGE AFLEVERINGER ?1

KNUD KNUBS FINDER SØREN SKUD MED EN PERFEKT AFLEVERING.
SØREN SKUD BRYDER IGENNEM!

SKUD ?2

SØREN SKUD RIVER ET BARSKT OG HÅRDT HÅNDEDSSKUD AF!!

SKUD MOD HVILKET OMRÅDE ?3

ET GODT UDSPARK AF BO BULEMÅL PUCKEN STYRES AF KARSTEN KUL
FEDUM HAR PUCKEN.

HVOR MANGE AFLEVERINGER ?2

NU ER DER 3 MAND MOD 2.
KUN PETER PUCK OG SØREN SKUD ER TILBAGE.
KARSTEN KUL AFLEVERER TIL RUNE HULRUM.
RUNE HULRUM DROPPER TIL ERIK ETNUL.

SKUD ?1

ERIK ETNUL FYRER ET HELT FANTASTISK SLAGSKUD AF!!!!

SKUD MOD HVILKET OMRÅDE ?4

SKRÆK OG RÆDSEL, ERIK ETNUL RAMMER HJØRNESTOLPEN TIL HØJRE FOR MARTIN MÅL, MENS MARTIN MÅLBESKYTTER MÅLET MOD DEN HERRELØSE PUCK!

+++++

NU LYDER SIRENEN.....KAMPEN ER FORBI

SLUTRESULTATET:

FEDUM: 1
SMØRGOD: 0

MÅLSTATISTIK FOR HOLD SMØRGOD

NAVN	MÅL	ASSISTERET
KNUD KNUBS	0	0
FREDE MÅLSTRUP	0	0
OTTO BOLD	0	0
PETER PUCK	0	0
SØREN SKUD	0	0

MÅLSTATISTIK FOR HOLD FEDUM

NAVN	MÅL	ASSISTERET
ERIK ETNUL	0	1
FINN FINTE	0	0
KARSTEN KUL	0	1
RUNE HULRUM	0	0
MOGENS MEDE	1	0

SKUD I NETTET: SMØRGOD: 2
FEDUM: 2

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                ISHOCKEY"
130 PRINT "-----"
140 X = 1
150 PRINT
160 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
170 INPUT C$
180 IF LEFT$(C$,1) = "N" THEN 230
190 IF LEFT$(C$,1) = "J" THEN 220
200 PRINT "SVAR 'J' FOR JA, 'N' FOR NEJ"
210 GOTO 160
220 GOTO 3730
230 DIM A$(7),B$(7),H(20),T(5),T1(5),T2(5),T3(5)
240 PRINT
250 PRINT "HVAD HEDDER DE TO HOLD ";
260 INPUT A$(7),B$(7)
270 PRINT
280 PRINT "INDTAST DET ANTAL MINUTTER,"
290 PRINT "SOM KAMPEN SKAL VARE ";
300 INPUT T6
310 PRINT
320 IF T6 < 1 THEN 280
330 PRINT "HOLDKAPTAJNEN FOR ";A$(7);" HOLDET"
340 PRINT "BEDES INDTASTE SPILLERNE NAVNE NU:"
350 PRINT
360 FOR I = 1 TO 6
370 PRINT "SPILLER NUMMER";I;
380 INPUT A$(I)
390 NEXT I
400 PRINT
410 PRINT "HOLDKAPTAJNEN FOR ";B$(7);" HOLDET"
420 PRINT "BEDES GØRE DET SAMME:"
430 PRINT
440 FOR T = 1 TO 6
450 PRINT "SPILLER NUMMER";T;
460 INPUT B$(T)
470 NEXT T
480 PRINT
490 PRINT "INDTAST DOMMERENS NAVN ";
500 INPUT R$
510 PRINT
520 PRINT A$(7);" STILLER OP:"
530 PRINT
540 FOR T = 1 TO 6
550 PRINT A$(T)
560 NEXT T
570 PRINT
580 PRINT B$(7);" STILLER OP:"
590 PRINT
600 FOR T = 1 TO 6
610 PRINT B$(T)
620 NEXT T
630 PRINT
640 PRINT
650 PRINT "VI ER KLAR TIL AFTENENS KAMP!"
660 PRINT R$" VIL SMIDE PUCKEN MELLEML"
670 PRINT A$(2);" OG ";B$(2);"."
680 FOR L = 1 TO T6
690 IF L = 1 THEN 720
700 PRINT
710 PRINT "OG VI ER PARAT TIL FACE-OFF."
720 C = INT (2 * RND (1)) + 1
730 ON C GOTO 740,760
740 PRINT A$(7);" HAR KONTROL OVER PUCKEN."
750 GOTO 770
760 PRINT B$(7);" HAR PUCKEN."
770 PRINT
780 PRINT "HVOR MANGE AFLEVERINGER ";
790 FOR N = 1 TO 3
800 H(N) = 0
810 INPUT P
820 PRINT
830 FOR N = 1 TO 3
840 H(N) = 0
850 NEXT N
860 IF P < 0 THEN 770
870 IF P > 3 THEN 770
880 FOR J = 1 TO (P + 2)
890 H(J) = INT (5 * RND (X)) + 1
900 NEXT J
910 IF H(J - 1) = H(J - 2) THEN 880
920 IF P + 2 < 3 THEN 950
930 IF H(J - 1) = H(J - 3) THEN 880
940 IF H(J - 2) = H(J - 3) THEN 880
950 IF P = 0 THEN 970
960 GOTO 1270
970 PRINT
980 PRINT "HVILKET SKUD ";
990 PRINT

```

```

1000 INPUT S
1010 IF S < 1 THEN 980
1020 IF S > 4 THEN 980
1030 ON C GOTO 1040,1230
1040 PRINT A$(H(J - 1));
1050 G = H(J - 1)
1060 G1 = 0:G2 = 0
1070 ON S GOTO 1080,1120,1160,1200
1080 PRINT " AFFYRER ET HÆRDT SKUD"
1090 PRINT "FRA DEN RØDE LINIE.!!"
1100 Z = 10
1110 GOTO 2030
1120 PRINT " FLIPPER ET"
1130 PRINT "LYNHURTIGT HÆNLEDSLAG HEN AD ISEN."
1140 Z = 17
1150 GOTO 2030
1160 PRINT "OG SKYDER ET BAGHÆNDSKUD MOD"
1170 PRINT "MÅLMANDEN."
1180 Z = 25
1190 GOTO 2030
1200 PRINT " SLÅR ET SKRUE ET SLAG"
1210 Z = 17
1220 GOTO 2030
1230 PRINT B$(H(J - 1));
1240 G1 = 0:G2 = 0
1250 G = H(J - 1)
1260 GOTO 1070
1270 ON C GOTO 1280,1530
1280 ON P GOTO 1290,1350,1430
1290 PRINT A$(H(J - 2));" FINDER ";A$(H(J - 1))
1300 PRINT "MED EN PERFEKT AFLEVERING."
1310 PRINT A$(H(J - 1));" BRYDER IGENNEM!"
1320 G = H(J - 1):G1 = H(J - 2)
1330 G2 = 0:Z1 = 3
1340 GOTO 1770
1350 PRINT A$(H(J - 2));" GIVER PUCKEN"
1360 PRINT "VIDERE TIL ";A$(H(J - 1));"."
1370 PRINT "DER NETOP PASSERER."
1380 PRINT A$(H(J - 3));" KASTER SIG OVER"
1390 PRINT B$(5);" OG ";B$(4);"."
1400 G = H(J - 3):G1 = H(J - 1):G2 = H(J - 2)
1410 Z1 = 2
1420 GOTO 1770
1430 PRINT "SR FOR SØREN.....DE ER 4 MAND MOD 2"
1440 PRINT A$(H(J - 3));" ER FORAN ";A$(H(J - 2));"."
1450 PRINT A$(H(J - 2));" VÆLTER SIG IND"
1460 PRINT "OVER MIDTLINIEN."
1470 PRINT A$(H(J - 2));" HAR PUCKEN,"
1480 PRINT "OG LÆGGER SIG BAG ";A$(H(J - 1))
1490 PRINT "GOD AFLEVERING TIL A$(H(J-1));"."
1500 G = H(J - 4):G1 = H(J - 1):G2 = H(J - 2)
1510 Z1 = 1
1520 GOTO 1770
1530 ON P GOTO 1540,1590,1670
1540 PRINT B$(H(J - 1));" SKYDER TIL ";B$(H(J - 2))
1550 PRINT "DER ER PÅ VEJ NED AF VENSTRE SIDE."
1560 G = H(J - 2):G1 = H(J - 1)
1570 G2 = 0:Z1 = 3
1580 GOTO 1770
1590 PRINT "NU ER DER 3 MAND MOD 2."
1600 PRINT "KUN ";A$(4);" OG ";A$(5)
1610 PRINT "ER TILBAGE."
1620 PRINT B$(H(J - 2));" AFLEVERER TIL ";B$(H(J - 1));"."
1630 PRINT B$(H(J - 1));" DROPPER TIL ";B$(H(J - 3));"."
1640 G = H(J - 3):G1 = H(J - 1):G2 = H(J - 2)
1650 Z1 = 2
1660 GOTO 1770
1670 PRINT "3-MOD-2 MED EN SPILLER I HÆLENE."
1680 PRINT B$(H(J - 4));" AFLEVERER TIL ";B$(H(J - 2));"."
1690 PRINT "DER HURTIGT GIVER PUCKEN VIDERE TIL"
1700 PRINT B$(H(J - 1));" DER AFLEVERER TVÆRS OVER"
1710 PRINT "MIDTLINIEN, HVOR DEN MODTAGES AF INGEN"
1720 PRINT "RINGERE END ";B$(H(J - 3));"!"
1730 PRINT "HOLD DA OP! HAN FØRSØGER ALENE!!!"
1740 G = H(J - 3):G1 = H(J - 1):G2 = H(J - 2)
1750 Z1 = 1
1760 GOTO 1770
1770 PRINT
1780 PRINT "SKUD ";
1790 INPUT S
1800 PRINT
1810 IF S > 4 THEN 1780
1820 IF S < 1 THEN 1780
1830 ON C GOTO 1840,2010
1840 PRINT A$(G);
1850 ON S GOTO 1860,1900,1940,1970
1860 PRINT " FYRER ET HELT FANTASTISK"
1870 PRINT "SLAGSKUD AF!!!!!"
1880 Z = 4:Z = Z + Z1

```

```


1890 GOTO 2030
1900 PRINT " RIVER ET BARSKT OG HÆRDT"
1910 PRINT "HÆNDELSSKUD AF!!"
1920 Z = 2:Z = Z + Z1
1930 GOTO 2030
1940 PRINT " FÆR ET FINTET BAGHÆNDELSSKUD IND."
1950 Z = 3:Z = Z + Z1
1960 GOTO 2030
1970 PRINT " FYRER ET OVERRASKENDE"
1980 PRINT "SMÆLDSKUD AF!!"
1990 Z = 2:Z = Z + Z1
2000 GOTO 2030
2010 PRINT B$(G);
2020 ON S GOTO 1860,1900,1940,1970
2030 PRINT
2040 PRINT "SKUD MOD HVILKET OMRÆDE ";
2050 INPUT A
2060 PRINT
2070 IF A < 1 THEN 2040
2080 IF A > 4 THEN 2040
2090 ON C GOTO 2100,2120
2100 S2 = S2 + 1
2110 GOTO 2130
2120 S3 = S3 + 1
2130 A1 = INT (4 * RND (1)) + 1
2140 IF A < > A1 THEN 2750
2150 H(20) = INT (100 * RND (1)) + 1
2160 IF INT (H(20) / Z) = H(20) / Z THEN 2650
2170 ON C GOTO 2180,2210
2180 PRINT "MÆL ";A$(7)
2190 H(9) = H(9) + 1
2200 GOTO 2230
2210 PRINT "RESULTAT ";B$(7)
2220 H(8) = H(8) + 1
2230 PRINT
2240 PRINT " STILLINGEN: ";
2250 IF H(8) > H(9) THEN 2310
2260 PRINT A$(7);";";H(9)
2270 PRINT " ";
2280 PRINT B$(7);";";H(8)
2290 PRINT
2300 GOTO 2340
2310 PRINT B$(7);";";H(8)
2320 PRINT " ";
2330 PRINT A$(7);";";H(9)
2340 ON C GOTO 2350,2500
2350 PRINT
2360 PRINT "MÆL SCORET AF: ";A$(G)
2370 IF G1 = 0 THEN 2420
2380 IF G2 = 0 THEN 2440
2390 PRINT "ASSISTERET AF: ";A$(G1)
2400 PRINT " OG: ";A$(G2)
2410 GOTO 2450
2420 PRINT " ALENE."
2430 GOTO 2450
2440 PRINT "ASSISTERET AF: ";A$(G1)
2450 T(G) = T(G) + 1
2460 T1(G1) = T1(G1) + 1
2470 T1(G2) = T1(G2) + 1
2480 GOTO 3240
2490 PRINT
2500 PRINT "MÆL SCORET AF: ";B$(G)
2510 IF G1 = 0 THEN 2560
2520 IF G2 = 0 THEN 2580
2530 PRINT "ASSISTERET AF: ";B$(G1)
2540 PRINT " OG: ";B$(G2)
2550 GOTO 2600
2560 PRINT " ALENE."
2570 GOTO 2600
2580 PRINT "ASSISTERET AF: ";B$(G1)
2590 GOTO 2600
2600 T2(G) = T2(G) + 1
2610 T3(G1) = T3(G1) + 1
2620 T3(G2) = T3(G2) + 1
2630 PRINT
2640 GOTO 3240
2650 A2 = INT (100 * RND (1)) + 1
2660 IF INT (A2 / 4) = A2 / 4 THEN 2680
2670 GOTO 2750
2680 ON C GOTO 2690,2720
2690 PRINT "REDDET AF ";B$(6);";";
2700 PRINT "PUCKEN ER PÆ BANEN IGEN."
2710 GOTO 2150
2720 PRINT "REDDET AF ";A$(6);";"; DER"
2730 PRINT "SENDER PUCKEN UD."
2740 GOTO 2150
2750 S1 = INT (6 * RND (1)) + 1
2760 ON C GOTO 2770,2990
2770 ON S1 GOTO 2780,2810,2850,2880,2920,2950
2780 PRINT "ET GODT UDSPARK AF ";B$(6)
2790 PRINT "PUCKEN STYRES AF ";B$(3)
2800 GOTO 720
2810 PRINT "HELT FÆNOMONAL REDNING MED HANDSKEN"
2820 PRINT "AF ";B$(6);";"; B$(6)
2830 PRINT "SLÆR PUCKEN UD BLANDT PUBLIKUM.""
2840 GOTO 3240
2850 PRINT "REDDET AF ";B$(6);";";S"
2860 PRINT "SKØJTEMEDER."
2870 GOTO 720
2880 PRINT "ET SKUD FRA ";A$(G);"; REDDES"
2890 PRINT "AF ";B$(6);";";S KNÆBESKYTTER."
2900 PRINT B$(6);"; DÆKKER."
2910 GOTO 3240
2920 PRINT " SKYDER EN HÆRD"
2930 PRINT "PUCK OVER HOVEDET PÆ ";B$(6);";"
2940 GOTO 720
2950 PRINT B$(6);"; FÆR PUCKEN LIGE I ANSIGTET!!"
2960 PRINT "HAN ER SKADET."
2970 PRINT "BACKEN ";B$(5);"; DÆKKER FORAN MÆLET."
2980 GOTO 3240
2990 ON S1 GOTO 3000,3030,3090,3130,3160,3200
3000 PRINT "REDDET AF ";A$(6);";S STAV."
3010 PRINT "PUCKEN SLÆRS UD AF ";A$(4);";"
3020 GOTO 720
3030 PRINT "SKRÆK OG RÆDSEL, ";B$(G)
3040 PRINT "RAMMER HJØRNESTOLPEN TIL"
3050 PRINT "HJØRE FOR ";A$(6);"; MENS"
3060 PRINT A$(6);";BESKYTTER MÆLET"
3070 PRINT "MOD DEN HERRELØGE PUCK!"
3080 GOTO 3240
3090 PRINT A$(6);"; REDDER MED SKØJTEN."
3100 PRINT A$(6);"; KNALDER PUCKEN UD"
3110 PRINT "BLANDT TILSKUERNE."
3120 GOTO 3240
3130 PRINT "REDDET AF ";A$(6);";";S"
3140 PRINT "STAV. HAN SLÆR PUCKEN VIDERE."
3150 GOTO 720
3160 PRINT "SPARKES UD AF ";A$(6);";"
3170 PRINT "MEN PUCKEN SPRINGER TILBÆGE,"
3180 PRINT "HELT UD PÆ MIDTERLINIEN."
3190 GOTO 720
3200 PRINT "REDDET AF ";A$(6);";";S"
3210 PRINT "HANDSKE. HAN HOLDER FAST PÆ"
3220 PRINT "PUCKEN."
3230 GOTO 3240
3240 NEXT L
3250 PRINT
3260 PRINT "+++++"
3270 PRINT
3280 PRINT "NU LYDER SIRENEN.....KAMPEN ER FORBI"
3290 PRINT
3300 PRINT TAB( 15);"SLUTRESULTATET:"
3310 IF H(8) > H(9) THEN 3350
3320 PRINT A$(7);";";H(9)
3330 PRINT B$(7);";";H(8)
3340 GOTO 3370
3350 PRINT B$(7);";";H(8)
3360 PRINT A$(7);";";H(9)
3370 PRINT
3380 PRINT "MÆLSTATISTIK FOR HOLD ";A$(7)
3390 PRINT
3400 PRINT "NAVN MÆL ASSISTERET"
3410 PRINT "-----"
3420 FOR I = 1 TO 5
3430 PRINT A$(I);
3440 FOR T = LEN (A$(I)) TO 20
3450 PRINT " ";
3460 NEXT T
3470 IF T(I) < 10 THEN PRINT " ";
3480 PRINT T(I);" ";
3490 IF T1(I) < 10 THEN PRINT " ";
3500 PRINT T1(I)
3510 NEXT I
3520 PRINT
3530 PRINT "MÆLSTATISTIK FOR HOLD ";B$(7)
3540 PRINT
3550 PRINT "NAVN MÆL ASSISTERET"
3560 PRINT "-----"
3570 FOR I = 1 TO 5
3580 PRINT B$(I);
3590 FOR T = LEN (B$(I)) TO 20
3600 PRINT " ";
3610 NEXT T
3620 IF T2(I) < 10 THEN PRINT " ";
3630 PRINT T2(I);" ";
3640 IF T3(I) < 10 THEN PRINT " ";
3650 PRINT T3(I)
3660 NEXT I

```

```

3670 PRINT
3680 PRINT "SKUD I NETTET: ";
3690 PRINT A$(7);":";S2
3700 PRINT "      ";
3710 PRINT B$(7);":";S3
3720 GOTO 4070
3730 PRINT
3740 PRINT "SPØRGSMÅL   SVAR"
3750 PRINT "-----"
3760 PRINT "AFLEVERING   INDTAST DET ANTAL AFLEVE-"
3770 PRINT "      RINGER DU ØNSKER AT LAVE,"
3780 PRINT "      FRA 0 TIL 3.
3790 PRINT
3800 PRINT "SKUD           INDTAST DET TAL DER SVARER"
3810 PRINT "      TIL DET SKUD DU VIL BRUGE."
3820 PRINT "      1 = SLAGSKUD"
3830 PRINT "      2 = HÅNDESSKUD"
3840 PRINT "      3 = BAGHÅNDESSKUD"
3850 PRINT "      4 = SMÅLDSKUD"
3860 PRINT
3870 PRINT "OMRÅDE        INDTAST DET TAL, DER SVARER"
3880 PRINT "      TIL DET OMRÅDE DU RETTER"
3890 PRINT "      DIT SKUD IMOD. INDTAST:"
3900 PRINT "      1 = ØVERSTE VENSTRE HJØRNE"
3910 PRINT "      2 = ØVERSTE HØJRE HJØRNE"
3920 PRINT "      3 = NEDERSTE VENSTRE HJØRNE"
3930 PRINT "      4 = NEDERSTE HØJRE HJØRNE"
3940 PRINT
3950 PRINT "<< TAST 'RETURN' FOR AT FORTSÆTTE >> ";
3960 INPUT C$
3970 PRINT
3980 PRINT "NÅR SPILLET STARTER VIL DU BLIVE SPURGT"
3990 PRINT "OM SPILERNES NAVNE. DE INDFASTES I FØL-"
4000 PRINT "GENE RÆKKEFØLGE: VENSTRE FORWARD, CEN-"
4010 PRINT "TERFORWARD, HØJRE FORWARD, VENSTRE BACK"
4020 PRINT "HØJRE BACK, MÅLMAND."
4030 PRINT
4040 PRINT "DE ANDRE SPØRGSMÅL I SPILLET HAR FOR-"
4050 PRINT "KLARENDE TEKST."
4060 GOTO 230
4070 END

```



Kalender

Dette program udskriver en kalender for et givet år. I programlinie 400 skal du selv huske at indføre ugedagen, som den første dag i det ønskede år har. (Søndag = 0, Mandag = 1, Tirsdag = 2, Onsdag = 3, Torsdag = 4, Fredag = 5, Lørdag = 6). I programlinie 200 indføres året. Ved at bruge programmet UGEDAG kan du let bestemme, hvilken ugedag den 1. januar i et givet år falder på. Du er også nødt til at foretage to ærinder i programlinierne 260 og 330, hvor der tages højde for skudår. Programlistningen beskriver de rettelser, der skal foretages. Når programmet køres, udskrives en nydelig 12 måneders kalender.

Programmet er skrevet af Geoffrey Chase fra The Abbey, Portsmouth, Rhode Island.

MORRISTOWN	CREATIVE COMPUTING	NEW JERSEY

KALENDER		

** 0 ***** JANUAR 1978 ***** 365 **

S	M	T	O	T	F	L

1	2	3	4	5	6	7
8	9	10	11	12	13	14
15	16	17	18	19	20	21
22	23	24	25	26	27	28
29	30	31				

** 31 ***** FEBRUAR 1978 ***** 334 **

S	M	T	O	T	F	L

			1	2	3	4
5	6	7	8	9	10	11
12	13	14	15	16	17	18
19	20	21	22	23	24	25
26	27	28				

** 59 ***** MARTS 1978 ***** 306 **

S	M	T	O	T	F	L

			1	2	3	4
5	6	7	8	9	10	11
12	13	14	15	16	17	18
19	20	21	22	23	24	25
26	27	28	29	30	31	

** 90 ***** APRIL 1978 ***** 275 **

S	M	T	O	T	F	L

						1
2	3	4	5	6	7	8
9	10	11	12	13	14	15
16	17	18	19	20	21	22
23	24	25	26	27	28	29
30						

** 120 ***** MAJ 1978 ***** 245 **

S	M	T	O	T	F	L

	1	2	3	4	5	6
7	8	9	10	11	12	13
14	15	16	17	18	19	20
21	22	23	24	25	26	27
28	29	30	31			

** 151 ***** JUNI 1978 ***** 214 **

S	M	T	O	T	F	L

				1	2	3
4	5	6	7	8	9	10
11	12	13	14	15	16	17
18	19	20	21	22	23	24
25	26	27	28	29	30	

** 181 ***** JULI 1978 ***** 184 **

S	M	T	O	T	F	L

						1
2	3	4	5	6	7	8
9	10	11	12	13	14	15
16	17	18	19	20	21	22
23	24	25	26	27	28	29
30	31					

** 212 ***** AUGUST 1978 ***** 153 **

S	M	T	O	T	F	L

			1	2	3	4
5	6	7	8	9	10	11
12	13	14	15	16	17	18
19	20	21	22	23	24	25
26	27	28	29	30	31	

** 243 ***** SEPTEMBER 1978 ***** 122 **

S	M	T	O	T	F	L

						1
2	3	4	5	6	7	8
9	10	11	12	13	14	15
16	17	18	19	20	21	22
23	24	25	26	27	28	29
30	31					

** 273 ***** OKTOBER 1978 ***** 92 **

S	M	T	O	T	F	L

						1
2	3	4	5	6	7	8
9	10	11	12	13	14	15
16	17	18	19	20	21	22
23	24	25	26	27	28	29
30	31					

** 304 ***** NOVEMBER 1978 ***** 61 **

S M T O T F L

			1	2	3	4
5	6	7	8	9	10	11
12	13	14	15	16	17	18
19	20	21	22	23	24	25
26	27	28	29	30		

** 334 ***** DECEMBER 1978 ***** 31 **

S M T O T F L

					1	2
3	4	5	6	7	8	9
10	11	12	13	14	15	16
17	18	19	20	21	22	23
24	25	26	27	28	29	30
31						

```

100 PRINT " MORRISTOWN CREATIVE COMPUTING NEW JERSEY "
110 PRINT "-----"
120 PRINT " KALENDER"
130 PRINT "-----"
140 PRINT
150 REM Her skal der indsættes værdierne for år, antal dage pr. år og
160 REM antal dage i hver måned inden programmet køres.
170 REM
180 REM Variabel Y skal sættes til året (her er brugt værdier for 1978).
190 REM
200 Y = 1978
210 REM
220 REM Variabel A skal sættes til antal dage i året.
230 REM Dvs. enten 365 eller 366 (hvis året er et skudår).
240 REM Der er anvendt værdi for 1978.
250 REM
260 A = 365
270 REM
280 REM I nedenstående DATA-sætning erstattes den tredje værdi (28)
290 REM med 29, hvis der er tale om et skudår. Den tredje værdi
300 REM skal svare til antal dage i februar måned.
310 REM Dvs. 0,31,29,31,..... på skudår
320 REM
330 DATA 0,31,28,31,30,31,30,31,31,30,31,30,31
340 REM
350 REM Variabel S skal sættes til den første ugedag i året, dvs.:
360 REM 0=søndag, 1=mandag, 2=tirsdag, 3=onsdag
370 REM 4=torsdag, 5=fredag, 6=lørdag
380 REM Første dag i 1978 (dvs. 1. januar 1978 er en søndag)
390 REM
400 S = 0
410 REM
420 PRINT
430 DIM M(12)
440 FOR I = 1 TO 4
450 PRINT
460 NEXT I
470 D = 0
480 REM indlæs hver måneds dage
490 FOR N = 0 TO 12
500 READ M(N)
510 NEXT N
520 FOR N = 1 TO 12
530 PRINT
540 PRINT
550 S = S + M(N - 1)
560 PRINT "***";S; TAB( 7);
570 FOR I = 1 TO 18
580 PRINT "*";
590 NEXT I
600 ON N GOTO 610,620,630,640,650,660,670,680,690,700,710,720
610 PRINT " JANUAR ";Y;: GOTO 730
620 PRINT " FEBRUAR ";Y;: GOTO 730
630 PRINT " MARTS ";Y;: GOTO 730
640 PRINT " APRIL ";Y;: GOTO 730
650 PRINT " MAJ ";Y;: GOTO 730
660 PRINT " JUNI ";Y;: GOTO 730
670 PRINT " JULI ";Y;: GOTO 730
680 PRINT " AUGUST ";Y;: GOTO 730
690 PRINT "SEPTEMBER";Y;: GOTO 730
700 PRINT " OKTOBER ";Y;: GOTO 730
710 PRINT " NOVEMBER";Y;: GOTO 730
720 PRINT " DECEMBER";Y;: GOTO 730
730 FOR I = 1 TO 13
740 PRINT "*";
750 NEXT I
760 PRINT A - S;"*"
770 PRINT
780 PRINT " S M T O T F L"
790 PRINT
800 FOR I = 1 TO 59
810 PRINT "*";
820 NEXT I
830 PRINT
840 FOR W = 1 TO 6
850 PRINT
860 PRINT TAB( 4)
870 FOR G = 1 TO 7
880 D = D + 1
890 D2 = D - S
900 IF D2 > M(N) THEN 960
910 IF D2 > 0 THEN PRINT D2;
920 PRINT TAB( 4 + 8 * G);
930 NEXT G
940 IF D2 = M(N) THEN 990
950 NEXT W
960 D = D - G
970 PRINT
980 PRINT
990 NEXT N
1000 FOR I = 1 TO 6
1010 PRINT
1020 NEXT I
1030 END

```

Kamp

I dette spil udkæmper du en mindre krig mod datamaten. Du har 72.000 soldater til rådighed, som du først skal fordele mellem din hær, dit flyvevåben og din flåde. Du kan fordele dem, ligesom du vil, bare det samlede antal soldater ikke overstiger 72.000.

Du angriber din modstander (datamaten) og angiver, hvor mange mand fra hæren, flyvevåbenet eller flåden du sætter ind. Derefter fortæller datamaten, hvad resultatet af slaget er blevet. Du får en ny udskrift af dine soldaters fordeling og bestemmer derefter næste træk.

Efter det andet slag bestemmes det ud fra tallene, om du har vundet eller tabt, eller om der bliver fred.

Dette program blev skrevet af Bob Dores, Milton, Massachusetts.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

KAMP

JEG ER I KRIG MED DIG. VI HAR 72000
SOLDATER HVER. DISSE SKAL FORDELES.

	MIG	DIG
HÆREN	30000	?25000
FLÅDEN	20000	?25000
FLYVEVÅBEN	22000	?22000

DU ANGRIBER FØRST.
(1=HÆR, 2=FLÅDE, 3=FLYVEVÅBEN) ?3

HVOR MANGE SOLDATER ?22000

DU UDSLETTEDE EN AF MINE HÆRPATROLJER,
MEN JEG ØDELAGDE 2 AF DINE FLÅDEBASER
OG 3 HÆRBASER.

	DIG	MIG
HÆREN	6250	20000
FLÅDEN	8333	20000
FLYVEVÅBEN	22000	22000

HVAD ER DIT NÆSTE TRÆK ?
(1=HÆR, 2=FLÅDE, 3=FLYVEVÅBEN) ?3

HVOR MANGE SOLDATER ?20000

MIN FLÅDE OG MIT FLYVEVÅBEN HAR I ET
FÆLLES ANGREB FORVANDLET DIT FØDELAND
TIL EN ØRKEN.


RESULTATET AF DINE 2 ANGREB FØRTE TIL,
AT DU TABTE. JEG HAR BESEJRET DIT LAND!
DET HAR DU GODT AF, DU KUNNE JO BARE
LADE VÆRE MED AT SPILLE DET HER
FJOLLEDE SPIL!!!!

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          KAMP"
130 PRINT "-----"
140 PRINT
150 PRINT "JEG ER I KRIG MED DIG. VI HAR 72000"
160 PRINT "SOLDATER HVER. DISSE SKAL FORDELES."
170 PRINT
180 PRINT "  ", "MIG", "DIG"
190 PRINT "HÆREN", 30000,
200 INPUT N
210 PRINT "FLÅDEN", 20000,
220 INPUT B
230 PRINT "FLYVEVÅBEN", 22000,
240 INPUT C
250 IF N + B + C > 72000 THEN 170
260 D = 30000: E = 20000: F = 22000
270 PRINT
280 PRINT "DU ANGRIBER FØRST."
290 PRINT "(1=HÆR, 2=FLÅDE, 3=FLYVEVÅBEN) ";
300 INPUT Y
310 PRINT
320 PRINT "HVOR MANGE SOLDATER ";
330 INPUT X
340 PRINT
350 IF X < 0 THEN 320
360 ON Y GOTO 370, 490, 670
370 IF X > N THEN 320
380 IF X < N / 3 THEN 410
390 IF X < 2 * N / 3 THEN 440
400 GOTO 600
410 PRINT "DU MISTEDE"; X; "SOLDATER."
420 N = INT (N - X)
430 GOTO 860
440 PRINT "DU TABTE"; INT (X / 3); "SOLDATER, MEN"
450 PRINT "JEG TABTE"; INT (2 * D / 3)
460 N = INT (N - X / 3)
470 D = 0
480 GOTO 860
490 IF X > B THEN 320
500 IF X < E / 3 THEN 530
510 IF X < 2 * E / 3 THEN 560
520 GOTO 600
530 PRINT "DIT ANGREB BLEV STANDSET!"
540 B = INT (B - X)
550 GOTO 860
560 PRINT "DU DRÆBTE"; INT (2 * E / 3);
570 PRINT "SOLDATER I MIN HÆR."
580 E = INT (E / 3)
590 GOTO 860
600 PRINT "DU SÆNKEDE 1 AF MINE PATROLJEØRDE, MEN"
610 PRINT "JEG UDSLETTEDE 2 AF DINE FLYVEBASER OG"
620 PRINT "3 HÆRBASER."
630 N = INT (N / 3)
640 C = INT (C / 3)
650 E = INT (2 * E / 3)
660 GOTO 860
670 IF X > C THEN 320
680 IF X < C / 3 THEN 710
690 IF X < 2 * C / 3 THEN 740
700 GOTO 800
710 PRINT "DIT ANGREB MISLYKKEDES."
720 C = INT (C - X)
730 GOTO 860
740 PRINT "VI KOM I NÆRKAMP.....DU VANDT,"
750 PRINT "OG DERMED ER DIN MISSION FULDFØRT."
760 D = INT (2 * D / 3)
770 E = INT (E / 3)
780 F = INT (F / 3)
790 GOTO 860
800 PRINT "DU UDSLETTEDE EN AF MINE HÆRPATROLJER,"
810 PRINT "MEN JEG ØDELAGDE 2 AF DINE FLÅDEBASER"
820 PRINT "OG 3 HÆRBASER."
830 N = INT (N / 4)
840 B = INT (B / 3)
850 D = INT (2 * D / 3)
860 PRINT
870 PRINT "  ", "DIG", "MIG"
880 PRINT "HÆREN", N, D
890 PRINT "FLÅDEN", B, E
900 PRINT "FLYVEVÅBEN", C, F
910 PRINT
920 PRINT "HVAD ER DIT NÆSTE TRÆK ?"
930 PRINT "(1=HÆR, 2=FLÅDE, 3=FLYVEVÅBEN) ";
940 INPUT G
950 PRINT
960 PRINT "HVOR MANGE SOLDATER ";
970 INPUT T
```

```

980 PRINT
990 IF T < 0 THEN 960
1000 ON G GOTO 1010,1090,1220
1010 IF T > N THEN 960
1020 IF T < D / 2 THEN 1060
1030 PRINT "DU TILINTETGJORDE MIN HÆR!"
1040 D = 0
1050 GOTO 1370
1060 PRINT "JEG ØDELAGDE DIT ANGREB TOTALT!"
1070 N = N - T
1080 GOTO 1370
1090 IF T > B THEN 960
1100 IF T < E / 2 THEN 1120
1110 GOTO 1170
1120 PRINT "JEG SÆNKEDE 2 AF DINE KRIGSSKIBE, OG"
1130 PRINT "UDRADEREDE DIN UBEVOGTEDE HOVEDSTAD."
1140 N = N / 4
1150 B = B / 2
1160 GOTO 1370
1170 PRINT "DIN FLÅDE SKØD 3 AF MINE FLY NED, OG"
1180 PRINT "DU HAR SÆNKET 3 AF MINE KRIGSSKIBE."
1190 F = 2 * F / 3
1200 E = (E / 2)
1210 GOTO 1370
1220 IF T > C THEN 960
1230 IF T > F / 2 THEN 1250
1240 GOTO 1320
1250 PRINT "MIN FLÅDE OG MIT FLYVEVÅBEN HAR I ET"
1260 PRINT "FÆLLES ANGREB FORVANDLET DIT FØDELAND"
1270 PRINT "TIL EN ØRKEN."
1280 N = N / 3
1290 B = B / 3
1300 C = C / 3
1310 GOTO 1370
1320 PRINT "EN AF DINE FLYVEMASKINER ER FALDET NED"
1330 PRINT "I DET HUS JEG BOR I!.....JEG ER DØD."
1340 PRINT "UDEN MIN LEDELSE FALDER MIT LAND FRA"
1350 PRINT "HINANDEN."
1360 GOTO 1450
1370 PRINT
1380 PRINT "RESULTATET AF DINE 2 ANGREB FØRTE TIL,"
1390 IF N + B + C > 3 / 2 * (D + E + F) THEN 1450
1400 IF N + B + C < 2 / 3 * (D + E + F) THEN 1470
1410 PRINT "AT PARIS-FORHANDLINGERNE BESLUTTER, AT"
1420 PRINT "VI BEHOLDER VORE RESPEKTIVE LANDE OG"
1430 PRINT "LEVER I FRED."
1440 GOTO 1510
1450 PRINT "AT DU VANDT.....SØRENS OSSE!"
1460 GOTO 1510
1470 PRINT "AT DU TABTE. JEG HAR BESEJRET DIT LAND!"
1480 PRINT "DET HAR DU GODT AF, DU KUNNE JO BARE"
1490 PRINT "LADE VÆRE MED AT SPILLE DET HER"
1500 PRINT "FJOLLEDE SPIL!!!!"
1510 END

```



Kanin

Dette program tegner en kanin på din lineskriver.

```

MORRISTOWN    CREATIVE COMPUTING    NEW JERSEY
-----
                                KANIN
-----

```

```

AN
KAN
KANINK
KANINKAN
ANINKANIN
NINKANINKA
INKANINKANI
NKANINKANIN
KANINKANINK
ANINKANINKA
NINKANINKAN
INKANINKANIN
NKANINKANINK
KANINKANINKA
ANINKANINKAN
NINKANINKAN
INKANINKAN
NKANINKA
ANINKANINKANI
INKANINKANINKANINK
ANINKANINKANINKANINKA
KANINKANINKANINKANINKAN
NINKANINKANIN
INKANINKAN
NKANINKA
ANINKANINKANI
INKANINKANINKANINK
ANINKANINKANINKANINKA
KANINKANINKANINKANINKAN
INKANINKANINKANINKANINKANI
NINKANINKANINKANINKANINKANIN
ANINKANI ANINKANINKANINKANIN
KANINKAN ANINKANINKANINKANINK
NKANINKAN ANINKANINKANINKANINK
INKANINKAN KANINKANINKANINKANINK
NINKANINKANINKANINKANINKANINKANINK
ANINKANINKANINKANINKANINKANINKANINK
NINKANINKANINKANINKANINKANINKANIN
INKANINKANINKANINKANINKANINKANIN
NKANINKANINKANINKANINKANINKANINKANI
ANINKANINKANINKANINKANINKANINKANI
KANINKANINKANINKANINKANINKAN N
NKAN NKANINK INKA K
KANIN INKANINK A
NKANI A NKANINK N
INKANI INKANIN INKANI
NINKANINKANINKANIN ANI
ANI N N N NKANINKA
KA NI N N N
NI ANIN
NIN
IN

```

```

100 PRINT " MORRISTOWN CREATIVE COMPUTING NEW JERSEY "
110 PRINT "-----"
120 PRINT " KANIN"
130 PRINT "-----"
140 PRINT
150 FOR I = 0 TO 4
160 READ B(I)
170 NEXT I
180 GOSUB 340
190 REM ASCII bogstavskode
200 L = 64
210 PRINT
220 READ X
230 IF X < 0 THEN 210
240 IF X > 128 THEN 320
250 PRINT TAB( X + 1);
260 READ Y
270 FOR I = X TO Y
280 J = I - 5 * INT ( I / 5 )
290 PRINT CHR$( L + B(J));
300 NEXT I
310 GOTO 220
320 GOSUB 340
330 GOTO 550
340 FOR I = 1 TO 6
350 PRINT
360 NEXT I
370 RETURN
380 REM DATA for 'kanin'
390 DATA 11,1,14,9,14
400 REM DATA for udskrift
410 DATA 1,2,-1,0,2,45,50,-1,0,5,43,52,-1,0,7,41,52,-1
420 DATA 1,9,37,50,-1,2,11,36,50,-1,3,13,34,49,-1,4,14,32,48,-1
430 DATA 5,15,31,47,-1,6,16,30,45,-1,7,17,29,44,-1,8,19,28,43,-1
440 DATA 9,20,27,41,-1,10,21,26,40,-1,11,22,25,38,-1,12,22,24,36,-1
450 DATA 13,34,-1,14,33,-1,15,31,-1,17,29,-1,18,27,-1
460 DATA 19,26,-1,16,28,-1,13,30,-1,11,31,-1,10,32,-1
470 DATA 8,33,-1,7,34,-1,6,13,16,34,-1,5,12,16,35,-1
480 DATA 4,12,16,35,-1,3,12,15,35,-1,2,35,-1,1,35,-1
490 DATA 2,34,-1,3,34,-1,4,33,-1,6,33,-1,10,32,34,34,-1
500 DATA 14,17,19,25,28,31,35,35,-1,15,19,23,30,36,36,-1
510 DATA 14,18,21,21,24,30,37,37,-1,13,18,23,29,33,38,-1
520 DATA 12,29,31,33,-1,11,13,17,17,19,19,22,22,24,31,-1
530 DATA 10,11,17,18,22,22,24,24,29,29,-1
540 DATA 22,23,26,29,-1,27,29,-1,28,29,-1,4096
550 END

```

Kast

Programmet tester dine grundlæggende kundskaber om mekanik. Det præsenterer dig for et simpelt problem: En bold kastet lodret op i luften med en tilfældigt valgt starthastighed. Du skal derefter besvare følgende tre spørgsmål om kastet:

1. Hvor højt når bolden?
2. Hvor mange sekunder går der, før den rammer jorden?
3. Hvilken hastighed har bolden på et tilfældigt valgt tidspunkt?

Datamaten bedømmer derefter dine svar og godkender en fejlmargen på 15%. Efter hver gennemkørsel får du et nyt spørgsmål, indtil du afbryder.

Kast er en forkortet udgave af Huntington Computer Project programmet KINERV, der blev skrevet af Richard Pav fra Patchogue High School, Patchogue, New York.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
KAST  
-----
```

```
EN BOLD BLIVER KASTET LIGE OP I LUFTEN  
MED EN STARHASTIGHED PÅ 30 METER PR.  
SEKUND.
```

```
HVOR HØJT VIL BOLDEN NÅ (I METER) ?44
```

```
DU ER TÆT PÅ.
```

```
DET RIGTIGE SVAR ER 45
```

```
HVOR MANGE SEKUNDER VIL DER GÅ,  
FØR DEN RAMMER JORDEN ?3
```

```
DU ER HELT VED SIDEN AF.....
```

```
DET RIGTIGE SVAR ER 6
```

```
HVAD ER BOLDENS HASTIGHED 2.6 SE-  
KUNDER EFTER START (M/SEC) ?3.5
```

```
DU ER TÆT PÅ.
```

```
DET RIGTIGE SVAR ER 4
```

```
2 RIGTIGE UD AF 3.
```

```
SLET IKKE DÆRLIGT!
```

```
EN BOLD BLIVER KASTET LIGE OP I LUFTEN  
MED EN STARHASTIGHED PÅ 15 METER PR.  
SEKUND.
```

```
HVOR HØJT VIL BOLDEN NÅ (I METER) ?9
```

```
DU ER HELT VED SIDEN AF.....
```

```
DET RIGTIGE SVAR ER 11.25
```

```
HVOR MANGE SEKUNDER VIL DER GÅ,  
FØR DEN RAMMER JORDEN ?3.5
```

```
DU ER HELT VED SIDEN AF.....
```

```
DET RIGTIGE SVAR ER 3
```

```
HVAD ER BOLDENS HASTIGHED 2.6 SE-  
KUNDER EFTER START (M/SEC) ?0
```

```
DU ER HELT VED SIDEN AF.....
```

```
DET RIGTIGE SVAR ER -11
```

```
0 RIGTIGE UD AF 3.
```

```
EN BOLD BLIVER KASTET LIGE OP I LUFTEN  
MED EN STARHASTIGHED PÅ 35 METER PR.  
SEKUND.
```

```
HVOR HØJT VIL BOLDEN NÅ (I METER) ?
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "KAST"  
130 PRINT "-----"  
140 PRINT  
150 Q = 0  
160 V = 5 + INT (35 * RND (1))  
170 PRINT "EN BOLD BLIVER KASTET LIGE OP I LUFTEN"  
180 PRINT "MED EN STARHASTIGHED PÅ";V;"METER PR."  
190 PRINT "SEKUND."  
200 PRINT  
210 A = .05 * V ^ 2  
220 PRINT "HVOR HØJT VIL BOLDEN NÅ (I METER) ";  
230 GOSUB 390  
240 A = V / 5  
250 PRINT "HVOR MANGE SEKUNDER VIL DER GÅ,"  
260 PRINT "FØR DEN RAMMER JORDEN ";  
270 GOSUB 390  
280 T = 1 + INT (2 * V * RND (1)) / 10  
290 A = V - 10 * T  
300 PRINT "HVAD ER BOLDENS HASTIGHED";T;"SE-"  
310 PRINT "KUNDER EFTER START (M/SEC) ";  
320 GOSUB 390  
330 PRINT  
340 PRINT Q;"RIGTIGE UD AF 3."  
350 IF Q < 2 THEN 140  
360 PRINT  
370 PRINT "SLET IKKE DÆRLIGT!"  
380 GOTO 140  
390 INPUT G  
400 PRINT  
410 IF ABS ((G - A) / A) < .15 THEN 440  
420 PRINT "DU ER HELT VED SIDEN AF....."  
430 GOTO 460  
440 PRINT "DU ER TÆT PÅ."  
450 Q = Q + 1  
460 PRINT  
470 PRINT "DET RIGTIGE SVAR ER";A  
480 PRINT  
490 RETURN  
500 END
```


Kemiker

Det fantasifulde kemiske stof kryptocyanid-syre kan kun fortyndes i forholdet 7 dele vand til 3 dele syre. Et andet blandingsforhold resulterer i, at syren bliver ustabil og eksploderer med det samme. Med udgangspunkt i en vis mængde syre skal du afgøre, hvor meget vand du skal fortynde den med. Hvis du rammer mere end 5% ved siden af, så mister du et af dine 9 liv. Programmet fortsætter, indtil alle dine ni liv er brugt, eller til du selv afbryder det.

Kemiker blev oprindeligt skrevet af Wayne Teeter fra Ridgecrest, Californien.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
KEMIKER  
-----
```

```
DET FANTASIFULDE KEMISKE STOF KRYPTO-  
CYANID-SYRE KAN KUN FORTYNDES I FORHOL-  
DET 7 DELE VAND TIL 3 DELE SYRE.
```

```
HVIS EN ANDEN FORTYNDINGSGRAD FORSØGES,  
BLIVER SYREN USTABIL OG EKSPODERER!
```

```
DU FÅR OPLYST HVOR MEGET SYRE DER ER,  
OG DU SKAL BESTEMME, HVOR MEGET VAND  
DER KAN TILSÆTTES.
```

```
HVIS DU RAMMER VED SIDEN AF, ER DU SELV  
UDE OM DET.....
```

```
DER ER 26 LITER KRYPTOCYANID-SYRE.  
HVOR MEGET VAND KAN TILSÆTTES ?45
```

```
ZZZZZZZYDE, BOBLE, DU ER NETOP BLEVET  
FORVANDLET TIL EN KLAT BÆVRENDE  
GELEAGTIG PROTOPLASMA!
```

```
MEN DU KAN FÅ LOV TIL AT PRØVE IGEN,  
MEN DENNEGANG MED ET ANDET LIV.....
```

```
DER ER 4 LITER KRYPTOCYANID-SYRE.  
HVOR MEGET VAND KAN TILSÆTTES ?9
```

```
GODT KLARET! DU KAN GODT TRÆKKE VEJRET  
IGEN, MEN PAS PÅ DE GIFTIGE DAMPE!
```

```
DER ER 45 LITER KRYPTOCYANID-SYRE.  
HVOR MEGET VAND KAN TILSÆTTES ?23
```

```
ZZZZZZZYDE, BOBLE, DU ER NETOP BLEVET  
FORVANDLET TIL EN KLAT BÆVRENDE  
GELEAGTIG PROTOPLASMA!
```

```
MEN DU KAN FÅ LOV TIL AT PRØVE IGEN,  
MEN DENNEGANG MED ET ANDET LIV.....
```

```
DER ER 18 LITER KRYPTOCYANID-SYRE.  
HVOR MEGET VAND KAN TILSÆTTES ?
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "          KEMIKER"  
130 PRINT "-----"  
140 PRINT  
150 PRINT "DET FANTASIFULDE KEMISKE STOF KRYPTO-"  
160 PRINT "CYANID-SYRE KAN KUN FORTYNDES I FORHOL-"  
170 PRINT "DET 7 DELE VAND TIL 3 DELE SYRE."  
180 PRINT  
190 PRINT "HVIS EN ANDEN FORTYNDINGSGRAD FORSØGES,"  
200 PRINT "BLIVER SYREN USTABIL OG EKSPODERER!"  
210 PRINT  
220 PRINT "DU FÅR OPLYST HVOR MEGET SYRE DER ER,"  
230 PRINT "OG DU SKAL BESTEMME, HVOR MEGET VAND"  
240 PRINT "DER KAN TILSÆTTES."  
250 PRINT  
260 PRINT "HVIS DU RAMMER VED SIDEN AF, ER DU SELV"  
270 PRINT "UDE OM DET....."  
280 PRINT  
290 A = INT ( RND (1) * 50)  
300 W = 7 * A / 3  
310 PRINT "DER ER";A;"LITER KRYPTOCYANID-SYRE."  
320 PRINT "HVOR MEGET VAND KAN TILSÆTTES ";  
330 INPUT R  
340 PRINT  
350 D = ABS ( W - R)  
360 IF D > W / 20 THEN 410  
370 PRINT "GODT KLARET! DU KAN GODT TRÆKKE VEJRET"  
380 PRINT "IGEN, MEN PAS PÅ DE GIFTIGE DAMPE!"  
390 PRINT  
400 GOTO 290  
410 PRINT "ZZZZZZZYDE, BOBLE, DU ER NETOP BLEVET"  
420 PRINT "FORVANDLET TIL EN KLAT BÆVRENDE"  
430 PRINT "GELEAGTIG PROTOPLASMA!"  
440 PRINT  
450 T = T + 1  
460 IF T = 9 THEN 510  
470 PRINT "MEN DU KAN FÅ LOV TIL AT PRØVE IGEN,"  
480 PRINT "MEN DENNEGANG MED ET ANDET LIV....."  
490 PRINT  
500 GOTO 290  
510 PRINT "DU HAR BRUGT ALLE DINE 9 LIV, MEN DU"  
520 PRINT "VIL ALTID BLIVE HUSKET SOM EN UFOR-"  
530 PRINT "GLEMMELIG BIDRAGYDER TIL DEN STORE"  
540 PRINT "VITTIGHEDSBOG OM KEMI!!!!!"  
550 END
```

Konge

Dette er et af de sværeste, mest udførlige og interessante spil, der handler om forvaltning af land og ressourcer. Hvis du aldrig har spillet et sådant spil før, bør du prøve Hammurabi først.

I dette spil er du statsminister på Kramnad, en lille kommunistisk ø, der måler 30 gange 70 kilometer. Det er din opgave at foretage beslutninger omkring landets budget og at fordele penge til dine undersåtter fra den fælles statskasse.

Pengesystemet er baseret på møntenheden Renork, og hver indbygger skal have 100 Renork pr. år for at overleve. Dit lands indtægter stammer fra jordbrug, og fra de turister der besøger landets vidunderlige skove for at jage og fiske, osv. En del af dit land er opdyrket, men der findes også store mineralforekomster, som du måske kan sælge til udenlandsk industri. Denne industri importerer og underholder deres egne arbejdere. Det koster mellem 10 og 15 Renork at så, opdyrke og høste en kvadratkilometer. Dit mål er at gennemføre en otte-årig embedsperiode uden større ulykker. Et advarende ord: det er ikke let!

Forfatteren til dette program er James A. Storer, som skrev det, mens han gik på Lexington High School.

```
MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY
-----
                          KONGE
-----
```

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

TIL LYKKE, DU ER NETOP BLEVET VALGT TIL STATSMINISTER PÅ ØEN 'KRAMNAD', EN LILLE Ø PÅ 30 GANGE 70 KILOMETER. DET ER DIN OPGAVER AT LÆGGE LANDETS BUDGET OG STYRE FORDELINGEN AF PENGE FRA STATSKASSEN TIL DINE LANDSMÆND.

MØNTENHEDEN HEDDER RENORK, OG HVER PERSON SKAL BRUGE 100 RENORK PR. ÅR FOR AT KUNNE OVERLEVE. DIT LANDS INDTÆGTER STAMMER FRA JORDBRUG OG FRA TURISTER, DER BESØGER LANDETS STORSLÆDE SKOVE FOR AT JAGE, FISKE OG NYDE NATUREN. HALV-DELEN AF DIT LAND ER LANDBRUGSJORD, DER OGSÅ INDEHOLDER STORE MÆNGDER MINERALER, OG SOM KAN SÆLGES TIL UDENLANDSKE INDUSTRIER, DER SELV IMPORTERER OG FORSØRGER DERES EGEN ARBEJDSKRAFT.

DET KOSTER MELLEM 10 OG 15 RENORK AT TILSÆ EN KVADRAT-KILOMETER LANDBRUGSJORD.

DIT MÅL ER AT FULDFØRE DIN 8 -ÅRIGE EMBEDSPERIODE UDEN PROBLEMER.....HELD OG LYKKE.

DU HAR NU 60465 RENORK I STATSKASSEN.
497 INDBYGGERE, 2000 KVADRATKILOMETER.
INDUSTRIERNE GIVER I ÅR 103 RENORK PR.
KVADRATKILOMETER JORD.
DET KOSTER I ØJEBLIKKET 10 RENORK PR.
KVADRATKILOMETER AT TILPLANTE JORDEN.

HVOR MANGE
KVADRATKILOMETER ØNSKER DU AT SÆLGE TIL INDUSTRIEN ?20
HVOR MANGE RENORK VIL DU GIVE INDBYGGERNE ?50000
HVOR MANGE KVADRATKILOMETER SKAL TILSÆS ?980
HVOR MANGE RENORK ØNSKES BRUGT TIL MILJØBESKYTTELSE ?2500

DER KOM 30 ARBEJDERE TIL LANDET, OG
99 INDBYGGERE KOM TIL ØEN.
AF DE 980 KVADRATKILOMETER DER BLEV TILSÆT,
HAR DU HØSTET 960 KVADRATKILOMETER AFGRØDER.
(GRUNDET LUFT- OG VAND-FORURENING FRA
DEN FREMMEDE INDUSTRI.)
FORTJENESTEN ER 49440 RENORK.
DU HAR TJENT 10921 RENORK PÅ TURISME.

DU HAR NU 60586 RENORK I STATSKASSEN.
596 INDBYGGERE, 30 UDENLANDSKE ARBEJDERE, OG
1980 KVADRATKILOMETER.
INDUSTRIERNE GIVER I ÅR 104 RENORK PR.
KVADRATKILOMETER JORD.
DET KOSTER I ØJEBLIKKET 10 RENORK PR.
KVADRATKILOMETER AT TILPLANTE JORDEN.

HVOR MANGE
KVADRATKILOMETER ØNSKER DU AT SÆLGE TIL INDUSTRIEN ?0
HVOR MANGE RENORK VIL DU GIVE INDBYGGERNE ?40000
HVOR MANGE KVADRATKILOMETER SKAL TILSÆS ?800
HVOR MANGE RENORK ØNSKES BRUGT TIL MILJØBESKYTTELSE ?30

196 INDBYGGERE ER DØDE AF SULT.
7 INDBYGGERE ER DØDE AF KULILTE- OG STØV-INDÅNDING.
DU ER NØDT TIL AT BRUGE 1827 RENORK PÅ
BEGRAVELSESOMKOSTNINGER.
2 INDBYGGERE FORLØD ØEN.
AF DE 800 KVADRATKILOMETER DER BLEV TILSÆT,
HAR DU HØSTET 781 KVADRATKILOMETER AFGRØDER.
(GRUNDET LUFT- OG VAND-FORURENING FRA
DEN FREMMEDE INDUSTRI.)
FORTJENESTEN ER 40612 RENORK.
DU HAR TJENT 8493 RENORK PÅ TURISME.
ET FALD, FORDI HOTELLERNE SER BESKIDTE OG
ANLØBNE UD GRUNDET SMOG.

203 INDBYGGERE DØDE PÅ ET ÅR!!!!!!

GRUNDET DENNE EKSTREME FEJLADMINISTRATION ER DU IKKE BLOT
KOMMET I UNRDE, OG BLEVET AFSAT, MEN DU HAR SANDELIG
OGSÅ VÆRE NØD TIL AT MODTAGE TITLEN SOM NATIONAL-IDIOT.

```
MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY
-----
                          KONGE
-----
```

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

TIL LYKKE, DU ER NETOP BLEVET VALGT TIL STATSMINISTER PÅ ØEN 'KRAMNAD', EN LILLE Ø PÅ 30 GANGE 70 KILOMETER. DET ER DIN OPGAVER AT LÆGGE LANDETS BUDGET OG STYRE FORDELINGEN AF PENGE FRA STATSKASSEN TIL DINE LANDSMÆND.

MØNTENHEDEN HEDDER RENORK, OG HVER PERSON SKAL BRUGE 100 RENORK PR. ÅR FOR AT KUNNE OVERLEVE. DIT LANDS INDTÆGTER STAMMER FRA JORDBRUG OG FRA TURISTER, DER BESØGER LANDETS STORSLÆDE SKOVE FOR AT JAGE, FISKE OG NYDE NATUREN. HALV-DELEN AF DIT LAND ER LANDBRUGSJORD, DER OGSÅ INDEHOLDER STORE MÆNGDER MINERALER, OG SOM KAN SÆLGES TIL UDENLANDSKE INDUSTRIER, DER SELV IMPORTERER OG FORSØRGER DERES EGEN ARBEJDSKRAFT.

DET KOSTER MELLEM 10 OG 15 RENORK AT TILSÆ EN KVADRAT-KILOMETER LANDBRUGSJORD.

DIT MÅL ER AT FULDFØRE DIN 8 -ÅRIGE EMBEDSPERIODE UDEN PROBLEMER.....HELD OG LYKKE.

DU HAR NU 60465 RENORK I STATSKASSEN.
497 INDBYGGERE, 2000 KVADRATKILOMETER.
INDUSTRIERNE GIVER I ÅR 103 RENORK PR.
KVADRATKILOMETER JORD.
DET KOSTER I ØJEBLIKKET 10 RENORK PR.
KVADRATKILOMETER AT TILPLANTE JORDEN.

HVOR MANGE
KVADRATKILOMETER ØNSKER DU AT SÆLGE TIL INDUSTRIEN ?10
HVOR MANGE RENORK VIL DU GIVE INDBYGGERNE ?50000
HVOR MANGE KVADRATKILOMETER SKAL TILSÆS ?980
HVOR MANGE RENORK ØNSKES BRUGT TIL MILJØBESKYTTELSE ?2000
GÅR IKKE. DU HAR KUN 1695 RENORK TILBAGE.
HVOR MANGE RENORK ØNSKES BRUGT TIL MILJØBESKYTTELSE ?1695

```

DER KOM 20 ARBEJDERE TIL LANDET, OG
67 INDBYGGERE KOM TIL ØEN.
AF DE 980 KVADRATKILOMETER DER BLEV TILSÆT,
HAR DU HØSTET 970 KVADRATKILOMETER AFGRØDER.
(GRUNDET LUFT- OG VAND-FORURENING FRA
DEN FREMMEDE INDUSTRI.)
FORTJENESTEN ER 49955 RENORK.
DU HAR TJENT 11071 RENORK PÅ TURISME.

-----

DU HAR NU 61026 RENORK I STATSKASSEN.
564 INDBYGGERE, 20 UDENLANDSKE ARBEJDERE, OG
1990 KVADRATKILOMETER.
INDUSTRIERNE GIVER I ÅR 104 RENORK PR.
KVADRATKILOMETER JORD.
DET KOSTER I ØJEBLIKKET 10 RENORK PR.
KVADRATKILOMETER AT TILPLANTE JORDEN.

-----

HVOR MANGE
KVADRATKILOMETER ØNSKER DU AT SÆLGE TIL INDUSTRIEN ?0
    HVOR MANGE RENORK VIL DU GIVE INDBYGGERNE ?45000
    HVOR MANGE KVADRATKILOMETER SKAL TILSÆS ?900
    HVOR MANGE RENORK ØNSKES BRUGT TIL MILJØBESKYTTELSE ?2000

114 INDBYGGERE ER DØDE AF SULT.
DU ER NØDT TIL AT BRUGE 1026 RENORK PÅ
BEGRAVELSESOMKOSTNINGER.
79 INDBYGGERE KOM TIL ØEN.
AF DE 900 KVADRATKILOMETER DER BLEV TILSÆT,
HAR DU HØSTET 891 KVADRATKILOMETER AFGRØDER.
(GRUNDET LUFT- OG VAND-FORURENING FRA
DEN FREMMEDE INDUSTRI.)
FORTJENESTEN ER 46332 RENORK.
DU HAR TJENT 9897 RENORK PÅ TURISME.
ET FALD, FORDI HOTELLERNE SER BESKIDTE OG
ANLØBNE UD GRUNDET SMOG.

DER VAR PENGE TILBAGE I STATSKASSEN, SOM DU VAR FOR FEDTET
TIL AT BRUGE. PÅ GRUND AF DETTE DØDE NOGLE AF DINE UNDER-
SÆTTER AF SULT. BEFOLKNINGEN ER FORARGET OG DU ER BLEVET
NØDT TIL ENTEN AT TRÆKKE DIG TILBAGE, ELLER AT BEGÅ SELV-
MORD. DU KAN SELV VÆLGE.
HVIS DU VÆLGER DET SIDSTE, SÅ HUSK AT SLUKKE FOR DATAMATEN
INDEN DU GØR DET.

100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY      "
110 PRINT "-----"
120 PRINT "                      KONGE"
130 PRINT "-----"
140 PRINT
150 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
160 INPUT Z$
170 N5 = 8
180 IF LEFT$(Z$,1) = "N" THEN 400
190 IF Z$ = "IGEN" THEN 2720
200 PRINT
210 PRINT "TIL LYKKE, DU ER NETOP BLEVET VALGT TIL STATSMINISTER PÅ ØEN"
220 PRINT "'KRAMNAD', EN LILLE Ø PÅ 30 GANGE 70 KILOMETER. DET"
230 PRINT "ER DIN OPGAVE AT LÆGGE LANDETS BUDGET OG STYRE FORDELINGEN"
240 PRINT "AF PENGE FRA STATSKASSEN TIL DINE LANDSMÆND."
250 PRINT
260 PRINT "MØNTENHEDEN HEDDER RENORK, OG HVER PERSON SKAL BRUGE 100"
270 PRINT "RENORK PR. ÅR FOR AT KUNNE OVERLEVE. DIT LANDS INDTÆGTER"
280 PRINT "STAMMER FRA JORDBRUG OG FRA TURISTER, DER BESØGER LANDETS"
290 PRINT "STORSLÆDE SKOVE FOR AT JAGE, FISKE OG NYDE NATUREN. HALV-"
300 PRINT "DELEN AF DIT LAND ER LANDBRUGSJORD, DER OGSÅ INDEHOLDER"
310 PRINT "STORE MÆNGDER MINERALER, OG SOM KAN SÆLGES TIL UDENLANDSKE"
320 PRINT "INDUSTRIER, DER SELV IMPORTERER OG FORSØRGER DERES EGEN"
330 PRINT "ARBEJDSKRAFT."
340 PRINT
350 PRINT "DET KOSTER MELLEM 10 OG 15 RENORK AT TILSÆ EN KVADRAT-"
360 PRINT "KILOMETER LANDBRUGSJORD."
370 PRINT
380 PRINT "DIT MÅL ER AT FULDFØRE DIN";N5;"-ÅRIGE EMBEDSPERIODE"
390 PRINT "UDEN PROBLEMER.....HELD OG LYKKE."
400 PRINT
410 Z = INT (60000 + (1000 * RND (1)) - (1000 * RND (1)))
420 B = INT (500 + (10 * RND (1)) - (10 * RND (1)))
430 D = 2000
440 W = INT (10 * RND (1) + 95)
450 PRINT "-----"
460 PRINT

470 PRINT "DU HAR NU";Z;"RENORK I STATSKASSEN."
480 PRINT INT (B);"INDBYGGERE, ";
490 V9 = INT ((( RND (1) / 2) * 10 + 10))
500 IF C = 0 THEN 520
510 PRINT INT (C);"UDENLANDSKE ARBEJDERE, OG"
520 PRINT INT (D);"KVADRATKILOMETER."
530 PRINT "INDUSTRIERNE GIVER I ÅR";W;"RENORK PR."
540 PRINT "KVADRATKILOMETER JORD."
550 PRINT "DET KOSTER I ØJEBLIKKET";V9;"RENORK PR."
560 PRINT "KVADRATKILOMETER AT TILPLANTE JORDEN."
570 PRINT
580 PRINT "-----"
590 PRINT
600 PRINT "HVOR MANGE"
610 PRINT "KVADRATKILOMETER ØNSKER DU AT SÆLGE TIL INDUSTRIEN ";
620 INPUT H
630 IF H < 0 THEN 600
640 IF H < = D - 1000 THEN 730
650 PRINT "  TÆNK DIG OM....."
660 PRINT "DU HAR KUN";D - 1000;"KVADRATKILOMETER LANDBRUGSJORD."
670 IF X < > 0 THEN 600
680 PRINT "(DEN UDENLANDSKE INDUSTRI VIL KUN KØBE LANDBRUGSJORD,"
690 PRINT "FORDI SKOV ER ØKONOMISK FOR MINEDRIFT,PÅ GRUND AF"
700 PRINT "TRÆERNE, TYKKERE OVERFLADE LAG, ETC.)"
710 X = 1
720 GOTO 600
730 D = INT (D - H)
740 Z = INT (Z + (H * W))
750 PRINT "      HVOR MANGE RENORK VIL DU GIVE INDBYGGERNE ";
760 INPUT I
770 IF I < 0 THEN 750
780 IF I < Z THEN 870
790 IF I = Z THEN 830
800 PRINT "  TAG DIG NU SAMMEN, DU HAR KUN";Z;"RENORK"
810 PRINT "  I STATSKASSEN."
820 GOTO 750
830 J = 0
840 K = 0
850 Z = 0
860 GOTO 1220
870 Z = INT (Z - I)
880 PRINT "      HVOR MANGE KVADRATKILOMETER SKAL TILSÆS ";
890 INPUT J
900 IF J < 0 THEN 880
910 IF J < = B * 2 THEN 940
920 PRINT "DESVÆRRE, HVER INDBYGGER KAN KUN TILSÆ 2 KVADRATKILOMETER."
930 GOTO 880
940 IF J < = D - 1000 THEN 980
950 PRINT "DESVÆRRE, MEN DU HAR KUN";D - 1000;"KVADRATKILOMETER"
960 PRINT "LANDBRUGSJORD, DER KAN TILSÆS."
970 GOTO 880
980 U1 = INT (J * V9)
990 IF U1 < Z THEN 1060
1000 IF U1 = Z THEN 1030
1010 PRINT "  OM IGEN, DU HAR KUN";Z;"RENORK I STATSKASSEN."
1020 GOTO 880
1030 K = 0
1040 Z = 0
1050 GOTO 1220
1060 Z = Z - U1
1070 PRINT "HVOR MANGE RENORK ØNSKES BRUGT TIL MILJØBESKYTTELSE ";
1080 INPUT K
1090 IF K < 0 THEN 1070
1100 IF K < = Z THEN 1220
1110 PRINT "GÅR IKKE. DU HAR KUN";Z;"RENORK TILBAGE."
1120 GOTO 1070
1130 IF H < > 0 THEN 1230
1140 IF I < > 0 THEN 1230
1150 IF J < > 0 THEN 1230
1160 IF K < > 0 THEN 1230
1170 PRINT
1180 PRINT "FARVEL."
1190 PRINT "(HVIS DU VIL FORTSÆTTE SPILLET SENERE, SKAL DU SVARE"
1200 PRINT "'IGEN' PÅ SPØRGSMALET OM DU VIL HAVE INSTRUKTIONER)"
1210 GOTO 2990
1220 GOTO 1130
1230 PRINT
1240 Z = INT (Z - K)
1250 Z4 = Z
1260 IF INT (I / 100 - B) > = 0 THEN 1290
1270 IF I / 100 < 50 THEN 2470
1280 PRINT INT (B - (I / 100));"INDBYGGERE ER DØDE AF SULT."
1290 F1 = INT ( RND (1) * (2000 - D))
1300 IF K < 25 THEN 1320
1310 F1 = INT (F1 / (K / 25))
1320 IF F1 < = 0 THEN 1340
1330 PRINT F1;"INDBYGGERE ER DØDE AF KULILTE- OG STØV-INDÅNDING."
1340 IF INT ((I / 100) - B) < 0 THEN 1370

```

```

1350 IF F1 > 0 THEN 1430
1360 GOTO 1530
1370 PRINT "DU ER NØDT TIL AT BRUGE"; INT ((F1 + (B - (I / 100))) * 9);
1380 PRINT "RENORK PÅ"
1390 PRINT "BEGRAVELSESOMKOSTNINGER."
1400 B5 = INT (F1 + (B - (I / 100)))
1410 Z = INT (Z - ((F1 + (B - (I / 100))) * 9))
1420 GOTO 1470
1430 PRINT "DU ER NØDT TIL AT BRUGE"; INT (F1 * 9);"RENORK PÅ"
1440 PRINT "BEGRAVELSESOMKOSTNINGER."
1450 B5 = F1
1460 Z = INT (Z - (F1 * 9))
1470 IF Z > = 0 THEN 1520
1480 PRINT "DU HAR IKKE ØKONOMISK RESERVE TIL AT KLARE OMKOSTNINGERNE."
1490 PRINT "DU ER NØDT TIL AT SÆLGE JORD."
1500 D = INT (D + (Z / W))
1510 Z = 0
1520 B = INT (B - B5)
1530 IF H = 0 THEN 1580
1540 C1 = INT (H + (RND (1) * 10) - (RND (1) * 20))
1550 IF C > 0 THEN 1570
1560 C1 = C1 + 20
1570 PRINT "DER KOM";C1;"ARBEJDERE TIL LANDET, OG"
1580 P1 = INT (((I / 100 - B) / 10) + (K / 25) - ((2000 - D) / 50) - (F1 / 2))
1590 PRINT ABS (P1);"INDBYGGERE ";
1600 IF P1 < 0 THEN 1630
1610 PRINT "KOM TIL";
1620 GOTO 1640
1630 PRINT "FORLOD";
1640 PRINT " ØEN."
1650 B = INT (B + P1)
1660 C = INT (C + C1)
1670 U2 = INT (((2000 - D) * ((RND (1) + 1.5) / 2)))
1680 IF C = 0 THEN 1700
1690 PRINT "AF DE "; INT (J);"KVADRATKILOMETER DER BLEV TILSÆT,"
1700 IF J > U2 THEN 1720
1710 U2 = J
1720 PRINT "HAR DU HØSTET"; INT (J - U2);"KVADRATKILOMETER AFGRØDER."
1730 IF U2 = 0 THEN 1800
1740 IF T1 > = 2 THEN 1800
1750 PRINT "(GRUNDET ";
1760 IF T1 = 0 THEN 1780
1770 PRINT "FORØGET ";
1780 PRINT "LUFT- OG VAND-FORURENING FRA"
1790 PRINT "DEN FREMMEDE INDUSTRI.)"
1800 Q = INT ((J - U2) * (W / 2))
1810 PRINT " FORTJENESTEN ER"; INT (Q);"RENORK."
1820 Z = INT (Z + Q)
1830 V1 = INT (((B - P1) * 22) + (RND (1) * 500))
1840 V2 = INT ((2000 - D) * 15)
1850 PRINT "DU HAR TJENT"; ABS (INT (V1 - V2));"RENORK PÅ TURISME."
1860 IF V2 = 0 THEN 1950
1870 IF V1 - V2 > = V3 THEN 1950
1880 PRINT "ET FALD, FORDI ";
1890 G1 = 10 * RND (1)
1900 IF G1 < = 2 THEN 1980
1910 IF G1 < = 4 THEN 2000
1920 IF G1 < = 6 THEN 2020
1930 IF G1 < = 8 THEN 2040
1940 IF G1 < = 10 THEN 2070
1950 V3 = INT (V1 - V2)
1960 Z = INT (Z + V3)
1970 GOTO 2100
1980 PRINT "FISKEBESTANDEN ER FALDET GRUNDET VAND-FORURENING."
1990 GOTO 1950
2000 PRINT "LUFT-FORURENINGEN DRÆBER FUGLEBESTANDEN."
2010 GOTO 1950
2020 PRINT "KURSTEDERNES KILDER ER ØDELÅGT AF VAND-FORURENING."
2030 GOTO 1950
2040 PRINT "UBEHAGELIG SMOG AFSKRÆKKER TURISTERNE"
2050 PRINT "FRA AT TAGE SOLBAD."
2060 GOTO 1950
2070 PRINT "HOTELLERNE SER BESKIDTE OG"
2080 PRINT "ANLØBNE UD GRUNDET SMOG."
2090 GOTO 1950
2100 IF B5 > 200 THEN 2300
2110 IF B < 343 THEN 2470
2120 IF (Z4 / 100) > 5 THEN 2530
2130 IF C > B THEN 2170
2140 IF N5 - 1 = X5 THEN 2630
2150 PRINT
2160 GOTO 2960
2170 PRINT
2180 PRINT
2190 PRINT "ANTALLET AF UDENLANDSKE ARBEJDERE HAR OVERSKREDT ANTAL-"
2200 PRINT "LET AF INDBYGGERE. DA DE NU ER I FLERTAL HAR DE GJORT"
2210 PRINT "OPRØR, OG HAR OVERTAGET STYRET I LANDET."

2220 IF RND (1) < = .5 THEN 2260
2230 PRINT "DU ER BLEVET FYRET SOM MINISTER,"
2240 PRINT "OG BEFINDER DIG NU I FÆNGSEL....."
2250 GOTO 2270
2260 PRINT "DU ER BLEVET SNIGHYRDET....."
2270 PRINT
2280 PRINT
2290 GOTO 2990
2300 PRINT
2310 PRINT
2320 PRINT B5;"INDBYGGERE DØDE PÅ ET ÅR!!!!!!"
2330 PRINT
2340 PRINT "GRUNDET DENNE EKSTREME FEJLADMINISTRATION ER DU IKKE BLOT"
2350 PRINT "KOMMET I UNÅDE, OG BLEVET AFSAT, MEN DU HAR SANDELIG"
2360 PRINT "OGSÅ ";
2370 M6 = INT (RND (1) * 10)
2380 IF M6 < = 3 THEN 2410
2390 IF M6 < = 6 THEN 2430
2400 IF M6 < = 10 THEN 2450
2410 PRINT "FRÆT DET VENSTRE ØJE STUKKET UD."
2420 GOTO 2270
2430 PRINT "FRÆT ET UHYRE DÅRLIGT RY."
2440 GOTO 2270
2450 PRINT "VÆRE NØD TIL AT MODTAGE TITLEN SOM NATIONAL-IDIOT."
2460 GOTO 2270
2470 PRINT
2480 PRINT
2490 PRINT "OVER EN TREDJEDEL AF BEFOLKNINGEN ER OMKOMMET SIDEN DU"
2500 PRINT "BLEV VALGT TIL MINISTER. BEFOLKNINGEN (DE FÅ DER ENDNU"
2510 PRINT "ER TILBAGE) HADER DIG AF ET GODT HJERTE."
2520 GOTO 2220
2530 IF B5 - F1 < 2 THEN 2130
2540 PRINT
2550 PRINT "DER VAR PENGE TILBAGE I STATSKASSEN, SOM DU VAR FOR FEDTET"
2560 PRINT "TIL AT BRUGE. PÅ GRUND AF DETTE DØDE NOGLE AF DINE UNDER-"
2570 PRINT "SÆTTER AF SULT. BEFOLKNINGEN ER FORARGET OG DU ER BLEVET"
2580 PRINT "NØDT TIL ENTEN AT TRÆKKE DIG TILBAGE, ELLER AT BEGR SELV-"
2590 PRINT "MORD. DU KAN SELV VÆLGE."
2600 PRINT "HVIS DU VÆLGER DET SIDSTE, SÅ HUSK AT SLUKKE FOR DATAMATEN"
2610 PRINT "INDEN DU GØR DET."
2620 GOTO 2270
2630 PRINT
2640 PRINT "TILLYKKE!!!!!!!!!!!!!!!!!!!!!!"
2650 PRINT "DU HAR GENMEMFØRT DIN";N5;"-RRS PERIODE MED SUCCES."
2660 PRINT "DU HAR SELVFLØGELIG VÆRET ENORMT HELDIG, MEN DET ER IKKE"
2670 PRINT "DESTO MINDRE EN STOR BEDRIFT DU HAR UDFØRT. FARVEL, OG"
2680 PRINT "HELD OG LYKKE, FOR DET VIL DU UDEN TVIVL FÅ BRUG FOR."
2690 PRINT "DU TILHØRER JO ÅBENBART DEN TYPE DUMME MENNESKER, DER"
2700 PRINT "KAN FINDE PÅ AT SPILLE ET ÅRDSVAGT SPIL SOM DETTE."
2710 GOTO 2270
2720 PRINT "HVOR MANGE ÅR HAVDE DU VÆRET MINISTER DA SPILLET BLEV"
2730 PRINT "AFBRUDT ";
2740 INPUT X5
2750 IF X5 < 0 THEN 2270
2760 IF X5 < 8 THEN 2790
2770 PRINT " HOLD NU OP! DU KAN HØJEST HAVE";N5;"ÅR."
2780 GOTO 2720
2790 PRINT "HVOR MEGET HAVDE DU I STATSKASSEN ";
2800 INPUT Z
2810 IF Z < 0 THEN 2270
2820 PRINT "HVOR STOR VAR BEFOLKNINGEN ";
2830 INPUT B
2840 IF B < 0 THEN 2270
2850 PRINT "HVOR MANGE ARBEJDERE VAR DER ";
2860 INPUT C
2870 IF C < 0 THEN 2270
2880 PRINT "HVOR MANGE KVADRATMETER LAND ";
2890 INPUT D
2900 IF D < 0 THEN 2270
2910 IF D > 2000 THEN 2930
2920 IF D > 1000 THEN 440
2930 PRINT "NEJ, NU MÅ DU HOLDE OP, DU BEGYNDTE JO MED 1000 KVADRAT-"
2940 PRINT "KILOMETER LANDBRUGSJORD OG LIGE SÅ MEGET SKOV."
2950 GOTO 2880
2960 X5 = X5 + 1
2970 B5 = 0
2980 GOTO 440
2990 END

```


Krig

Dette program spiller kortspillet Krig. Kortene blandes, og hver spiller får et kort. Spillerne skal sammenligne kortene, og den, som har kortet med den højeste (numeriske) værdi, vinder. Hvis kortene har samme værdi, bliver det uafgjort. Spillet er slut, når hele kortspillet er brugt op (dvs. 52 kort, 26 omgange), eller hvis man afbryder spillet.

Datamaten giver kort og angiver farve og værdi. F.eks. angives Spar 7 som S-7.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
                    KRIG  
-----
```

```
DETT ER KORTSPILLET 'KRIG'.
```

```
HVERT KORT ANGIVES MED ET FARVENUMMER,  
SOM F.EKS. S-7 FOR SPAR SYV.
```

```
ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J
```

```
DATAMATEN GIVER SIG SELV OG DIG ET  
'KORT'. DEN HØJESTE KORT-VÆRDI  
VINDER. ES ER HØJESTE KORT.
```

```
SPILLET SLUTTER NÅR DU VÆLGER IKKE AT  
FORTSÆTTE, ELLER NÅR DER IKKE ER  
FLERE KORT.
```

```
DU: R-10 DATAMATEN: R-4
```

```
**** DU VINDER ****  
DU HAR 1 POINT  
DATAMATEN HAR 0 POINT.
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
```

```
DU: S-2 DATAMATEN: S-8
```

```
*** DATAMATEN VINDER ***  
DU HAR 1 POINT  
DATAMATEN HAR 1 POINT.
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
```

```
DU: H-9 DATAMATEN: S-A
```

```
*** DATAMATEN VINDER ***  
DU HAR 1 POINT  
DATAMATEN HAR 2 POINT.
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
```

```
DU: S-K DATAMATEN: H-3
```

```
**** DU VINDER ****  
DU HAR 2 POINT  
DATAMATEN HAR 2 POINT.
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
```

```
DU: R-5 DATAMATEN: K-2
```

```
**** DU VINDER ****  
DU HAR 3 POINT  
DATAMATEN HAR 2 POINT.
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
```

```
DU: H-7 DATAMATEN: R-7
```

```
**** UAFGJORT ****
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
```

```
DU: H-8 DATAMATEN: K-Q
```

```
*** DATAMATEN VINDER ***  
DU HAR 3 POINT  
DATAMATEN HAR 3 POINT.
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
```

```
DU: S-6 DATAMATEN: S-4
```

```
**** DU VINDER ****  
DU HAR 4 POINT  
DATAMATEN HAR 3 POINT.
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?J
```

```
DU: H-5 DATAMATEN: K-5
```

```
**** UAFGJORT ****
```

```
VIL DU SPILLE IGEN ('J' ELLER 'N') ?N
```

```
TAK FOR SPILLET!  
DET VAR HYGGELIGT.
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "                    KRIG"  
130 PRINT "-----"  
140 PRINT  
150 PRINT "    DETTE ER KORTSPILLET 'KRIG'."  
160 PRINT  
170 PRINT "HVERT KORT ANGIVES MED ET FARVENUMMER,"  
180 PRINT "SOM F.EKS. S-7 FOR SPAR SYV."  
190 PRINT  
200 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";  
210 INPUT B$  
220 IF LEFT$(B$,1) = "N" THEN 340  
230 IF LEFT$(B$,1) = "J" THEN 280  
240 PRINT  
250 PRINT "SVAR VENLIGST 'JA' ELLER 'NEJ'"  
260 PRINT  
270 GOTO 200  
280 PRINT  
290 PRINT "DATAMATEN GIVER SIG SELV OG DIG ET"  
300 PRINT "'KORT'. DEN HØJESTE KORT-VÆRDI"  
310 PRINT "VINDER. ES ER HØJESTE KORT."  
320 PRINT  
330 PRINT "SPILLET SLUTTER NÅR DU VÆLGER IKKE AT"  
340 PRINT "FORTSÆTTE, ELLER NÅR DER IKKE ER"  
350 PRINT "FLERE KORT."  
360 PRINT
```

```
370 DIM A$(52),L(54)
```

```
380 FOR I = 1 TO 52
```

```
390 READ A$(I)
```

```
400 NEXT I
```

```
410 FOR J = 1 TO 52
```

```
460 GOTO 420
```

```
420 L(J) = INT(52 * RND(1)) + 1
```

```
470 NEXT K
```

```
430 IF J = 1 THEN 480
```

```
480 NEXT J
```

```
440 FOR K = 1 TO J - 1
```

```
490 P = P + 1
```

```
450 IF L(K) < > L(J) THEN 470
```

```
500 M1 = L(P)
```

```
510 P = P + 1
```

```
520 M2 = L(P)
```

```
530 PRINT
```

```
540 PRINT "DU: ";A$(M1);" DATAMATEN: ";A$(M2)
```

```
550 N1 = INT((M1 - .5) / 4)
```

```
560 N2 = INT((M2 - .5) / 4)
```

```
570 IF N1 > = N2 THEN 630
```

```
580 A1 = A1 + 1
```

```
590 PRINT "*** DATAMATEN VINDER ***"
```

```
600 PRINT "    DU HAR";B1;"POINT"
```

```
610 PRINT "DATAMATEN HAR";A1;"POINT."
```

```
620 GOTO 700
```

```
630 IF N1 = N2 THEN 690
```

```
640 B1 = B1 + 1
```

```
650 PRINT "**** DU VINDER ****"
```

```
660 PRINT "    DU HAR";B1;"POINT"
```

```
670 PRINT "DATAMATEN HAR";A1;"POINT."
```

```
680 GOTO 700
```

```
690 PRINT "**** UAFGJORT ****"
```

```
700 IF L(P + 1) = 0 THEN 780
```

```
710 PRINT
```

```
720 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
```

```
730 INPUT V$
```

```
740 IF LEFT$(V$,1) = "J" THEN 490
```

```
750 IF LEFT$(V$,1) = "N" THEN 840
```

```
760 PRINT "TAST VENLIGST 'JA' ELLER 'NEJ'."
```

```
770 GOTO 710
```

```
780 PRINT
```

```
790 PRINT
```

```
800 PRINT "DER ER IKKE FLERE KORT TILBAGE!"
```

```
810 PRINT "SLUTRESULTAT:"
```

```
820 PRINT "    DU HAR";B1;"POINT"
```

```
830 PRINT "DATAMATEN HAR";A1;"POINT."
```

```
840 PRINT
```

```
850 PRINT "TAK FOR SPILLET!"
```

```
860 PRINT "DET VAR HYGGELIGT."
```

```
870 PRINT
```

```
880 DATA "S-2","H-2","K-2","R-2"
```

```
890 DATA "S-3","H-3","K-3","R-3"
```

```
900 DATA "S-4","H-4","K-4","R-4"
```

```
910 DATA "S-5","H-5","K-5","R-5"
```

```
920 DATA "S-6","H-6","K-6","R-6"
```

```
930 DATA "S-7","H-7","K-7","R-7"
```

```
940 DATA "S-8","H-8","K-8","R-8"
```

```
950 DATA "S-9","H-9","K-9","R-9"
```

```
960 DATA "S-10","H-10","K-10","R-10"
```

```
970 DATA "S-J","H-J","K-J","R-J"
```

```
980 DATA "S-Q","H-Q","K-Q","R-Q"
```

```
990 DATA "S-K","H-K","K-K","R-K"
```

```
1000 DATA "S-A","H-A","K-A","R-A"
```

```
1010 END
```

Kryds og bolle

Spillet Kryds og bolle behøver vist ingen nærmere introduktion. I denne variant spiller du mod datamaten. Træk indtastes ifølge nedenstående diagram:

```
1 2 3
4 5 6
7 8 9
```

Hvis du gør et fejlagtigt træk, vinder datamaten. Hvis datamaten gør et dårligt træk, så kan du vinde, i modsat fald ender spillet uafgjort.

Den anden version af spillet udskriver brættet efter hvert træk. Denne version er mere velegnet for dataskærm, især hvis du modificerer programmet, så brættet ikke udskrives på ny efter hvert træk, men du i stedet kan bruge markøren til at angive trækket.

Det første af programmerne er skrevet af Tom Kloos, som på det tidspunkt var studerende ved Oregon Museum of Science and Industry. Dette program blev senere modificeret af Steve North fra Creative Computing. Forfatteren til det andet spil er Curt Flick fra Akron, Ohio.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

KRYDS-OG-BOLLE

SPILLEBRÆTTET NUMMERERES SÅLEDES:

```
1 2 3
8 9 4
7 6 5
```

```
DATAMATENS TRÆK :9
  DIT TRÆK ?3
DATAMATENS TRÆK :4
  DIT TRÆK ?8
DATAMATENS TRÆK :6
  DIT TRÆK ?7
DATAMATENS TRÆK :2
OG VI HAR EN VINDER *****
```

```
DATAMATENS TRÆK :9
  DIT TRÆK ?5
DATAMATENS TRÆK :6
  DIT TRÆK ?2
DATAMATENS TRÆK :8
  DIT TRÆK ?4
DATAMATENS TRÆK :3
  DIT TRÆK ?7
DATAMATENS TRÆK :1
SPILLET ER UAFGJORT.
```

```
DATAMATENS TRÆK :9
  DIT TRÆK ?
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          KRYDS-OG-BOLLE"
130 PRINT "-----"
140 PRINT
150 PRINT
160 REM Dette program spiller kryds-og-bolle
170 REM Datamaten begynder
180 PRINT "SPILLEBRÆTTET NUMMERERES SÅLEDES:"
190 PRINT
200 PRINT "1 2 3"
210 PRINT "8 9 4"
220 PRINT "7 6 5"
230 PRINT
240 DEF FN M(X) = X - 8 * INT ((X - 1) / 8)
```

```
250 PRINT
260 PRINT
270 A = 9
280 M = A
290 GOSUB 670
300 P = M
310 B = FN M(P + 1)
320 M = B
330 GOSUB 670
340 Q = M
350 IF Q = FN M(B + 4) THEN 400
360 C = FN M(B + 4)
370 M = C
380 GOSUB 710
390 GOTO 730
400 C = FN M(B + 2)
410 M = C
420 GOSUB 670
430 R = M
440 IF R = FN M(C + 4) THEN 490
450 D = FN M(C + 4)
460 M = D
470 GOSUB 710
480 GOTO 730
490 IF P / 2 < > INT (P / 2) THEN 540
500 D = FN M(C + 7)
510 M = D
520 GOSUB 710
530 GOTO 730
540 D = FN M(C + 3)
550 M = D
560 GOSUB 670
570 S = M
580 IF S = FN M(D + 4) THEN 620
590 E = FN M(D + 4)
600 M = E
610 GOSUB 710
620 E = FN M(D + 6)
630 M = E
640 GOSUB 710
650 PRINT "SPILLET ER UAFGJORT."
660 GOTO 250
670 GOSUB 710
680 PRINT "          DIT TRÆK ";
690 INPUT M
700 RETURN
710 PRINT "DATAMATENS TRÆK :";M
720 RETURN
730 PRINT "OG VI HAR EN VINDER *****"
740 GOTO 250
750 END
```

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

KRYDS-OG-BOLLE, VERSION 2

SPILLER BRÆTTET NUMMERERES SÅLEDES:

```
1 2 3
4 5 6
7 8 9
```

ØNSKER DU 'O' ELLER 'X' ?X

HVAD VÆLGER DU ?4

```
! !
---+---+---
X ! !
---+---+---
! !
```

COMPUTEREN VÆLGER.....

```
! !
---+---+---
X ! O !
---+---+---
! !
```

```

HVAD VÆLGER DU ?1
X ! !
-----
X ! 0 !
-----
! !

COMPUTEREN VÆLGER.....
X ! !
-----
X ! 0 !
-----
0 ! !

HVAD VÆLGER DU ?3
X ! ! X
-----
X ! 0 !
-----
0 ! !

COMPUTEREN VÆLGER.....
X ! 0 ! X
-----
X ! 0 !
-----
0 ! !

HVAD VÆLGER DU ?8
X ! 0 ! X
-----
X ! 0 !
-----
0 ! X !

COMPUTEREN VÆLGER.....
X ! 0 ! X
-----
X ! 0 !
-----
0 ! X ! 0

HVAD VÆLGER DU ?6
X ! 0 ! X
-----
X ! 0 ! X
-----
0 ! X ! 0

SPILLET ER UAFGJORT. TAK!

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          KRYDS-OG-BOLLE, VERSION 2"
130 PRINT "-----"
140 PRINT
150 PRINT
160 PRINT "SPILLER BRÆTTET NUMMERERES SÅLEDES:"
170 PRINT
180 PRINT " 1 2 3"
190 PRINT " 4 5 6"
200 PRINT " 7 8 9"
210 PRINT
220 PRINT
230 DIM S(9)
240 PRINT "ØNSKER DU '0' ELLER 'X' ";
250 INPUT C$
260 PRINT
270 IF LEN (C$) < > 1 THEN 240
280 IF LEFT$ (C$,1) = "X" THEN 1030
290 P$ = "0"
300 Q$ = "X"
310 G = - 1:H = 1
320 IF S(5) < > 0 THEN 340
330 S(5) = - 1: GOTO 930
340 IF S(5) < > 1 THEN 370
350 IF S(1) < > 0 THEN 410
360 S(1) = - 1: GOTO 930
370 IF S(2) = 1 AND S(1) = 0 THEN 880
380 IF S(4) = 1 AND S(1) = 0 THEN 880
390 IF S(6) = 1 AND S(9) = 0 THEN 920
400 IF S(8) = 1 AND S(9) = 0 THEN 920
410 IF G = 1 THEN 430
420 GOTO 480
430 J = 3 * INT ((M - 1) / 3) + 1
440 IF 3 * INT ((M - 1) / 3) + 1 = M THEN K = 1
450 IF 3 * INT ((M - 1) / 3) + 2 = M THEN K = 2
460 IF 3 * INT ((M - 1) / 3) + 3 = M THEN K = 3
470 GOTO 500
480 FOR J = 1 TO 7 STEP 3
490 FOR K = 1 TO 3
500 IF S(J) < > G THEN 550
510 IF S(J + 2) < > G THEN 600
520 IF S(J + 1) < > 0 THEN 640
530 S(J + 1) = - 1
540 GOTO 930
550 IF S(J) = H THEN 640
560 IF S(J + 2) < > G THEN 640
570 IF S(J + 1) < > G THEN 640
580 S(J) = - 1
590 GOTO 930
600 IF S(J + 2) < > 0 THEN 640
610 IF S(J + 1) < > G THEN 640
620 S(J + 2) = - 1
630 GOTO 930
640 IF S(K) < > G THEN 690
650 IF S(K + 6) < > G THEN 740
660 IF S(K + 3) < > 0 THEN 780
670 S(K + 3) = - 1
680 GOTO 930
690 IF S(K) = H THEN 780
700 IF S(K + 6) < > G THEN 780
710 IF S(K + 3) < > G THEN 780
720 S(K) = - 1
730 GOTO 930
740 IF S(K + 6) < > 0 THEN 780
750 IF S(K + 3) < > G THEN 780
760 S(K + 6) = - 1
770 GOTO 930
780 GOTO 970
790 IF S(3) = G AND S(7) = 0 THEN 910
800 IF S(9) = G AND S(1) = 0 THEN 880
810 IF S(7) = G AND S(3) = 0 THEN 900
820 IF S(9) = 0 AND S(1) = G THEN 920
830 IF G = - 1 THEN G = 1:H = - 1: GOTO 410
840 IF S(9) = 1 AND S(3) = 0 THEN 890
850 FOR I = 2 TO 9: IF S(I) < > 0 THEN 870
860 S(I) = - 1: GOTO 930
870 NEXT I
880 S(1) = - 1: GOTO 930
890 IF S(1) = 1 THEN 850
900 S(3) = - 1: GOTO 930
910 S(7) = - 1: GOTO 930
920 S(9) = - 1
930 PRINT
940 PRINT "COMPUTEREN VÆLGER....."
950 GOSUB 1180
960 GOTO 1050
970 IF G = 1 THEN 1010
980 IF J = 7 AND K = 3 THEN 1010
990 NEXT K
1000 NEXT J
1010 IF S(5) = G THEN 790
1020 GOTO 830
1030 P$ = "X"
1040 Q$ = "0"
1050 PRINT
1060 PRINT "HVAD VÆLGER DU ";
1070 INPUT M
1080 IF M = 0 THEN PRINT "TAK FOR SPILLET": GOTO
1620
1090 IF M > 9 THEN 1110
1100 IF S(M) = 0 THEN 1140
1110 PRINT "DET FELT ER OPTAGET"
1120 PRINT
1130 GOTO 1050
1140 G = 1
1150 S(M) = 1
1160 GOSUB 1180
1170 GOTO 310
1180 PRINT
1190 FOR I = 1 TO 9
1200 PRINT " ";
1210 IF S(I) < > - 1 THEN 1240
1220 PRINT Q$;" ";
1230 GOTO 1280
1240 IF S(I) < > 0 THEN 1270
1250 PRINT " ";
1260 GOTO 1280
1270 PRINT P$;" ";
1280 IF I < > 3 AND I < > 6 THEN 1320
1290 PRINT
1300 PRINT "-----"
1310 GOTO 1340
1320 IF I = 9 THEN 1340
1330 PRINT "!!";
1340 NEXT I
1350 PRINT
1360 PRINT
1370 FOR I = 1 TO 7 STEP 3
1380 IF S(I) < > S(I + 1) THEN 1420
1390 IF S(I) < > S(I + 2) THEN 1420
1400 IF S(I) = - 1 THEN 1590
1410 IF S(I) = 1 THEN 1570
1420 NEXT I
1430 FOR I = 1 TO 3
1440 IF S(I) < > S(I + 3) THEN 1480
1450 IF S(I) < > S(I + 6) THEN 1480
1460 IF S(I) = - 1 THEN 1590
1470 IF S(I) = 1 THEN 1570
1480 NEXT I
1490 FOR I = 1 TO 9: IF S(I) = 0 THEN 1520
1500 NEXT I
1510 GOTO 1610
1520 IF S(5) < > G THEN 1550
1530 IF S(1) = G AND S(9) = G THEN 1560
1540 IF S(3) = G AND S(7) = G THEN 1560
1550 RETURN
1560 IF G = - 1 THEN 1590
1570 PRINT "DU TÆVEDE MIG!! GODT SPILLET!"
1580 GOTO 1620
1590 PRINT "JEG VINDER, KLUMMERHOVED!"
1600 GOTO 1620
1610 PRINT "SPILLET ER UAFGJORT. TAK!"
1620 END

```

Kryds og bolle i 3-D

Dette spil spilles ligesom almindelig kryds og bolle, men i en kubus med sider på 4 felter ($4 * 4 * 4$). I dette spil skal man have 4 »boller« eller »krydser« på række (eller diagonalt i det tredimensionale plan) for at vinde.

Hvert træk angives med et trecifret tal (de tre cifre skal ikke adskilles af kommaer). Hvert ciffer skal ligge mellem 1 og 4. Cifrene angiver i orden niveau (målt i højden), kolonne og række for trækket. Det kan lade sig gøre at vinde over datamaten, hvis du spiller rigtigt, men dette spil er betydeligt vanskeligere end almindelig kryds og bolle.

Denne version kommer fra Dartmouth College.

```

MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY
-----
                KRYDS-OG-BOLLE I 3-D
-----

```

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

DETTE ER KRYDS-OG-BOLLE SPILLET I 3 DIMENSIONER.

DET SPILLES I EN KASSE, DER MÅLER 4 X 4 X 4 FELTER PR. SIDE.

HVERT TRÆK ANGIVES MED ET TRE-CIFRET TAL, HVOR HVER CIFFER LIGGER MELLEML 1 OG 4 (BEGGE INKLUSIVE). CIFRENE ANGIVER RESPEKTIVT DYBDE, RÆKKE OG KOLONNE I KASSEN.

HVIS DU ØNSKER UDSKRIFT AF SPILLEBRÆTTET SKAL DU BLOT TASTE 'O' SOM ET TRÆK, OG PROGRAMMET VIL UDSKRIVE BRÆTTET MED DINE TRÆK VIST MED ET (D), OG COMPUTERENS TRÆK VIST MED ET (C). FELTER, DER ENDNU IKKE ER BRUGT VISES MED ().

HVIS DU ØNSKER AT STANDSE PROGRAMMET KAN DU TASTE '1' SOM ET TRÆK.

VIL DU BEGYNDE ?JA

DIT TRÆK ?122

DATAMATEN VÆLGER 111

DIT TRÆK ?112

DATAMATEN VÆLGER 411

DIT TRÆK ?412

DATAMATEN VÆLGER 414

DIT TRÆK ?212

GODT FORSØG. DATAMATEN VÆLGER 312

DIT TRÆK ?0

```

(C)      (D)      ( )      ( )
  ( )      (D)      ( )      ( )
    ( )      ( )      ( )      ( )
      ( )      ( )      ( )      ( )

( )      (D)      ( )      ( )
  ( )      ( )      ( )      ( )
    ( )      ( )      ( )      ( )
      ( )      ( )      ( )      ( )

```

```

( )      (C)      ( )      ( )
  ( )      ( )      ( )      ( )
    ( )      ( )      ( )      ( )
      ( )      ( )      ( )      ( )

(C)      (D)      ( )      (C)
  ( )      ( )      ( )      ( )
    ( )      ( )      ( )      ( )
      ( )      ( )      ( )      ( )

```

DIT TRÆK ?442

DATAMATEN VÆLGER 114

DIT TRÆK ?214

DATAMATEN VÆLGER 213 , OG VINDER FØLGENDE:

411 312 213 114

ØNSKER DU ET SPIL TIL ?N

```

100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY      "
110 PRINT "-----"
120 PRINT "                KRYDS-OG-BOLLE I 3-D"
130 PRINT "-----"
140 PRINT
150 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
160 INPUT CS
170 IF LEFT$(CS,1) = "N" THEN 370
180 IF LEFT$(CS,1) = "J" THEN 210
190 PRINT "FORKERT SVAR. TAST VENLIGST 'JA' ELLER 'NEJ'"
200 GOTO 160
210 PRINT
220 PRINT "DETTE ER KRYDS-OG-BOLLE SPILLET I 3 DIMENSIONER."
230 PRINT
240 PRINT "DET SPILLES I EN KASSE, DER MÅLER 4 X 4 X 4 FELTER PR."
250 PRINT "SIDE."
260 PRINT "HVERT TRÆK ANGIVES MED ET TRE-CIFRET TAL, HVOR HVER CIFFER"
270 PRINT "LIGGER MELLEML 1 OG 4 (BEGGE INKLUSIVE). CIFRENE ANGIVER"
280 PRINT "RESPEKTIVT DYBDE, RÆKKE OG KOLONNE I KASSEN."
290 PRINT
300 PRINT "HVIS DU ØNSKER UDSKRIFT AF SPILLEBRÆTTET SKAL DU BLOT"
310 PRINT "TASTE 'O' SOM ET TRÆK, OG PROGRAMMET VIL UDSKRIVE BRÆTTET"
320 PRINT "MED DINE TRÆK VIST MED ET (D), OG COMPUTERENS TRÆK VIST"
330 PRINT "MED ET (C). FELTER, DER ENDNU IKKE ER BRUGT VISES MED ( )."
340 PRINT
350 PRINT "HVIS DU ØNSKER AT STANDSE PROGRAMMET KAN DU TASTE '1' SOM"
360 PRINT "ET TRÆK."
370 PRINT
380 DIM X(64),L(76),M(76,4),Y(16)
390 FOR I = 1 TO 16
400 READ Y(I)
410 NEXT I
420 FOR I = 1 TO 76
430 FOR J = 1 TO 4
440 READ M(I,J)
450 NEXT J
460 NEXT I
470 FOR I = 1 TO 64
480 X(I) = 0
490 NEXT I
500 Z = 1
510 PRINT "VIL DU BEGYNDE ";
520 INPUT SS
530 IF LEFT$(SS,1) = "N" THEN 800
540 IF LEFT$(SS,1) = "J" THEN 580
550 PRINT "FORKERT SVAR. SVAR VENLIGST ENTEN 'JA' ELLER 'NEJ' "
560 PRINT
570 GOTO 510
580 PRINT
590 PRINT
600 PRINT "DIT TRÆK ";
610 INPUT J1
620 PRINT
630 IF J1 = 1 THEN 3050
640 IF J1 < > 0 THEN 680
650 PRINT

```

```

660 GOSUB 2790
670 GOTO 600
680 IF J1 < 111 THEN 3020
690 IF J1 > 444 THEN 3020
700 GOSUB 2740
710 K1 = INT (J1 / 100)
720 J2 = (J1 - K1 * 100)
730 K2 = INT (J2 / 10)
740 K3 = J1 - K1 * 100 - K2 * 10
750 M = 16 * K1 + 4 * K2 + K3 - 20
760 IF X(M) = 0 THEN 790
770 PRINT "DET FELT ER OPTAGET. PRØV IGEN."
780 GOTO 580
790 X(M) = 1
800 GOSUB 1880
810 J = 1
820 I = 1
830 IF J = 1 THEN 910
840 IF J = 2 THEN 980
850 IF J = 3 THEN 1130
860 I = I + 1
870 IF I < = 76 THEN 830
880 J = J + 1
890 IF J < = 3 THEN 820
900 GOTO 1520
910 IF L(I) < > 4 THEN 860
920 PRINT "DU VINDER MED FØLGENDE RÆKKE:"
930 FOR J = 1 TO 4
940 M = M(I,J)
950 GOSUB 1810
960 NEXT J
970 GOTO 1720
980 IF L(I) < > 15 THEN 860
990 FOR J = 1 TO 4
1000 M = M(I,J)
1010 IF X(M) < > 0 THEN 1050
1020 X(M) = 5
1030 PRINT "DATAMATEN VÆLGER";
1040 GOSUB 1810
1050 NEXT J
1060 PRINT ", OG VINDER FØLGENDE:"
1070 PRINT
1080 FOR J = 1 TO 4
1090 M = M(I,J)
1100 GOSUB 1810
1110 NEXT J
1120 GOTO 1720
1130 IF L(I) < > 3 THEN 860
1140 PRINT "GODT FORSØG. DATAMATEN VÆLGER";
1150 FOR J = 1 TO 4
1160 M = M(I,J)
1170 IF X(M) < > 0 THEN 1210
1180 X(M) = 5
1190 GOSUB 1810
1200 GOTO 580
1210 NEXT J
1220 GOTO 1520
1230 I = 1
1240 L(I) = X(M(I,1)) + X(M(I,2)) + X(M(I,3)) + X(M(I,4))
1250 L = L(I)
1260 IF L < 2 THEN 1330
1270 IF L > = 3 THEN 1330
1280 IF L > 2 THEN 2470
1290 FOR J = 1 TO 4
1300 IF X(M(I,J)) < > 0 THEN 1320
1310 X(M(I,J)) = 1 / 8
1320 NEXT J
1330 I = I + 1
1340 IF I < = 76 THEN 1240
1350 GOSUB 1880
1360 I = 1
1370 IF L(I) = 1 / 2 THEN 2600
1380 IF L(I) = 1 + 3 / 8 THEN 2600
1390 I = I + 1
1400 IF I < = 76 THEN 1370
1410 GOTO 2070
1420 Z = 1
1430 IF X(Y(Z)) = 0 THEN 1470
1440 Z = Z + 1
1450 IF Z < > 17 THEN 1430
1460 GOTO 1960
1470 M = Y(Z)
1480 X(M) = 5
1490 PRINT "DATAMATEN VÆLGER";
1500 GOSUB 1810
1510 GOTO 580
1520 I = 1
1530 L(I) = X(M(I,1)) + X(M(I,2)) + X(M(I,3)) + X(M(I,4))

1540 L = L(I)
1550 IF L < 10 THEN 1620
1560 IF L > = 11 THEN 1620
1570 IF L > 10 THEN 2470
1580 FOR J = 1 TO 4
1590 IF X(M(I,J)) < > 0 THEN 1610
1600 X(M(I,J)) = 1 / 8
1610 NEXT J
1620 I = I + 1
1630 IF I < = 76 THEN 1530
1640 GOSUB 1880
1650 I = 1
1660 IF L(I) = .5 THEN 2600
1670 IF L(I) = 5 + 3 / 8 THEN 2600
1680 I = I + 1
1690 IF I < = 76 THEN 1660
1700 GOSUB 2740
1710 GOTO 1230
1720 PRINT
1730 PRINT
1740 PRINT "ØNSKER DU ET SPIL TIL ";
1750 INPUT X$
1760 IF LEFT$(X$,1) = "J" THEN 470
1770 IF LEFT$(X$,1) = "N" THEN 1800
1780 PRINT "FORKERT SVAR ('JA' ELLER 'NEJ') ";
1790 GOTO 1750
1800 GOTO 3050
1810 K1 = INT ((M - 1) / 16) + 1
1820 J2 = M - 16 * (K1 - 1)
1830 K2 = INT ((J2 - 1) / 4) + 1
1840 K3 = M - (K1 - 1) * 16 - (K2 - 1) * 4
1850 M = K1 * 100 + K2 * 10 + K3
1860 PRINT M;
1870 RETURN
1880 FOR S = 1 TO 76
1890 J1 = M(S,1)
1900 J2 = M(S,2)
1910 J3 = M(S,3)
1920 J4 = M(S,4)
1930 L(S) = X(J1) + X(J2) + X(J3) + X(J4)
1940 NEXT S
1950 RETURN
1960 FOR I = 1 TO 64
1970 IF X(I) < > 0 THEN 2040
1980 X(I) = 5
1990 M = I
2000 PRINT "DATAMATEN FORETRÆKKER";
2010 GOSUB 1810
2020 PRINT
2030 GOTO 580
2040 NEXT I
2050 PRINT "SPILLET ER UAFGJORT."
2060 GOTO 1720
2070 FOR K = 1 TO 18
2080 P = 0
2090 FOR I = 4 * K - 3 TO 4 * K
2100 FOR J = 1 TO 4
2110 P = P + X(M(I,J))
2120 NEXT J
2130 NEXT I
2140 IF P < 4 THEN 2180
2150 IF P < 5 THEN 2210
2160 IF P < 9 THEN 2180
2170 IF P < 10 THEN 2210
2180 NEXT K
2190 GOSUB 2740
2200 GOTO 1420
2210 S = 1 / 8
2220 FOR I = 4 * K - 3 TO 4 * K
2230 GOTO 2610
2240 NEXT I
2250 S = 0
2260 GOTO 2220
2270 DATA 1,49,52,4,13,61,64,16,22,39,23,38,26,42,27,43
2280 DATA 1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20
2290 DATA 21,22,23,24,25,26,27,28,29,30,31,32,33,34,35,36,37,38
2300 DATA 39,40,41,42,43,44,45,46,47,48,49,50,51,52,53,54,55,56
2310 DATA 57,58,59,60,61,62,63,64
2320 DATA 1,17,33,49,5,21,37,53,9,25,41,57,13,29,45,61
2330 DATA 2,18,34,50,6,22,38,54,10,26,42,58,14,30,46,62
2340 DATA 3,19,35,51,7,23,39,55,11,27,43,59,15,31,47,63
2350 DATA 4,20,36,52,8,24,40,56,12,28,44,60,16,32,48,64
2360 DATA 1,5,9,13,17,21,25,29,33,37,41,45,49,53,57,61
2370 DATA 2,6,10,14,18,22,26,30,34,38,42,46,50,54,58,62
2380 DATA 3,7,11,15,19,23,27,31,35,39,43,47,51,55,59,63
2390 DATA 4,8,12,16,20,24,28,32,36,40,44,48,52,56,60,64
2400 DATA 1,6,11,16,17,22,27,32,33,38,43,48,49,54,59,64
2410 DATA 13,10,7,4,29,26,23,20,45,42,39,36,61,58,55,52

```

```


2420 DATA 1,21,41,61,2,22,42,62,3,23,43,63,4,24,44,64
2430 DATA 49,37,25,13,50,38,26,14,51,39,27,15,52,40,28,16
2440 DATA 1,18,35,52,5,22,39,56,9,26,43,60,13,30,47,64
2450 DATA 49,34,19,4,53,38,23,8,57,42,27,12,61,46,31,16
2460 DATA 1,22,43,64,16,27,38,49,4,23,42,61,13,26,39,52
2470 FOR J = 1 TO 4
2480 IF X(M(I,J)) < > 1 / 8 THEN 2570
2490 X(M(I,J)) = 5
2500 IF L(I) < 5 THEN 2530
2510 PRINT "LAD OS SR SE HVORDAN DU KOMMER UD AF DENNEHER:"
2520 GOTO 2540
2530 PRINT "DIN RÆV. DET VAR I SIDSTE ØJEBLIK. DATAMATEN FLYTTER TIL"
2540 M = M(I,J)
2550 GOSUB 1810
2560 GOTO 580
2570 NEXT J
2580 PRINT "DATAMATEN OPGIVER!"
2590 GOTO 1720
2600 S = 1 / 8
2610 IF I - INT ( I / 4 ) * 4 > 1 THEN 2640
2620 A = 1
2630 GOTO 2650
2640 A = 2
2650 FOR J = A TO 5 - A STEP 5 - 2 * A
2660 IF X(M(I,J)) = S THEN 2690
2670 NEXT J
2680 GOTO 2240
2690 X(M(I,J)) = 5
2700 M = M(I,J)
2710 PRINT "DATAMATENS TRÆK";
2720 GOSUB 1810
2730 GOTO 580
2740 FOR I = 1 TO 64
2750 IF X(I) < > 1 / 8 THEN 2770
2760 X(I) = 0
2770 NEXT I
2780 RETURN
2790 FOR I = 1 TO 4
2800 FOR J = 1 TO 4
2810 FOR I1 = 1 TO J
2820 PRINT " ";
2830 NEXT I1
2840 FOR K = 1 TO 4
2850 Q = 16 * I + 4 * J + K - 20
2860 IF X(Q) < > 0 THEN 2880
2870 PRINT "( ) ";
2880 IF X(Q) < > 5 THEN 2900
2890 PRINT "(C) ";
2900 IF X(Q) < > 1 THEN 2920
2910 PRINT "(D) ";
2920 IF X(Q) < > 1 / 8 THEN 2940
2930 PRINT "( ) ";
2940 NEXT K
2950 PRINT
2960 PRINT
2970 NEXT J
2980 PRINT
2990 PRINT
3000 NEXT I
3010 RETURN
3020 PRINT "ULOVLIGT TRÆK, PRØV IGEN --"
3030 PRINT
3040 GOTO 610
3050 END

```

Kubus

Kubus er et spil, der spilles på tre tilstødende sider af en kubus, der har sidelængden 2. Enhver position bestemmes ved hjælp af tre tal, f.eks. 1,2,1. Målet med spillet er at bevæge sig fra 1,1,1 til 3,3,3 ved at flytte en vandret eller lodret position (men ikke en diagonal!) ad gangen, uden at støde på en af de 5 miner, som datamaten har placeret tilfældigt i fem af felterne. Når spillet begynder, får du 500 kroner, og for hvert spil skal du sætte et beløb på, om du mener, at du kommer til at nå dit mål. Du har tabt, hvis du støder på en mine, eller hvis du foretager et forkert træk (f.eks. hvis du ændrer mere end ét ciffer på din foregående position).

Kubus blev skabt af Jerimac Ratliff fra Fort Worth, Texas.



MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

KUBUS

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?

I DETTE SPIL ER DATAMATENS TILFÆLDIGE BESLUTNINGER DIN MODSTANDER. SPILLEBRÆTTET ER EN KUBUS MED EN SIDELÆNGDE PÅ 3. HVER AF DE 27 POSITIONER KAN ANGIVES VED AT INDTASTE 3 TAL, F.EKS.: '2,3,1'.

NÅR SPILLET BEGYNDER BLIVER DU AUTOMATISK PLACERET PÅ POSITION 1,1,1. DET GÆLDER OM AT KOMME HEN TIL POSITION 3,3,3. DET LYDER MÅSKE IKKE SÅ SVÆRT, MEN COMPUTEREN VÆLGER - TILFÆLDIGT - 5 POSITIONER, HVOR DER PLACERES LANDMINER!

<< TAST 'RETURN' FOR AT FORTSÆTTE >> ?

HVIS DU GÅR HEN PÅ EN POSITION, HVOR DER ER EN LANDMINE, SÅ VIL DU TABE.

EN ANDEN DETALJE ER, AT DU KUN KAN BEVÆGE DIG 1 POSITION I 1 RETNING AD GANGEN, OG IKKE DIAGONALT. F.EKS. KAN DU GÅ FRA 1,1,2 TIL 2,1,2 ELLER 1,1,3. DU MÅ ALTSÅ IKKE UDSKIFTE TO AF TALLENE I

DET SAMME TRÆK. HVIS DU FORETAGER ET ULOVLIGT TRÆK, SÅ TABER DU, OG DATAMATEN TAGER ALLE DE PENGE DU HAR SATSET I DEN RUNDE.

ALLE JA/NEJ-SPØRGSMÅL I SPILLET SKAL BESVARES MED 1 FOR JA, OG 0 FOR NEJ. NÅR DU INDTASTER INDSATSENS STØRRELSE, SÅ SKRIV KUN KRONEBELØBET, F.EKS. '100' DU FÅR AUTOMATISK INDSAT 500 KRONER PÅ DIN KONTO NÅR SPILLET STARTER.

HELD OG LYKKE.....

VIL DU GØRE EN INDSATS (1=JA, 0=NEJ) ?1

HVOR STOR SKAL INDSATSEN VÆRE ?170

DET ER DIN TUR ---

INDTAST POSITION ?1,1,1
 NÆSTE TRÆK ?1,2,1
 NÆSTE TRÆK ?1,2,2
 NÆSTE TRÆK ?2,2,2
 NÆSTE TRÆK ?2,3,2
 NÆSTE TRÆK ?2,3,3
 NÆSTE TRÆK ?3,3,3

TILLYKKE!

NU HAR DU 670 KRONER.

VIL DU SPILLE IGEN ('J' ELLER 'N') ?J

VIL DU GØRE EN INDSATS (1=JA, 0=NEJ) ?1

HVOR STOR SKAL INDSATSEN VÆRE ?200

DET ER DIN TUR ---

INDTAST POSITION ?1,1,1
 NÆSTE TRÆK ?1,2,1
 NÆSTE TRÆK ?2,2,1
 NÆSTE TRÆK ?2,3,1
 NÆSTE TRÆK ?2,3,2
 NÆSTE TRÆK ?2,3,3

***** B A N G *****

DU HAR TABT.....

DU HAR NU 470 KRONER.

VIL DU SPILLE IGEN ('J' ELLER 'N') ?

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                KUBUS"
130 PRINT "-----"
140 PRINT
150 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
160 INPUT A$
170 PRINT
180 IF LEFT$(A$,1) = "N" THEN 590
190 PRINT "I DETTE SPIL ER DATAMATENS TILFÆLDIGE"
200 PRINT "BESLUTNINGER DIN MODSTANDER. SPILLE-"
210 PRINT "BRÆTTET ER EN KUBUS MED EN SIDELÆNGDE-"
220 PRINT "PÅ 3. HVER AF DE 27 POSITIONER KAN"
230 PRINT "ANGIVES VED AT INDTASTE 3 TAL, F.EKS.:"
240 PRINT "'2,3,1'."
250 PRINT
260 PRINT "NÅR SPILLET BEGYNDER BLIVER DU AUTOMA-"
270 PRINT "TISK PLACERET PÅ POSITION 1,1,1."
280 PRINT "DET GÆLDER OM AT KOMME HEN TIL POSI-"
290 PRINT "TION 3,3,3. DET LYDER MÅSKE IKKE SÅ"
300 PRINT "SVÆRT, MEN COMPUTEREN VÆLGER - TILFÆL-"
310 PRINT "DIGT - 5 POSITIONER, HVOR DER PLACERES"
320 PRINT "LANDMINER!"
330 PRINT
340 PRINT "<< TAST 'RETURN' FOR AT FORTSÆTTE >> ";
350 INPUT A$
360 PRINT
370 PRINT "HVIS DU GÅR HEN PÅ EN POSITION, HVOR"
380 PRINT "DER ER EN LANDMINE, SÅ VIL DU TABE."
390 PRINT

```

```

400 PRINT "EN ANDEN DETALJE ER, AT DU KUN KAN BE-"
410 PRINT "VÆGE DIG 1 POSITION I 1 RETNING AD GAN-"
420 PRINT "GEN, OG IKKE DIAGONALT. F.EKS. KAN DU"
430 PRINT "GR FRA 1,1,2 TIL 2,1,2 ELLER 1,1,3. DU"
440 PRINT "MR ALTSÅ IKKE UDSKIFTE TO AF TALLENE I"
450 PRINT "DET SAMME TRÆK. HVIS DU FORETAGER ET"
460 PRINT "ULOVLIGT TRÆK, SÅ TABER DU, OG DATAMA-"
470 PRINT "TEN TAGER ALLE DE PENGE DU HAR SATSET I"
480 PRINT "DEN RUNDE."
490 PRINT
500 PRINT "ALLE JA/NEJ-SPØRGSMÅL I SPILLET SKAL"
510 PRINT "BESVARES MED 1 FOR JA, OG 0 FOR NEJ."
520 PRINT "NÅR DU INDTASTER INDSATSENS STØRRELSE,"
530 PRINT "SÅ SKRIV KUN KRONEBELØBET, F.EKS. '100'"
540 PRINT "DU FÅR AUTOMATISK INDSAT 500 KRONER PÅ"
550 PRINT "DIN KONTO NÅR SPILLET STARTER."
560 PRINT
570 PRINT "HELD OG LYKKE....."
580 PRINT
590 A1 = 500
600 A = INT (3 * RND (1))
610 IF A < > 0 THEN 630
620 A = 3
630 B = INT (3 * RND (1))
640 IF B < > 0 THEN 660
650 B = 2
660 C = INT (3 * RND (1))
670 IF C < > 0 THEN 690
680 C = 3
690 D = INT (3 * RND (1))
700 IF D < > 0 THEN 720
710 D = 1
720 E = INT (3 * RND (1))
730 IF E < > 0 THEN 750
740 E = 3
750 F = INT (3 * RND (1))
760 IF F < > 0 THEN 780
770 F = 3
780 G = INT (3 * RND (1))
790 IF G < > 0 THEN 810
800 G = 3
810 H = INT (3 * RND (1))
820 IF H < > 0 THEN 840
830 H = 3
840 I = INT (3 * RND (1))
850 IF I < > 0 THEN 870
860 I = 2
870 J = INT (3 * RND (1))
880 IF J < > 0 THEN 900
890 J = 3
900 K = INT (3 * RND (1))
910 IF K < > 0 THEN 930
920 K = 2
930 L = INT (3 * RND (1))
940 IF L < > 0 THEN 960
950 L = 3
960 M = INT (3 * RND (1))
970 IF M < > 0 THEN 990
980 M = 3
990 N = INT (3 * RND (1))
1000 IF N < > 0 THEN 1020
1010 N = 1
1020 O = INT (3 * RND (1))
1030 IF O < > 0 THEN 1050
1040 O = 3
1050 PRINT "VIL DU GØRE EN INDSATS (1=JA, 0=NEJ) ";
1060 Z1 = 0
1070 INPUT Z
1080 IF Z = 0 THEN 1130
1090 PRINT
1100 PRINT "HVOR STOR SKAL INDSATSEN VÆRE ";
1110 INPUT Z1
1120 IF A1 < Z1 THEN 1840
1130 W = 1
1140 X = 1
1150 Y = 1
1160 PRINT
1170 PRINT "DET ER DIN TUR ---"
1180 PRINT
1190 PRINT "INDTAST POSITION ";
1200 INPUT P,Q,R
1210 IF P > W + 1 THEN 1300
1220 IF P = W + 1 THEN 1270
1230 IF Q > X + 1 THEN 1300
1240 IF Q = X + 1 THEN 1280
1250 IF R > Y + 1 THEN 1300
1260 GOTO 1340
1270 IF Q > = X + 1 THEN 1300

1280 IF R > = Y + 1 THEN 1300
1290 GOTO 1340
1300 PRINT
1310 PRINT "ULOVLIGT TRÆK!!!!!! *&~%`&(f DU TABER"
1320 PRINT
1330 GOTO 1740
1340 W = P
1350 X = Q
1360 Y = R
1370 IF P = 3 THEN 1390
1380 GOTO 1420
1390 IF Q = 3 THEN 1410
1400 GOTO 1420
1410 IF R = 3 THEN 1870
1420 IF P = (A) THEN 1440
1430 GOTO 1470
1440 IF Q = B THEN 1460
1450 GOTO 1470
1460 IF R = C THEN 1690
1470 IF P = D THEN 1490
1480 GOTO 1520
1490 IF Q = E THEN 1510
1500 GOTO 1520
1510 IF R = F THEN 1690
1520 IF P = G THEN 1540
1530 GOTO 1570
1540 IF Q = H THEN 1560
1550 GOTO 1570
1560 IF R = I THEN 1690
1570 IF P = J THEN 1590
1580 GOTO 1620
1590 IF Q = K THEN 1610
1600 GOTO 1620
1610 IF R = L THEN 1690
1620 IF P = M THEN 1640
1630 GOTO 1670
1640 IF Q = N THEN 1660
1650 GOTO 1670
1660 IF R = O THEN 1690
1670 PRINT "NÆSTE TRÆK"
1680 GOTO 1200
1690 PRINT
1700 PRINT "***** B A N G *****"
1710 PRINT
1720 PRINT "DU HAR TABT....."
1730 PRINT
1740 IF Z = 0 THEN 1940
1750 PRINT
1760 Z2 = A1 - Z1
1770 IF Z2 > 0 THEN 1800
1780 PRINT "DU ER HELT FLAD....."
1790 GOTO 1980
1800 PRINT "DU HAR NU";Z2;"KRONER."
1810 PRINT
1820 A1 = Z2
1830 GOTO 1940
1840 PRINT "NU SNYDER DU IGEN! NY INDSATS ";
1850 PRINT
1860 GOTO 1110
1870 PRINT "TILLYKKE!"
1880 PRINT
1890 IF Z = 0 THEN 1940
1900 Z2 = A1 + Z1
1910 PRINT "NU HAR DU";Z2;"KRONER."
1920 PRINT
1930 A1 = Z2
1940 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1950 INPUT A$
1960 PRINT
1970 IF LEFT$(A$,1) = "J" THEN 600
1980 PRINT "DET VAR ÆRGELIGT."
1990 PRINT
2000 PRINT "NR, MEN FARVEL MED DIG."
2010 END
2020 X = RND ( - 1)
2030 FOR I = 1 TO 1000
2040 PRINT RND (1)
2050 NEXT
2060 END

```



```

1190 IF R = H THEN 1290
1200 IF W(R + 1,S) < > 0 THEN 1290
1210 IF S < > V THEN 1250
1220 IF Z = 1 THEN 1280
1230 Q = 1
1240 GOTO 1440
1250 IF W(R,S + 1) < > 0 THEN 1280
1260 X = INT ( RND (1) * 2 + 1)
1270 ON X GOTO 1500,1590
1280 GOTO 1500
1290 IF S < > V THEN 1330
1300 IF Z = 1 THEN 1350
1310 Q = 1
1320 GOTO 1340
1330 IF W(R,S + 1) < > 0 THEN 1350
1340 GOTO 1590
1350 GOTO 1790
1360 W(R - 1,S) = C
1370 C = C + 1
1380 V(R - 1,S) = 2
1390 R = R - 1
1400 IF C = H * V + 1 THEN 1800
1410 Q = 0
1420 GOTO 620
1430 W(R,S - 1) = C
1440 C = C + 1
1450 V(R,S - 1) = 1
1460 S = S - 1
1470 IF C = H * V + 1 THEN 1800
1480 Q = 0
1490 GOTO 620
1500 W(R + 1,S) = C
1510 C = C + 1
1520 IF V(R,S) = 0 THEN 1550
1530 V(R,S) = 3
1540 GOTO 1560
1550 V(R,S) = 2
1560 R = R + 1
1570 IF C = H * V + 1 THEN 1800
1580 GOTO 980
1590 IF Q = 1 THEN 1690
1600 W(R,S + 1) = C
1610 C = C + 1
1620 IF V(R,S) = 0 THEN 1650
1630 V(R,S) = 3
1640 GOTO 1660
1650 V(R,S) = 1
1660 S = S + 1
1670 IF C = H * V + 1 THEN 1800
1680 GOTO 620
1690 Z = 1
1700 IF V(R,S) = 0 THEN 1740
1710 V(R,S) = 3
1720 Q = 0
1730 GOTO 1790
1740 V(R,S) = 1
1750 Q = 0
1760 R = 1
1770 S = 1
1780 GOTO 610
1790 GOTO 520
1800 FOR J = 1 TO V
1810 PRINT "!" ;
1820 FOR I = 1 TO H
1830 IF V(I,J) < 2 THEN 1860
1840 PRINT " " ;
1850 GOTO 1870
1860 PRINT " !" ;
1870 NEXT I
1880 PRINT
1890 FOR I = 1 TO H
1900 IF V(I,J) = 0 THEN 1940
1910 IF V(I,J) = 2 THEN 1940
1920 PRINT ": " ;
1930 GOTO 1950
1940 PRINT ":-" ;
1950 NEXT I
1960 PRINT ":" ;
1970 NEXT J
1980 END

```

Lige vinder

Dette spil spilles mellem dig og datamaten. Før spillet starter, placeres et lige antal genstande på en række, f.eks. glaskugler, jetoner eller tændstikker. Du og datamaten skiftes til at tage mellem 1 og 4 glaskugler op. Spillet er slut, når der ikke er flere kugler at tage. Den, der har taget et lige antal kugler op, har vundet.

Der findes to versioner af spillet. Selv om de for spilleren ser ud til at være ens, er de programmeret på meget forskellig vis. Lige vinder (den første version) er deterministisk, dvs. datamaten spiller efter bestemte sikre regler og kan ikke overvindes, hvis man ikke selv ved, hvordan spillet skal spilles. Der startes altid med 27 glaskugler, selv om denne værdi kan ændres i programlinjerne 430 og 1330.

Den anden version (Lige vinder, version 2) er meget mere interessant, fordi datamaten starter med kun at kende spillets regler. Ved hjælp af en enkel teknik for kunstig intelligens (kybernetik) lærer datamaten af sine fejltrin indtil den spiller udmærket. Efter tyve spil er datamaten virkelig svær at slå. Hvis mennesket spiller med forskellige taktikker, synes datamaten at lære spillet hurtigere. Hvis du prøver at optegne indlæringskurven for programmet, vil du opdage, at den minder meget om den klassiske, som vi kender fra psykologiske eksperimenter.

Eric Peters fra DEC skrev Lige vinder, version 2. Den oprindelige forfatter til Lige vinder er ukendt.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

LIGE VINDER

DETTE ER ET SPIL FOR TO PERSONER, DER HEDDER 'LIGE VINDER'. SPILLERNE SKAL BRUGE 27 GLASKUGLER ELLER 27 ANDRE OBJEKTER, SOM LÆGGES PÅ ET BORD.

DE TO SPILLERE SKIFTES TIL AT FJERNE FRA 1 TIL 4 KUGLER FRA BORDET. SPILLET ER SLUT NÅR DER IKKE ER FLERE KUGLER TILBAGE.

DEN SPILLER, DER HAR TAGET ET LIGE ANTAL KUGLER HAR VUNDET!

DE ENESTE REGLER DER FINDES ER DISSE:

- (1) SPILLERNE SKAL SKIFTES.
- (2) DER MÅ KUN TAGES 1-4 KUGLER PR. TRÆK.
- (3) DU MÅ IKKE SIDDE ET TRÆK OVER.

TAST '1' HVIS DU VIL BEGYNDE,
'0' HVIS JEG SKAL BEGYNDE. ?1

ANTAL PÅ BORDET: 27

HVOR MANGE TAGER DU ?4

ANTAL PÅ BORDET: 23

DU HAR IALT 4 GLASKUGLER.

JEG TAGER 4 GLASKUGLER.

ANTAL PÅ BORDET: 19

JEG HAR IALT 4 GLASKUGLER.

HVOR MANGE TAGER DU ?4

ANTAL PÅ BORDET: 15

DU HAR IALT 8 GLASKUGLER.

JEG TAGER 2 GLASKUGLER.

ANTAL PÅ BORDET: 13

JEG HAR IALT 6 GLASKUGLER.

HVOR MANGE TAGER DU ?4

ANTAL PÅ BORDET: 9

DU HAR IALT 12 GLASKUGLER.

JEG TAGER 2 GLASKUGLER.

ANTAL PÅ BORDET: 7

JEG HAR IALT 8 GLASKUGLER.

HVOR MANGE TAGER DU ?4

ANTAL PÅ BORDET: 3

DU HAR IALT 16 GLASKUGLER.

JEG TAGER 2 GLASKUGLER.

ANTAL PÅ BORDET: 1

JEG HAR IALT 10 GLASKUGLER.

HVOR MANGE TAGER DU ?1

DET VAR DET.....

JEG HAR 10 - OG DU HAR 17

JEG VANDT....

VIL DU SPILLE IGEN ('J' ELLER 'N') ?N

DET ER IORDEN, PÅ GENSYN EN ANDEN GANG!

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                LIGE VINDER"
130 PRINT "-----"
140 PRINT
150 Y1 = 0
160 M1 = 0
170 DIM M(20),Y(20)
180 PRINT "DETTE ER ET SPIL FOR TO PERSONER, DER"
190 PRINT "HEDDER 'LIGE VINDER'. SPILLERNE SKAL"
200 PRINT "BRUGE 27 GLASKUGLER ELLER 27 ANDRE"
210 PRINT "OBJEKTER, SOM LÆGGES PÅ ET BORD."
220 PRINT
230 PRINT "DE TO SPILLERE SKIFTES TIL AT FJERNE"
240 PRINT "FRA 1 TIL 4 KUGLER FRA BORDET. SPILLET"
250 PRINT "ER SLUT NÅR DER IKKE ER FLERE KUGLER"
260 PRINT "TILBAGE."
270 PRINT
280 PRINT "DEN SPILLER, DER HAR TAGET ET LIGE"
290 PRINT "ANTAL KUGLER HAR VUNDET!"
300 PRINT
310 PRINT "DE ENESTE REGLER DER FINDES ER DISSE:"
320 PRINT
330 PRINT " (1) SPILLERNE SKAL SKIFTES."
340 PRINT " (2) DER MÅ KUN TAGES 1-4 KUGLER PR."
350 PRINT "     TRÆK."
360 PRINT " (3) DU MÅ IKKE SIDDE ET TRÆK OVER."
370 PRINT
380 PRINT "TAST '1' HVIS DU VIL BEGYNDE,"
390 PRINT "     '0' HVIS JEG SKAL BEGYNDE. ";
400 INPUT C
410 IF C = 0 THEN 430
420 GOTO 1330
430 T = 27
440 M = 2
450 PRINT
460 PRINT "ANTAL PÅ BORDET: ";T
470 PRINT
480 M1 = M1 + M
490 T = T - M
500 PRINT "JEG TAGER ";M;"GLASKUGLER."
510 PRINT
520 IF T = 0 THEN 1110
530 PRINT "ANTAL PÅ BORDET: ";T
540 PRINT
550 PRINT "JEG HAR IALT ";M1;"GLASKUGLER."
560 PRINT
570 PRINT "HVOR MANGE TAGER DU ";
580 INPUT Y
590 IF Y < 1 THEN 1400
600 IF Y > 4 THEN 1400
610 IF Y < = T THEN 660
620 PRINT
621 PRINT "DU FORSØGER AT TAGE FLERE KUGLER END"
630 PRINT "DER FINDES PÅ BORDET.....PRØV IGEN"
640 PRINT
650 GOTO 580
660 Y1 = Y1 + Y
670 T = T - Y
680 IF T = 0 THEN 1110
690 PRINT
700 PRINT "ANTAL PÅ BORDET: ";T
710 PRINT
720 PRINT "DU HAR IALT ";Y1;"GLASKUGLER."
730 PRINT
740 IF T < .5 THEN 1110
750 R = T - 6 * INT (T / 6)
760 IF INT (Y1 / 2) = Y1 / 2 THEN 950
770 IF T < 4.2 THEN 830
780 IF R > 3.4 THEN 870
790 M = R + 1
800 M1 = M1 + M
810 T = T - M
820 GOTO 500
830 M = T
840 T = T - M
850 GOTO 1070
860 REM   Ved 250 vinder jeg.
870 IF R < 4.7 THEN 910
880 IF R > 3.5 THEN 910
890 M = 1
900 GOTO 920
910 M = 4
920 T = T - M
930 M1 = M1 + M
940 GOTO 500
950 REM   Jeg er rede til at beregne strategien,
960 REM   hvis modstanderens totale antal er lige.

```

```

970 IF R < 1.5 THEN 1290
980 IF R > 5.3 THEN 1290
990 M = R - 1
1000 M1 = M1 + M
1010 T = T - M
1020 IF T < .2 THEN 1040
1030 GOTO 500
1040 PRINT "JEG TAGER";M;"GLASKUGLER."
1050 PRINT
1060 GOTO 1110
1070 PRINT "JEG TAGER";M;"GLASKUGLE."
1080 PRINT
1090 PRINT "DER ER IKKE FLERE KUGLER PÅ BORDET!!!!"
1100 M1 = M1 + M
1110 PRINT
1111 PRINT "DET VAR DET....."
1120 PRINT
1130 PRINT "JEG HAR";M1;"- OG DU HAR";Y1
1140 PRINT
1150 IF INT (M1 / 2) = M1 / 2 THEN 1190
1160 PRINT "DU VANDT!!!!!"
1170 PRINT
1180 GOTO 1210
1190 PRINT "JEG VANDT...."
1200 PRINT
1210 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1220 INPUT A$
1240 PRINT
1250 IF LEFT$(A$,1) = "N" THEN 1300
1260 M1 = 0
1270 Y1 = 0
1280 GOTO 380
1290 GOTO 890
1300 PRINT "DET ER IORDEN, PÅ GENSYN EN ANDEN GANG!"
1320 GOTO 1460
1330 T = 27
1340 PRINT
1350 PRINT "ANTAL PÅ BORDET:";T
1360 PRINT
1370 PRINT "HVOR MANGE TAGER DU ";
1380 INPUT Y
1390 GOTO 590
1400 PRINT
1410 PRINT "DET ANTAL KUGLER DU INDTASTER, SKAL"
1420 PRINT "VÆRE ET POSITIVT TAL MELLEM 1 OG 4."
1430 PRINT
1440 PRINT "HVOR MANGE TAGER DU ";
1450 GOTO 580
1460 END

```

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

LIGE VINDER, VERSION 2

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

VED SPILLET'S BEGYNDELSE, BLIVER ET TIL-
FÆLDIGT ANTAL GLASKUGLER PLACERET PÅ
BORDET. ANTALLET AF KUGLER ER ALTID
ULIGE. EFTER TUR SKAL DE TO SPILLERE
TAGE 1, 2, 3 ELLER 4 KUGLER. VINDEREN
ER DEN SPILLER SOM - NÅR DER IKKE ER
FLERE KUGLER PÅ BORDET - HAR TAGET ET
LIGE ANTAL KUGLER.

<< TAST 'RETURN' FOR AT FORTSÆTTE >> ?

VED SPILLET'S BEGYNDELSE KENDER DATAMA-
TEN KUN SPILLET'S REGLER. JO FLERE SPIL
DER SPILLES, JO DYGTIGERE BLIVER DEN
TIL AT SPILLE. DET VIL DERFOR VÆRE VAN-
SKELIGT AT VINDE OVER DATAMATEN 20 GAN-
I TRÆK.....MEN PRØV ALLIGEVEL!!!!

HVIS DU ØNSKER AT HOLDE OP MED AT SPIL-
LE, SKAL DU BLOT BRUGE VÆRDIEN '0' FOR
ET TRÆK.

DER ER 17 GLASKUGLER PÅ BORDET.
DATAMATEN TAGER 4 GLASKUGLER.
DER ER NU 13 TILBAGE PÅ BORDET.
HVOR MANGE TAGER DU ?4

DER ER 9 GLASKUGLER PÅ BORDET.
DATAMATEN TAGER 4 GLASKUGLER.
DER ER NU 5 TILBAGE PÅ BORDET.
HVOR MANGE TAGER DU ?3

DER ER 2 GLASKUGLER PÅ BORDET.
DATAMATEN TAGER 2 GLASKUGLER.

SPILET ER FORBI.....JEG VANDT!!!!

DER ER 9 GLASKUGLER PÅ BORDET.
DATAMATEN TAGER 4 GLASKUGLER.
DER ER NU 5 TILBAGE PÅ BORDET.
HVOR MANGE TAGER DU ?1

DER ER 4 GLASKUGLER PÅ BORDET.
DATAMATEN TAGER 4 GLASKUGLER.

SPILET ER FORBI.....JEG VANDT!!!!

DER ER 19 GLASKUGLER PÅ BORDET.
DATAMATEN TAGER 4 GLASKUGLER.
DER ER NU 15 TILBAGE PÅ BORDET.
HVOR MANGE TAGER DU ?3

DER ER 12 GLASKUGLER PÅ BORDET.
DATAMATEN TAGER 4 GLASKUGLER.
DER ER NU 8 TILBAGE PÅ BORDET.
HVOR MANGE TAGER DU ?1

DER ER 7 GLASKUGLER PÅ BORDET.
DATAMATEN TAGER 4 GLASKUGLER.
DER ER NU 3 TILBAGE PÅ BORDET.
HVOR MANGE TAGER DU ?2

DER ER 1 GLASKUGLE TILBAGE PÅ BORDET.
DATAMATEN TAGER 1 GLASKUGLE.

SPILET ER FORBI.....DU VINDER!!!!

DER ER 13 GLASKUGLER PÅ BORDET.
DATAMATEN TAGER 1 GLASKUGLE.
DER ER NU 12 TILBAGE PÅ BORDET.
HVOR MANGE TAGER DU ?

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          LIGE VINDER, VERSION 2"
130 PRINT "-----"
140 PRINT
150 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
160 INPUT A$
170 IF LEFT$(A$,1) = "N" THEN 400
180 PRINT "VED SPILLET'S BEGYNDELSE, BLIVER ET TIL-"
190 PRINT "FÆLDIGT ANTAL GLASKUGLER PLACERET PÅ"
200 PRINT "BORDET. ANTALLET AF KUGLER ER ALTID"
210 PRINT "ULIGE. EFTER TUR SKAL DE TO SPILLERE"
220 PRINT "TAGE 1, 2, 3 ELLER 4 KUGLER. VINDEREN"
230 PRINT "ER DEN SPILLER SOM - NÅR DER IKKE ER"
240 PRINT "FLERE KUGLER PÅ BORDET - HAR TAGET ET"
250 PRINT "LIGE ANTAL KUGLER."
260 PRINT
270 PRINT "<< TAST 'RETURN' FOR AT FORTSÆTTE >> ";
280 INPUT A$
290 PRINT
300 PRINT "VED SPILLET'S BEGYNDELSE KENDER DATAMA-"
310 PRINT "TEN KUN SPILLET'S REGLER. JO FLERE SPIL"
320 PRINT "DER SPILLES, JO DYGTIGERE BLIVER DEN"
330 PRINT "TIL AT SPILLE. DET VIL DERFOR VÆRE VAN-"
340 PRINT "SKELIGT AT VINDE OVER DATAMATEN 20 GAN-"
350 PRINT "I TRÆK.....MEN PRØV ALLIGEVEL!!!!!"
360 PRINT
370 PRINT "HVIS DU ØNSKER AT HOLDE OP MED AT SPIL-"
380 PRINT "LE, SKAL DU BLOT BRUGE VÆRDIEN '0' FOR"
390 PRINT "ET TRÆK."
400 PRINT
410 DIM R(1,5)
420 L = 0
430 B = 0
440 FOR I = 0 TO 5
450 R(1,I) = 4
460 R(0,I) = 4
470 NEXT I
480 A = 0
490 B = 0
500 P = INT ((13 * RND (1) + 9) / 2) * 2 + 1
510 IF P = 1 THEN 1000
520 PRINT "DER ER";P;"GLASKUGLER PÅ BORDET."
530 E1 = E
540 L1 = L
550 E = (A / 2 - INT (A / 2)) * 2
560 L = INT ((P / 6 - INT (P / 6)) * 6 + .5)

```

```

570 IF R(E,L) > = P THEN 760
580 M = R(E,L)
590 IF M < = 0 THEN 820
600 P = P - M .
610 IF M = 1 THEN 960
620 PRINT "DATAMATEN TAGER";M;"GLASKUGLER."
630 PRINT "DER ER NU";P;"TILBAGE PÅ BORDET."
640 PRINT "HVOR MANGE TAGER DU ";
650 B = B + M
660 INPUT M
670 PRINT
680 M = INT (M)
690 IF M < 1 THEN 900
700 IF M > 4 THEN 910
710 IF M > P THEN 910
720 IF M = P THEN 810
730 P = P - M
740 A = A + M
750 GOTO 510
760 IF P = 1 THEN 1020
770 PRINT "DATAMATEN TAGER";P;"GLASKUGLER."
780 PRINT
790 R(E,L) = P
800 B = B + P
810 IF B / 2 = INT (B / 2) THEN 870
820 PRINT "SPILLET ER FORBI.....DU VINDER!!!!"
830 PRINT
840 IF R(E,L) = 1 THEN 930
850 R(E,L) = R(E,L) - 1
860 GOTO 480
870 PRINT "SPILLET ER FORBI.....JEG VANDT!!!!"
880 PRINT
890 GOTO 480
900 IF M = 0 THEN 1050
910 PRINT "DU KAN IKKE TAGE";M;"PRØV IGEN."
920 GOTO 660
930 IF R(E1,L1) = 1 THEN 480
940 R(E1,L1) = R(E1,L1) - 1
950 GOTO 480
960 PRINT "DATAMATEN TAGER 1 GLASKUGLE."
970 PRINT "DER ER NU";P;"TILBAGE PÅ BORDET."
980 PRINT "HVOR MANGE TAGER DU ";
990 GOTO 650
1000 PRINT "DER ER 1 GLASKUGLE TILBAGE PÅ BORDET."
1010 GOTO 530
1020 PRINT "DATAMATEN TAGER 1 GLASKUGLE."
1030 PRINT
1040 GOTO 790
1050 END

```

Litteratur quiz

Dette er et enkelt program af CAI-typen. CAI betyder Computer Aided Instruction, dvs. indlæring ved hjælp af datamat. Programmet stiller fire spørgsmål, hvor du skal vælge mellem nogle svarmuligheder. Spørgsmålene drejer sig om klassiske børnebøger. Når programmet køres, gives de nødvendige forklaringer.

Programmet er skrevet af Pamela McGinley fra DEC.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
-----
                LITTERATUR QUIZ
-----
```

HER KAN DU TESTE DIN VIDEN OM
BØRNEBØGER.

DET ER EN MULTIPLE-CHOICE QUIZ.

TAST 1, 2, 3 ELLER 4 EFTER SPØRGSMÅLS-
TEGNET.

HELD OG LYKKE!

HVAD HEDDER KATTEN I 'PINOCCHIO' ?

1=TIGER 2=CICERO 3=FIGARO 4=GUIPETTO ?1

DESVÆRRE....DEN HED FIGARO.

FRA HVIS HAVE STJÆLER SNURRE SNUP GULE-
RØDDER ?

1=HR. JENSSENS 2=ELMER FJOTS
3=CLEM JUDD'S 4=STROMBOLI'S ?3

ÆRGELIGT. DET ER FRA ELMER FJOTS HAVE.

HVAD HEDDER DOROTHY'S HUND I EVENTYRET
'TROLDMANDEWN FRA OZ' ?

1=CICERO 2=TRIXIE 3=KING 4=TOTO ?3

LÆS BOGEN IGEN...HUNDEN HED TOTO.

HVAD HED DEN SMUKKE MØ, DER SPISTE DET
FORGIFTEDE ÆBLE ?

1=EVA 2=ASKEPOT 3=SNEHVIDE 4=WENDY ?1

HOLD DA OP, DET VAR DA SNEHVIDE!

BAH. DET VAR DEFINITIVT IKKE NOGET AT
PRALE MED. TILBAGE TIL BØRNEHAVEN MED
DIG, GAMLE VEN....

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "                LITTERATUR QUIZ"
130 PRINT "-----"
140 PRINT
150 R = 0
160 PRINT "HER KAN DU TESTE DIN VIDEN OM"
170 PRINT "BØRNEBØGER."
180 PRINT
190 PRINT "DET ER EN MULTIPLE-CHOICE QUIZ."
200 PRINT
210 PRINT "TAST 1, 2, 3 ELLER 4 EFTER SPØRGSMÅLS-"
220 PRINT "TEGNET."
230 PRINT
240 PRINT "HELD OG LYKKE!"
250 PRINT
260 PRINT "HVAD HEDDER KATTEN I 'PINOCCHIO' ?"
270 PRINT
280 PRINT "1=TIGER 2=CICERO 3=FIGARO 4=GUIPETTO ";
290 INPUT A
300 PRINT
310 IF A = 3 THEN 340
320 PRINT "DESVÆRRE....DEN HED FIGARO."
330 GOTO 360
340 PRINT "UDMÆRKET! RIGTIGT! HER ER ET TIL:"
350 R = R + 1
360 PRINT
370 PRINT "FRA HVIS HAVE STJÆLER SNURRE SNUP GULE-"
380 PRINT "RØDDER ?"
390 PRINT "        1=HR. JENSSENS  2=ELMER FJOTS"
400 PRINT "        3=CLEM JUDD'S  4=STROMBOLI'S ";
410 INPUT A
420 IF A = 2 THEN 460
430 PRINT
440 PRINT "ÆRGELIGT. DET ER FRA ELMER FJOTS HAVE."
450 GOTO 490
460 PRINT
470 PRINT "MEGET DYGTIGT!"
480 R = R + 1
490 PRINT
500 PRINT "HVAD HEDDER DOROTHY'S HUND I EVENTYRET"
510 PRINT "'TROLDMANDEWN FRA OZ' ?"
520 PRINT
530 PRINT "1=CICERO 2=TRIXIE 3=KING 4=TOTO ";
540 INPUT A
550 PRINT
560 IF A = 4 THEN 590
570 PRINT "LÆS BOGEN IGEN...HUNDEN HED TOTO."
580 GOTO 610
590 PRINT "JAAAH, DU ER ET LITTERATUR-GENI!"
600 R = R + 1
610 PRINT
620 PRINT "HVAD HED DEN SMUKKE MØ, DER SPISTE DET"
630 PRINT "FORGIFTEDE ÆBLE ?"
640 PRINT
650 PRINT "1=EVA 2=ASKEPOT 3=SNEHVIDE 4=WENDY ";
660 INPUT A
670 PRINT
680 IF A = 3 THEN 710
690 PRINT "HOLD DA OP, DET VAR DA SNEHVIDE!"
700 GOTO 730
710 PRINT "DU HAR VIST EN GOD HUKOMMELSE, HVA'!"
720 R = R + 1
730 PRINT
740 IF R = 4 THEN 800
750 IF R < 2 THEN 850
760 PRINT "IKKE DÆRLIGT, MEN DU ER NOK NØDT TIL AT"
770 PRINT "BRUGE LIDT MERE TID PÅ AT LÆSE NOGLE"
780 PRINT "AF BØRNEBOGSLITTERATURENS HOVEDVÆRKER."
790 GOTO 880
800 PRINT "WOW, DET ER ALLETIDES. DU KAN VIRKELIG"
810 PRINT "FINDE RUNDT I BØRNEVÆRSELSES-LEKTUREN."
820 PRINT "DEN NÆSTE QUIZ VIL HANDLE OM KINESISK"
830 PRINT "LITTERATUR I DET 2. ÅRHUNDREDE (HA-HA)."
840 GOTO 880
850 PRINT "BAH. DET VAR DEFINITIVT IKKE NOGET AT"
860 PRINT "PRALE MED. TILBAGE TIL BØRNEHAVEN MED"
870 PRINT "DIG, GAMLE VEN...."
880 END
```

Liv

Liv blev første gang beskrevet i »Scientific America« i oktober 1970 i en artikel af Martin Gardner. Selve spillet blev opfundet af John Conway fra Gonville and Caius College, University of Cambridge, England.

I den »manuelle« version af spillet vises cellerne ved hjælp af brikker på et stort skakbræt. Cellerne forøger sig og dør i én uendelighed ifølge nogle enkle arvelighedslove. De regler, som Conway opstillede, da han bestemte spillets arvelighedslove, er baseret på følgende:

1. Der må ikke findes et begyndelsesmønster, der kan bevise, at cellerne kan vokse ud over alle grænser.
2. Der skal findes et begyndelsesmønster, som tilsyneladende vokser uden begrænsninger.
3. Der skal findes begyndelsesmønstre, der vokser og forandrer sig i lang tid, før cellerne standser. Dette skal skyldes én af følgende tre grunde:
 - a) de forsvinder helt på grund af overbefolkning, eller fordi de bliver for få.
 - b) de standser i et stabilt system, hvor de forbliver uændrede, eller
 - c) de går ind i en periodisk fase, hvor de hele tiden gentager en uendelig cyklus med to eller flere perioder.

Sammenfattende kan man sige, at reglerne skal være sådan, at cellernes opførsel skal være relativt svær at forudsige. Conways arvelighedsregler er elegante og enkle. Læg mærke til, at hver celle på brættet (som antages at være uendeligt stort) har otte naboceller, fire vinkelrette og fire diagonale.

Arvelighedsreglerne er:

1. Overlevelse. Hver celle, der har to eller tre naboceller, overlever til næste generation.

2. Død. Celler med fire eller flere naboceller dør (fjernes) på grund af overbefolkning. Celler med kun én nabocelle dør på grund af isolation.
3. Fødsel. I en tom plads med præcis 3 naboceller – hverken mere eller mindre – fødes der en ny celle i næste generation.

Det er vigtigt at forstå, at alle fødsler og dødsfald sker samtidigt. Tilsammen udgør de en enkelt generation, eller, som vi kalder det, et »træk« i begyndelsesmønstrets livshistorie.

Du vil opdage, at alle cellerne hele tiden forandrer sig med usædvanlige og nogle gange smukke og uventede mønstre. I nogle tilfælde vil hele cellesamfundet uddø fuldstændigt (alle celler forsvinder), selv om dette sker efter mange generationer. De fleste begyndelsesmønstre når enten stabile værdier – Conway kalder dem for »stilleben« – som ikke forandrer sig, eller man får mønstre, som svinger frem og tilbage for evigt.

Asymmetriske begyndelsesmønstre tenderer mod at blive symmetriske. Når denne symmetri er opnået, kan den forsvinde igen eller forøges yderligere.

Conway brugte en DEC PDP-7 med en grafisk skærm til at iagttage de celler, der lever i lang tid. Du vil sandsynligvis bedre kunne lide at betragte cellerne på en skærm end at se dem på en udskrift.

Eftersom MITS 8K BASIC ikke har instruktionen LINE INPUT, der tillader indtastning af blanktegn i begyndelsesmønstret, skal der i stedet indtastes et punktum (.) som første tegn, når begyndelsesmønstret indtastes. Dette punktum vil af BASIC blive omdannet til et blanktegn. Hvis du skriver »SLUT« angiver du, at du ønsker at stoppe indtastningen af begyndelsesmønstret. Se kørselseksemplet.

Clark Baker fra projektet DELTA skrev originalversionen af programmet. Dette blev yderligere modificeret af Steve North fra Creative Computing.

```

MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY
-----
                      LIV
-----

INDTAST DIT MØNSTER:
?_ ***
?_ * *
?_* *
?SLUT

GENERATION: 0      POPULATION: 7
                    ***
                    * *
                    * *

GENERATION: 1      POPULATION: 8
                    *
                    ***
                    ** **

GENERATION: 2      POPULATION: 9
                    ***
                    * *
                    ** **
  
```

```

GENERATION: 3      POPULATION: 10
                    *
                    ***
                    * *
                    ** **

GENERATION: 4      POPULATION: 12
                    ***
                    ***
                    * *
                    ** **

GENERATION: 5      POPULATION: 11
                    *
                    * *
                    * *
                    * *
                    ** **

GENERATION: 6      POPULATION: 16
                    *
                    ***
                    ** **
                    ** **
                    ** **
  
```

GENERATION: 7

POPULATION: 15

```

***
* *
*   *
*   *
*** **

```

GENERATION: 14

POPULATION: 39

```

*
*
***
*   *
*   *
**  *  **
** *** **
* *** ** *
** * * **
**  **

```

GENERATION: 8

POPULATION: 23

```

*
***
*****
** **
** **
** **
* *

```

GENERATION: 15

POPULATION: 37

```

***
*
*** **
* * *** *
*
** ** ** **
* * * *
*** **

```

GENERATION: 9

POPULATION: 20

```

***
* *
*   *
* * *
*   *
* * **
** **

```

GENERATION: 16

POPULATION: 48

```

*
***
***
*   *
*** * ***
* ** * ** *
** ** * ** **
*** **
* *
** **
* *

```

GENERATION: 10

POPULATION: 27

```

*
***
*****
*   *
** **
** **
* * * *
** **

```

GENERATION: 17

POPULATION: 46

```

***
* *
*** * * ***
** * *** * **
* * * * *
** ** ** **
* *
*** ***
** **

```

GENERATION: 11

POPULATION: 30

```

***
* *
* *
** *** **
* * * * *
*   *
** * * **
** **

```

GENERATION: 18

POPULATION: 54

```

*
*
* *
*   *
*   *
** *** **
** * * **
* * * * *
** *** ** **
* * * * *
* * * *
* * * *
* * * *

```

GENERATION: 12

POPULATION: 42

```

*
***
* * *
* * *
** *** **
* * * * *
** * * **
** * * **
*** **

```

GENERATION: 13

POPULATION: 30

```

***
* *
* *
*   *
** * * **
* * * * *
* * * * *
** * * **
* *

```

GENERATION: 19

POPULATION: 56

```

***
*
** ***** **
** * * * * **
* * * * *
* * * * *
** * * **
** ** ** **

```


GENERATION: 20

POPULATION: 61

```
      *
     ***
    ****
   *****
  ** * * * * **
 * * * * * * *
* * * * * * *
  * * * * *
   **          **
  *****      *****
```

```
100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY      "
110 PRINT "-----"
120 PRINT "                      LIV"
130 PRINT "-----"
140 PRINT
150 PRINT "INDTAST DIT MØNSTER: "
160 X1 = 1
170 Y1 = 1
180 X2 = 24
190 Y2 = 70
200 DIM A(24,70),B$(24)
210 C = 1
220 INPUT B$(C)
230 IF B$(C) = "SLUT" THEN B$(C) = "": GOTO 270
240 IF LEFT$(B$(C),1) = "." THEN B$(C) = " " + RIGHT$(B$(C), LEN
(B$(C)) - 1)
250 C = C + 1
260 GOTO 220
270 C = C - 1
280 L = 0
290 FOR X = 1 TO C - 1
300 IF LEN (B$(X)) > L THEN L = LEN (B$(X))
310 NEXT X
320 X1 = 11 - C / 2
330 Y1 = 33 - L / 2
340 FOR X = 1 TO C
350 FOR Y = 1 TO LEN (B$(X))
360 IF MID$(B$(X),Y,1) < > " " THEN A(X1 + X,Y1 + Y) = 1:P = P + 1
370 NEXT Y
380 NEXT X
390 PRINT
400 PRINT
410 PRINT "GENERATION:";G"          POPULATION:";P;
420 IF I9 < > 0 THEN PRINT "UGYLDIG";
430 X3 = 24
440 Y3 = 70
450 X4 = 1
460 Y4 = 1
470 P = 0
480 G = G + 1
490 FOR X = 1 TO X1 - 1
500 PRINT
510 NEXT X
520 FOR X = X1 TO X2
530 PRINT
540 FOR Y = Y1 TO Y2
550 IF A(X,Y) = 2 THEN A(X,Y) = 0: GOTO 630
560 IF A(X,Y) = 3 THEN A(X,Y) = 1: GOTO 580
570 IF A(X,Y) < > 1 THEN 630
580 PRINT TAB( Y);"*";
590 IF X < X3 THEN X3 = X
600 IF X > X4 THEN X4 = X
610 IF Y < Y3 THEN Y3 = Y
620 IF Y > Y4 THEN Y4 = Y
630 NEXT Y
640 NEXT X
650 FOR X = X2 + 1 TO 24
660 PRINT
670 NEXT X
680 X1 = X3
690 X2 = X4
700 Y1 = Y3
710 Y2 = Y4
720 IF X1 < 3 THEN X1 = 3:I9 = - 1
730 IF X2 > 22 THEN X2 = 22:I9 = - 1
740 IF Y1 < 3 THEN Y1 = 3:I9 = - 1
750 IF Y2 > 68 THEN Y2 = 68:I9 = - 1
760 P = 0
770 FOR X = X1 - 1 TO X2 + 1
780 FOR Y = Y1 - 1 TO Y2 + 1
790 C = 0
800 FOR I = X - 1 TO X + 1
810 FOR J = Y - 1 TO Y + 1
820 IF A(I,J) = 1 OR A(I,J) = 2 THEN C = C + 1
830 NEXT J
840 NEXT I
850 IF A(X,Y) = 0 THEN 890
860 IF C < 3 OR C > 4 THEN A(X,Y) = 2: GOTO 880
870 P = P + 1
880 GOTO 900
890 IF C = 3 THEN A(X,Y) = 3:P = P + 1
900 NEXT Y
910 NEXT X
920 X1 = X1 - 1
930 Y1 = Y1 - 1
940 X2 = X2 + 1
950 Y2 = Y2 + 1
960 GOTO 410
970 END
```

Liv for to

Liv for to er baseret på Conway's spil Liv. Du er nødt til at kende reglerne for Liv, inden du begynder at spille Liv for to.

Spillet er beregnet for to deltagere og spilles på et bræt med 5 gange 5 felter. Hver spiller har et symbol eller en markering, som svarer til hans brikker i spillet. Levende celler, der tilhører spiller 1 vises med (*), mens de for spiller 2 vises med symbolet (+).

Tegnene '*' og '+' betragtes som ens, undtagen når det skal besluttes, om en levende celle skal skabes. En tom plads, der er omgivet af to '+' og en '*' nabocelle, giver en '+', dvs. den celle, der fødes, kommer til at tilhøre spilleren med flest celler, omkring det sted hvor en ny celle kan fødes. F.eks.:

	1	2	3	4	5
1					
2			*		
3				#	
4			#		
5					

En ny celle vil blive skabt på (3,3), og det bliver en '+' fordi der er to '+', men kun en '*'. Dette kommer derefter til at se således ud:

	1	2	3	4	5
1					
2					
3			#	#	
4					
5					

I det første træk placerer hver spiller 3 levende celler på brættet, idet koordinaterne for disse indtastes. Hvis den samme celle vælges af de to spillere, lades denne tom.

Brættet justeres derefter til at vise næste cellegeneration.

Ved hvert af de følgende træk placerer hver af spillerne en celle på brættet i et forsøg på at tilintetgøre alle modstanderens celler. Brættet udskrives for næste celle-generation, så snart begge spillere har indtastet deres træk.

Spillet fortsætter, indtil en af spillerne ikke har flere levende celler. Datamaten tegner derefter brættet og skriver, hvem der har vundet.

Ideen til spillet og ovenstående spillebeskrivelse kommer fra Brian Wyvill fra Bradford University i Yorkshire, England.

LIV FOR TO

LIV FOR TO

SPILLER 1 - 3 LEVENDE BRIKKER.

X,Y ?3,3
X,Y ?3,5
X,Y ?3,2

SPILLER 2 - 3 LEVENDE BRIKKER.

X,Y ?5,1
X,Y ?4,2
X,Y ?5,3

0	1	2	3	4	5	0
1						+ 1
2			*	+		2
3			*		+	3
4						4
5			*			5
0	1	2	3	4	5	0

0	1	2	3	4	5	0
1				+		1
2			*		+	2
3			*			3
4				*		4
5						5
0	1	2	3	4	5	0

SPILLER 1 X,Y ?4,2

SPILLER 2 X,Y ?4,3

0	1	2	3	4	5	0
1			*	+	+	1
2					+	2
3						3
4			*	*		4
5						5
0	1	2	3	4	5	0

SPILLER 1 X,Y ?4,2

SPILLER 2 X,Y ?2,1

0	1	2	3	4	5	0
1			*		+	1
2					+	2
3			*	*		3
4						4
5						5
0	1	2	3	4	5	0

SPILLER 1 X,Y ?4,2

SPILLER 2 X,Y ?4,1

0	1	2	3	4	5	0
1			*		+	1
2					+	2
3			*			3
4						4
5						5
0	1	2	3	4	5	0

SPILLER 1 X,Y ?4,1

SPILLER 2 X,Y ?5,2

0	1	2	3	4	5	0
1			*	+		1
2					+	2
3						3
4						4
5						5
0	1	2	3	4	5	0

SPILLER 1 X,Y ?5,3

SPILLER 2 X,Y ?3,1

0	1	2	3	4	5	0
1				*	+	1
2					+	2
3						3
4						4
5						5
0	1	2	3	4	5	0

SPILLER 1 X,Y ?4,2

SPILLER 2 X,Y ?5,3

0	1	2	3	4	5	0
1				*	+	1
2					+	2
3				+	+	3
4						4
5						5
0	1	2	3	4	5	0

SPILLER 1 X,Y ?5,2

SPILLER 2 X,Y ?3,1

0	1	2	3	4	5	0
1				*	+	1
2				+		2
3				+	+	3
4						4
5						5
0	1	2	3	4	5	0

SPILLER 1 X,Y ?3,3

SPILLER 2 X,Y ?4,2

0	1	2	3	4	5	0
1			+	*	+	1
2						2
3			*		+	3
4				+		4
5						5
0	1	2	3	4	5	0

SPILLER 1 X,Y ?4,2

SPILLER 2 X,Y ?2,2

0	1	2	3	4	5	0
1			+	*	+	1
2			+			2
3			*		+	3
4				+		4
5						5
0	1	2	3	4	5	0

SPILLER 1 X,Y ?3,2

SPILLER 2 X,Y ?4,2

0	1	2	3	4	5	0
1			+		+	1
2			+			2
3			*		+	3
4				+		4
5						5
0	1	2	3	4	5	0

SPILLER 1 X,Y ?5,2

SPILLER 2 X,Y ?2,4

```

0 1 2 3 4 5 0
1                1
2 + + + + * 2
3 + *      + + 3
4          +    4
5                5
0 1 2 3 4 5 0

```

SPILLER 1 X,Y ?3,3

SPILLER 2 X,Y ?5,1

```

0 1 2 3 4 5 0
1  + +      + 1
2 +          2
3 +          + 3
4          +  4
5                5
0 1 2 3 4 5 0

```

SPILLER 2 VINDER.

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          LIV FOR TO"
130 PRINT "-----"
140 PRINT
150 DIM N(6,6),K(18),A(16),X(2),Y(2)
160 DATA 3,102,103,120,130,121,112,111,12
170 DATA 21,30,1020,1030,1011,1021,1003,1002,1012
180 FOR M = 1 TO 18
190 READ K(M)
200 NEXT M
210 DATA -1,0,1,0,0,-1,0,1,-1,-1,1,-1,-1,1,1,1
220 FOR O1 = 1 TO 16
230 READ A(O1)
240 NEXT O1
250 GOTO 730
260 FOR J = 1 TO 5
270 FOR K = 1 TO 5
280 IF N(J,K) > 99 THEN GOSUB 500
290 NEXT K
300 NEXT J
310 K = 0
320 M2 = 0
330 M3 = 0
340 FOR J = 0 TO 6
350 PRINT
360 FOR K = 0 TO 6
370 IF J < > 0 THEN IF J < > 6 THEN 410
380 IF K = 6 THEN PRINT 0;: GOTO 470
390 PRINT K;
400 GOTO 460
410 IF K < > 0 THEN IF K < > 6 THEN 450
420 IF J = 6 THEN PRINT 0;: GOTO 480
430 PRINT J;

```

```

440 GOTO 460
450 GOSUB 560
460 NEXT K
470 NEXT J
480 PRINT
490 RETURN
500 B = 1
510 IF N(J,K) > 999 THEN B = 10
520 FOR O1 = 1 TO 15 STEP 2
530 N(J + A(O1),K + A(O1 + 1)) = N(J + A(O1),K + A(O1
+ 1)) + B
540 NEXT O1
550 RETURN
560 IF N(J,K) < 3 THEN 700
570 FOR O1 = 1 TO 18
580 IF N(J,K) = K(O1) THEN 610
590 NEXT O1
600 GOTO 700
610 IF O1 > 9 THEN 660
620 N(J,K) = 100
630 M2 = M2 + 1
640 PRINT " * ";
650 RETURN
660 N(J,K) = 1000
670 M3 = M3 + 1
680 PRINT " + ";
690 RETURN
700 N(J,K) = 0
710 PRINT " ";
720 RETURN
730 PRINT "          LIV FOR TO"
740 M2 = 0
750 M3 = 0
760 FOR J = 1 TO 5
770 FOR K = 1 TO 5
780 N(J,K) = 0
790 NEXT K
800 NEXT J
810 FOR B = 1 TO 2
820 P1 = 3
830 IF B = 2 THEN P1 = 30
840 PRINT
850 PRINT "SPILLER";B;" - 3 LEVENDE BRIKKER."
860 PRINT
870 FOR K1 = 1 TO 3
880 GOSUB 1140
890 N(X(B),Y(B)) = P1
900 NEXT K1
910 NEXT B
920 GOSUB 310
930 PRINT
940 GOSUB 260
950 IF M2 = 0 THEN IF M3 = 0 THEN 990
960 IF M3 = 0 THEN B = 1: GOTO 1020
970 IF M2 = 0 THEN B = 2: GOTO 1020
980 GOTO 1050
990 PRINT
1000 PRINT "UAFGJORT."
1010 GOTO 1300
1020 PRINT
1030 PRINT "SPILLER";B;"VINDER."
1040 GOTO 1300
1050 FOR B = 1 TO 2
1060 PRINT
1070 PRINT "SPILLER";B;
1080 GOSUB 1140
1090 IF B = 99 THEN 930
1100 NEXT B
1110 N(X(1),Y(1)) = 100
1120 N(X(2),Y(2)) = 1000
1130 GOTO 930
1140 PRINT "X,Y ";
1150 INPUT Y(B),X(B)
1160 IF X(B) < = 5 THEN IF X(B) > 0 THEN 1180
1170 GOTO 1280
1180 IF Y(B) < = 5 THEN IF Y(B) > 0 THEN 1200
1190 GOTO 1280
1200 IF N(X(B),Y(B)) < > 0 THEN 1280
1210 IF B = 1 THEN RETURN
1220 IF X(1) = X(2) THEN IF Y(1) = Y(2) THEN 1240
1230 RETURN
1240 PRINT "SAMME KOORDINAT. ANGIV 0."
1250 N(X(B) + 1,Y(B) + 1) = 0
1260 B = 99
1270 RETURN
1280 PRINT "ULOVLIGE KOORDINATER, PRØV IGEN."
1290 GOTO 1140
1300 END

```


Mastermind

I marts-april nummeret af Creative Computing, årgang 1976, offentliggjorde vi en datamatversion af logikspillet Mastermind. Mastermind spilles af to personer, den ene kaldes koderen, den anden kodeløseren. Når spillet starter, laver koderen en kode, som består af en kombination af farvede stifter. Koden holdes skjult for kodeløseren. Derefter forsøger kodeløseren at løse koden ved at placere stifter i forskellige kombinationer på brættet, en ad gangen. Når dette er gjort, giver koderen kodeløseren nogle fingerpeg om, hvor nær gættet var på den rigtige kombination. For hver stift, der er på rigtig plads i koden, og som har den rigtige farve, vises en sort stift. For hver stift i kodeløserens kode, der har rigtig farve, men som ikke står på rigtig plads, vises en hvid stift. Bemærk, at disse sorte og hvide stifter ikke fortæller, hvilke stifter i kodeløserens kode der var rigtige, men kun at en eller flere af stifterne var i orden. Hvis koden f.eks. var:

GUL, RØD, RØD, GRØN

og jeg gættede på:

RØD, RØD, GUL, SORT

Så ville jeg få to hvide og en sort stift for dette gæt. Jeg kan derimod ikke vide (medmindre jeg sammenligner med tidligere gæt), hvilke af stifterne i mit gæt der har rigtig farve, og hvilken der var rigtigt placeret.

Mange programmører har skrevet programmer, hvor datamaten spiller en passiv rolle, dvs. som koder, mens mennesket er kodeløser. Dette er forholdsvis simpelt, den rigtige udfordring ligger dog i at skrive et program, hvor datamaten deltager som kodeløser.

Egentlig er det ikke særlig svært at få datamaten til at beregne den rigtige kombination. Forestil dig, at man f.eks. lavede en liste over alle de mulige koder. Til at begynde med vælger man en tilfældig kode fra denne liste. Derefter stryger man de umulige kombinationer fra listen, efterhånden som man får flere og flere ledetråde. Hvis man f.eks. gætter Rød, Rød, Grøn, Grøn og slet ikke får en stift, ja, så ved man, at kombinationen hverken indeholder Rød eller Grøn. Alle kombinationer i listen, der indeholder disse farver, kan derfor stryges. Denne proces fortsætter, indtil den rigtige løsning er fundet, eller indtil der ikke er flere alternativer tilbage på listen (og så ved man, at koderen har begået en fejl, da han uddelte sorte og hvide stifter!).

Bemærk, at vi i dette program egentlig aldrig laver en sådan liste, men at vi i stedet holder rede på, hvilke gæt (i rækkefølge) der kan være rigtige. Med dette system kan vi let sige, at den 523. kombination er den rigtige. Men når vi skal finde ud af, hvad denne kombination består i, så må vi tælle helt fra den første kombination til den 523'ende. Dette problem kan forenkles ved at omregne fra talbase 10 til en talbase, der svarer til det antal farver, som spillet opererer med. Herefter skal værdierne, der bruges i MID\$-funktionen, laves om så denne ikke forsøger at finde det 0'te tegn i strengen. Vi har prøvet et program, hvor listen med samtlige kombinationer blev indlæst som indicerede streng-variabler. Dette program var betydeligt hurtigere end nærværende, men optog frygtelig megen plads i datamatens hukommelse.

I spillets begyndelse angiver du, hvor mange farver og hvor mange pladser (hvilket direkte har betydning for antallet af kombinationer) du ønsker, at spillet kan bruge. Du indtaster også, hvor mange omgange du vil spille. Når du er kodeløser, kan du, hvis du skriver »BRÆT«, få en liste over alle dine tidligere gæt og

ledetråde. Hvis du skriver »SLUT«, afsluttes spillet. Bemærk, at denne version anvender indicerede strengvariabler, men det er udelukkende af hensyn til bekvemligheden, og programmet kan let konverteres til de BASIC-varianter, der ikke tillader streng-håndtering. Dog kræves, at MID\$-instruktionen findes. Denne konvertering kan lade sig gøre, fordi de indicerede streng-variabler kun har én dimension og aldrig indeholder flere tegn end antallet af positioner.

MORRISTOWN/CREATIVE COMPUTING/N. JERSEY

MASTER MIND

ANTAL FARVER ?4

ANTAL POSITIONER ?4

ANTAL RUNDER ?1

TOTALE ANTAL MULIGHEDER = 256

BOGSTAV	FARVE
=====	=====
S	SORT
H	HVID
R	RØD
B	BLÅ

OMGANG NUMMER 1 ----

GÆT MIN KOMBINATION.

TRÆK NR. 1 GÆT ?SHRB
DU FÅR 1 SORTE OG 2 HVIDE.

TRÆK NR. 2 GÆT ?RHBB
DU FÅR 1 SORTE OG 2 HVIDE.

TRÆK NR. 3 GÆT ?RRBS
DU FÅR 1 SORTE OG 2 HVIDE.

TRÆK NR. 4 GÆT ?RBBS
DU FÅR 2 SORTE OG 2 HVIDE.

TRÆK NR. 5 GÆT ?SBBR

DU GÆTTEDE DET PR 5 TRÆK.

RESULTAT

DATAMATEN 0
MENNESKET 5

NU GÆTTER JEG. TÆNK PR EN KOMBINATION.
TAST 'RETURN' NÅR DU ER PARAT ?

MIT GÆT ER: HHSH - (SORTE, HVIDE) ?1,0
MIT GÆT ER: RRRH - (SORTE, HVIDE) ?2,1
MIT GÆT ER: RRSR - (SORTE, HVIDE) ?1,1
MIT GÆT ER: RHRB - (SORTE, HVIDE) ?2,2
MIT GÆT ER: HRRB - (SORTE, HVIDE) ?4,0

JEG RAMTE RIGTIGT PR 5 FORSØG.

RESULTAT

DATAMATEN 5
MENNESKET 5

SPILET ER FORBI

SLUTRESULTAT

DATAMATEN 5
MENNESKET 5

 MASTER MIND

ANTAL FARVER ?5

ANTAL POSITIONER ?4

ANTAL RUNDER ?1

TOTALE ANTAL MULIGHEDER = 625

BOGSTAV	FARVE
=====	=====
S	SORT
H	HVID
R	RØD
B	BLÅ
O	ORANGE

OMGANG NUMMER 1 ----

GÆT MIN KOMBINATION.

TRÆK NR. 1 GÆT ?SOHR
 DU FÅR 0 SORTE OG 2 HVIDE.

TRÆK NR. 2 GÆT ?SOBR
 DU FÅR 1 SORTE OG 1 HVIDE.

TRÆK NR. 3 GÆT ?RRBH
 DU FÅR 2 SORTE OG 1 HVIDE.

TRÆK NR. 4 GÆT ?RHBB
 DU FÅR 2 SORTE OG 2 HVIDE.

TRÆK NR. 5 GÆT ?HRBB

DU GÆTTEDE DET PÅ 5 TRÆK.

RESULTAT

 DATAMATEN 0
 MENNESKET 5

NU GÆTTER JEG. TÆNK PÅ EN KOMBINATION.
 TAST 'RETURN' NÅR DU ER PARAT ?

MIT GÆT ER: SOHH - (SORTE, HVIDE) ?2,0
 MIT GÆT ER: SBRH - (SORTE, HVIDE) ?1,0
 MIT GÆT ER: SOSO - (SORTE, HVIDE) ?2,2
 MIT GÆT ER: SOOS - (SORTE, HVIDE) ?4,0

JEG RAMTE RIGTIGT PÅ 4 FORSØG.

RESULTAT

 DATAMATEN 4
 MENNESKET 5

SPILLET ER FORBI

SLUTRESULTAT

 DATAMATEN 4
 MENNESKET 5

```

100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "          MASTER MIND"
130 PRINT "-----"
140 PRINT
150 PRINT "ANTAL FARVER ";
160 INPUT C9
170 PRINT
180 IF C9 > 8 THEN PRINT "IKKE FLERE END 8
FARVER!": GOTO 140
190 PRINT "ANTAL POSITIONER ";
200 INPUT P9
210 PRINT
220 PRINT "ANTAL RUNDER ";
230 INPUT R9
240 P = C9 ^ P9

```

```

250 PRINT
260 PRINT "TOTALE ANTAL MULIGHEDER =";P
270 H = 0
280 C = 0
290 DIM Q(P9),S(10,2),S$(10),A$(P9),G$(P9),I(P),
H$(P9)
300 L$ = "SHRBOGLP"
310 PRINT
320 PRINT
330 PRINT "BOGSTAV      FARVE"
340 PRINT "=====      ====="
350 FOR X = 1 TO C9
360 READ X$
370 PRINT " "; MID$(L$,X,1);"          ";X$
380 NEXT X
390 PRINT
400 FOR R = 1 TO R9
410 PRINT
420 PRINT "OMGANG NUMMER";R;"----"
430 PRINT
440 PRINT "GÆT MIN KOMBINATION."
450 REM Lav en kombination
460 A = INT (P * RND (1) + 1)
470 GOSUB 1830
480 FOR X = 1 TO A
490 GOSUB 1900
500 NEXT X
510 FOR M = 1 TO 10
520 PRINT
530 PRINT "TRÆK NR.";M;"GÆT ";
540 INPUT X$
550 IF X$ = "BRÆT" THEN 1590
560 IF X$ = "SLUT" THEN 1710
570 IF LEN (X$) < > P9 THEN PRINT "FORKERT ANTAL
POSITIONER.": GOTO 520
580 REM Indfør X$ i G$(1-P9)
590 FOR X = 1 TO P9
600 FOR Y = 1 TO C9
610 IF MID$(X$,X,1) = MID$(L$,Y,1) THEN 650
620 NEXT Y
630 PRINT """; MID$(X$,X,1)"" FINDES IKKE."
640 GOTO 520
650 G$(X) = MID$(X$,X,1)
660 NEXT X
670 REM Q(1-P9) forvandles til A$(1-P9)
680 REM Hvor A$(1-P9) er det virkelige gæt
690 GOSUB 2050
700 REM Udregn antallet af sorte og hvide
710 GOSUB 2120
720 IF B = P9 THEN 880
730 REM Fortæl det til den der spiller
740 PRINT "DU FÅR";B;"SORTE OG";W;"HVIDE."
750 REM Gem dette til senere udskrift
760 S$(M) = X$
770 S(M,1) = B
780 S(M,2) = W
790 NEXT M
800 PRINT : PRINT "DU LØB TØR FOR GÆT. DU FÅR IKKE
FLERE."
810 GOTO 900
820 GOSUB 2050
830 PRINT "DEN HEMMELIGE KOMBINATION VAR: ";
840 FOR X = 1 TO P9
850 PRINT A$(X);
860 NEXT X
870 PRINT
880 PRINT
890 PRINT "DU GÆTTEDE DET PÅ";M;"TRÆK."
900 H = H + M
910 PRINT
920 GOSUB 2370
930 REM
940 REM Datamaten gætter
950 REM
960 FOR X = 1 TO P
970 I(X) = 1
980 NEXT X
990 PRINT "NU GÆTTER JEG. TÆNK PÅ EN KOMBINATION."
1000 PRINT "TAST 'RETURN' NÅR DU ER PARAT ";
1010 INPUT X$
1020 PRINT
1030 FOR M = 1 TO 10
1040 GOSUB 1830
1050 REM Lav et gæt
1060 G = INT (P * RND (1) + 1)
1070 IF I(G) = 1 THEN 1210
1080 FOR X = G TO P
1090 IF I(X) = 1 THEN 1200

```

```

1100 NEXT X
1110 FOR X = 1 TO G
1120 IF I(X) = 1 THEN 1200
1130 NEXT X
1140 PRINT
1150 PRINT "IDIOT, DU HAR GIVET MIG FEJLAGTIGE OG"
1160 PRINT "VILDLEDENDE OPLYSNINGER. LAD OS FORSØGE"
1170 PRINT "EN GANG TIL.....VÆR MERE OMHYGGELIG!"
1180 PRINT
1190 GOTO 930
1200 G = X
1210 REM Omregn gæt nr. G til G$
1220 FOR X = 1 TO G
1230 GOSUB 1900
1240 NEXT X
1250 GOSUB 2530
1260 PRINT "MIT GÆT ER: ";
1270 FOR X = 1 TO P9
1280 PRINT H$(X);
1290 NEXT X
1300 PRINT " - (SORTE, HVIDE) ";
1310 INPUT B1,W1
1320 IF B1 = P9 THEN 1470
1330 GOSUB 1830
1340 FOR X = 1 TO P
1350 GOSUB 1900
1360 IF I(X) = 0 THEN 1410
1370 GOSUB 2600
1380 GOSUB 2050
1390 GOSUB 2120
1400 IF B1 < > B OR W1 < > W THEN I(X) = 0
1410 NEXT X
1420 NEXT M
1430 PRINT "JEG HAR BRUGT ALLE MINE OMGANGE!"
1440 PRINT "DET SER UD TIL, AT MIN PROCESSOR HAR"
1450 PRINT "EN DÆRLIG DAG!"
1460 GOTO 1500
1470 PRINT
1480 PRINT "JEG RAMTE RIGTIGT PÅ";M;"FORSØG."
1490 PRINT
1500 C = C + M
1510 GOSUB 2370
1520 NEXT R
1530 PRINT "SPILLET ER FORBI"
1540 PRINT
1550 PRINT "SLUTRESULTAT"
1560 PRINT "-----"
1570 GOSUB 2420
1580 GOTO 2700
1590 REM
1600 REM Rutine der udskriver brættet
1610 REM
1620 PRINT
1630 PRINT "BRÆTTET"
1640 PRINT "TRÆK GÆT          SORT HVID"
1650 FOR Z = 1 TO M - 1
1660 PRINT Z; TAB( 6);S$(Z); TAB( 20);
1670 PRINT S(Z,1); TAB( 26);S(Z,2)
1680 NEXT Z
1690 PRINT
1700 GOTO 520
1710 REM
1720 REM Afslutningsrutine
1730 REM
1740 PRINT "BANGEBUKS!"
1750 PRINT "MIN KOMBINATION VAR: ";
1760 GOSUB 2050
1770 FOR X = 1 TO P9
1780 PRINT A$(X);
1790 NEXT X
1800 PRINT
1810 PRINT "FARVEL MED DIG."
1820 GOTO 2700
1830 REM
1840 REM Sæt Q(1-P9) til nuller
1850 REM
1860 FOR S = 1 TO P9
1870 Q(S) = 0
1880 NEXT S
1890 RETURN
1900 REM
1910 REM Justering af Q(1-P9)
1920 REM
1930 IF Q(1) > 0 THEN 1990
1940 REM Hvis lig med 0, juster til 1
1950 FOR S = 1 TO P9
1960 Q(S) = 1
1970 NEXT S

1980 RETURN
1990 Q = 1
2000 Q(Q) = Q(Q) + 1
2010 IF Q(Q) < = C9 THEN RETURN
2020 Q(Q) = 1
2030 Q = Q + 1
2040 GOTO 2000
2050 REM
2060 REM Forvandling af Q(1-P9) til A$(1-P9)
2070 REM
2080 FOR S = 1 TO P9
2090 A$(S) = MID$( L$,Q(S),1)
2100 NEXT S
2110 RETURN
2120 REM
2130 REM Find antal sorte (B) og hvide (W)
2140 REM Bland G$ og A$ samtidig
2150 REM
2160 B = 0
2170 W = 0
2180 F = 0
2190 FOR S = 1 TO P9
2200 IF G$(S) < > A$(S) THEN 2260
2210 B = B + 1
2220 G$(S) = CHR$( F)
2230 A$(S) = CHR$( F + 1)
2240 F = F + 2
2250 GOTO 2350
2260 FOR T = 1 TO P9
2270 IF G$(S) < > A$(T) THEN 2340
2280 IF G$(T) = A$(T) THEN 2340
2290 W = W + 1
2300 A$(T) = CHR$( F)
2310 G$(S) = CHR$( F + 1)
2320 F = F + 2
2330 GOTO 2350
2340 NEXT T
2350 NEXT S
2360 RETURN
2370 REM
2380 REM Udskriv resultatet
2390 REM
2400 PRINT "RESULTAT"
2410 PRINT "-----"
2420 PRINT "DATAMATEN";C
2430 PRINT "HENNESKET";H
2440 PRINT
2450 RETURN
2460 REM
2470 REM Forvandling af Q(1-P9) til G$(1-P9)
2480 REM
2490 FOR S = 1 TO P9
2500 G$(S) = MID$( L$,Q(S),1)
2510 NEXT S
2520 RETURN
2530 REM
2540 REM Forvandling af Q(1-P9) til H$(1-P9)
2550 REM
2560 FOR S = 1 TO P9
2570 H$(S) = MID$( L$,Q(S),1)
2580 NEXT S
2590 RETURN
2600 REM
2610 REM Kopier H$ til G$
2620 REM
2630 FOR S = 1 TO P9
2640 G$(S) = H$(S)
2650 NEXT S
2660 RETURN
2670 REM Programdata for farvenavne
2680 DATA "SORT","HVID","RØD","BLÅ","ORANGE","GUL",
"LILLA","PURPUR"
2690 REM Nu er det desværre tid til at slutte
2700 END

```

Matematisk terning

Dette program giver visuel øvelse i at lægge sammen ved hjælp af terninger. Programmet udskriver ingen tekst. Programmet kan anvendes, når børn skal starte med at lære matematik, og da svaret kan findes ved at tælle øjnene på terningerne, kan barnet få en fornemmelse af visse grundlæggende matematiske kendsgerninger.

Programmet blev oprindeligt skrevet af Jim Gerrish, der er lærer ved Bernice A. Ray School i Hanover, New Hampshire.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
-----
MATEMATISK TERNING
-----
```

```
DETTE PROGRAM TEGNER BILLEDER AF TO
TERNINGER PÅ SKÆRMEN.
```

```
NÅR DE TO TERNINGER OG ET LIGHEDSTEGN
EFTERFULGT AF ET SPØRGSMÅLSTEGN ER
BLEVET UDSKREVET, SKAL DU INDTASTE TER-
NINGERNES TOTALE ANTAL ØJNE OG RETURN.
```

```
NÅR DU ØNSKER AT STANDSE PROGRAMMET,
KAN DU BRUGE 'CONTROL-C' SOM INDTAST.
```

```
-----
! * * !
! * * !
! * * !
-----
```

+

```
-----
! * * !
! * * !
! * * !
-----
```

= ?12

RIGTIGT!!!

TERNINGERNE ER KASTET ... IGEN.

```
-----
! * !
! * !
! * !
-----
```

+

```
-----
! * !
! * !
! * !
-----
```

= ?4

NEJ, TÆL ØJNENE OG SVAR IGEN.

= ?2

NEJ ENDNU EN GANG! DER ER 3 ØJNE IALT.

TERNINGERNE ER KASTET ... IGEN.

```
-----
! * !
! * !
! * !
-----
```

+

```
-----
! * * !
! * !
! * * !
-----
```

= ?8

RIGTIGT!!!

TERNINGERNE ER KASTET ... IGEN.

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"
110 PRINT "-----"
120 PRINT "MATEMATISK TERNING"
130 PRINT "-----"
140 PRINT
150 PRINT "DETTE PROGRAM TEGNER BILLEDER AF TO"
160 PRINT "TERNINGER PÅ SKÆRMEN."
170 PRINT
180 PRINT "NÅR DE TO TERNINGER OG ET LIGHEDSTEGN"
190 PRINT "EFTERFULGT AF ET SPØRGSMÅLSTEGN ER"
200 PRINT "BLEVET UDSKREVET, SKAL DU INDTASTE TER-"
210 PRINT "NINGERNES TOTALE ANTAL ØJNE OG RETURN."
220 PRINT
230 PRINT "NÅR DU ØNSKER AT STANDSE PROGRAMMET,"
240 PRINT "KAN DU BRUGE 'CONTROL-C' SOM INDTAST."
250 PRINT
260 PRINT
270 N = N + 1
280 D = INT (6 * RND (1) + 1)
290 PRINT "-----"
300 IF D = 1 THEN 370
310 IF D = 2 THEN 350
320 IF D = 3 THEN 350
330 PRINT "! * * !"
340 GOTO 380
350 PRINT "! * !"
360 GOTO 380
370 PRINT "! * * !"
380 IF D = 2 THEN 430
390 IF D = 4 THEN 430
400 IF D = 6 THEN 450
410 PRINT "! * * !"
420 GOTO 460
430 PRINT "! * !"
440 GOTO 460
450 PRINT "! * * !"
460 IF D = 1 THEN 530
470 IF D = 2 THEN 510
480 IF D = 3 THEN 510
490 PRINT "! * * !"
500 GOTO 540
510 PRINT "! * !"
520 GOTO 540
530 PRINT "! * * !"
540 PRINT "-----"
550 PRINT
560 IF N = 2 THEN 610
570 PRINT " + "
580 PRINT
590 A = D
600 GOTO 270
610 T = D + A
620 PRINT " = ";
630 INPUT T1
640 PRINT
650 IF T1 = T THEN 740
660 PRINT "NEJ, TÆL ØJNENE OG SVAR IGEN."
670 PRINT
680 PRINT " = ";
690 INPUT T2
700 PRINT
710 IF T2 = T THEN 740
720 PRINT "NEJ ENDNU EN GANG! DER ER";T;"ØJNE IALT."
730 GOTO 270
740 PRINT "RIGTIGT!!!"
750 PRINT
760 PRINT "TERNINGERNE ER KASTET ... IGEN."
770 PRINT
780 N = 0
790 GOTO 270
800 END
```


Mål

I dette program skyder du med et våben fra et rumskib i et tredimensionalt rum. Det rumskib – Stjerneskipet Enterprise – befinder sig i origo (0,0,0) i et tredimensionalt koordinatsystem (x,y,z). Du får at vide omtrent, hvor målet befinder sig i retvinklede koordinater, den omtrentlige vinkel fra x- og z-akserne i både grader og radianer og den omtrentlige afstand til målet.

Med udgangspunkt i disse oplysninger skal du forsøge at ramme målet. Et skud, der detonerer inden for en radius af 20 km fra målet, tilintetgør det. Efter hvert skud får du at vide, hvor dit skud eksploderede, samt en noget bedre angivelse af målets position. Heldigvis er dette bare en øvelse, og målet skyder ikke tilbage. Når du har øvet dig lidt, bør du kunne ramme målet på 3-4 skud. Det kan imidlertid godt tage lidt tid!

Forfatter til dette spil er H. David Crockett fra Fort Worth, Texas.

```
RRUN
  MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY
-----
                          MÅL
-----
```

DU ER VÅBEN-OFFICER PÅ RUMSKIBET ENTERPRISE, OG DU SKAL GENNEMGÅ DENNE TEST SÅ DINE OVERORDNEDE KAN UNDERSØGE, HVOR NØJAGTIGT DU KAN SKYDE I ET TRE-DIMENSIONALT RUM. DU VIL BLIVE INFORMERET OM X- OG Y-AKSERNES VINKLER (I RADIANTER), MENS MÅLETS POSITION ANGIVES I RETVINKLEDE KOORDINATER, SAMT EN CIRKA-VÆRDI FOR MÅLETS AFSTAND (I GRADER) FRA AKSERNE.

DU VIL DEREFTER FORTSÆTTE MED AT SKYDE PÅ MÅLET, INDTIL DET ER ØDELAGT.

HELD OG LYKKE!!

RADIANER FRA X-AKSEN = 4.63829356
FRA Z-AKSEN = 1.71346911

MÅLET ER SIDST IAGTTAGET PÅ FØLGENDE KOORDINATER:

X-KOORDINAT = -2196.35075
Y-KOORDINAT = -29587.928
Z-KOORDINAT = -4261.96417
ANSLÆT AFSTAND = 29970

INDTAST VINKELAFVIGELSE FRA X ?230
VINKELAFVIGELSE FRA Z ?110
AFSTAND ?66000

RADIANER FRA X-AKSEN = 4.01424183
FRA Z-AKSEN = 1.91985479

SKUDET RAMTE 37669.9934 KM. BAGVED MÅLET.
SKUDET RAMTE 17921.4404 KM. TIL HØJRE FOR MÅLET.
SKUDET RAMTE 18310.9071 KM. UNDER MÅLET.

SKUDETS ANSLÆDE POSITION:
X-KOORDINAT = -39866.3442
Y-KOORDINAT = -47509.3684
Z-KOORDINAT = -22572.8713
AFSTAND FRA MÅLET = 45557.6091

ANSLÆT AFSTAND = 29972

INDTAST VINKELAFVIGELSE FRA X ?600
VINKELAFVIGELSE FRA Z ?50
AFSTAND ?45000

RADIANER FRA X-AKSEN = 10.4719352
FRA Z-AKSEN = .872661268

SKUDET RAMTE 15040.8037 KM. BAGVED MÅLET.
SKUDET RAMTE 264.920906 KM. TIL HØJRE FOR MÅLET.
SKUDET RAMTE 33187.5224 KM. OVER MÅLET.

SKUDETS ANSLÆDE POSITION:
X-KOORDINAT = -17237.1544
Y-KOORDINAT = -29852.8489
Z-KOORDINAT = 28925.5582
AFSTAND FRA MÅLET = 36437.7222

ANSLÆT AFSTAND = 29973

INDTAST VINKELAFVIGELSE FRA X ?700
VINKELAFVIGELSE FRA Z ?80
AFSTAND ?25000

RADIANER FRA X-AKSEN = 12.2172578
FRA Z-AKSEN = 1.39625803

SKUDET RAMTE 25331.3474 KM. FORAN MÅLET.
SKUDET RAMTE 21166.2461 KM. TIL VENSTRE FOR MÅLET.
SKUDET RAMTE 8603.3009 KM. OVER MÅLET.

SKUDETS ANSLÆDE POSITION:
X-KOORDINAT = 23134.9966
Y-KOORDINAT = -8421.68195
Z-KOORDINAT = 4341.33673
AFSTAND FRA MÅLET = 34113.1048

ANSLÆT AFSTAND = 29973.885

INDTAST VINKELAFVIGELSE FRA X ?500
VINKELAFVIGELSE FRA Z ?12
AFSTAND ?10

DU HAR SPRÆNGT DIG SELV I LUFTEN!!

NÆSTE MÅL.....

RADIANER FRA X-AKSEN = 5.51970655
FRA Z-AKSEN = .324994369

MÅLET ER SIDST IAGTTAGET PÅ FØLGENDE KOORDINATER:

X-KOORDINAT = 1533.43121
Y-KOORDINAT = -1467.63898
Z-KOORDINAT = 6299.57146
ANSLÆT AFSTAND = 6640

INDTAST VINKELAFVIGELSE FRA X ?

```
100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY      "
110 PRINT "-----"
120 PRINT "                          MÅL"
130 PRINT "-----"
140 PRINT
150 R = 1
160 R1 = 57.296
170 P = 3.141592
180 PRINT "DU ER VÅBEN-OFFICER PÅ RUMSKIBET ENTERPRISE, OG DU SKAL"
190 PRINT "GENNEMGÅ DENNE TEST SÅ DINE OVERORDNEDE KAN UNDERSØGE,"
200 PRINT "HVOR NØJAGTIGT DU KAN SKYDE I ET TRE-DIMENSIONALT RUM."
210 PRINT "DU VIL BLIVE INFORMERET OM X- OG Y-AKSERNES VINKLER"
220 PRINT "(I RADIANTER), MENS MÅLETS POSITION ANGIVES I RETVINK-"
230 PRINT "LEDE KOORDINATER, SAMT EN CIRKA-VÆRDI FOR MÅLETS"
240 PRINT "AFSTAND (I GRADER) FRA AKSERNE."
250 PRINT
260 PRINT "DU VIL DEREFTER FORTSÆTTE MED AT SKYDE PÅ MÅLET, INDTIL"
270 PRINT "DET ER ØDELAGT."
280 PRINT
290 PRINT "HELD OG LYKKE!!"
300 PRINT
310 A = RND (1) * 2 * P
320 B = RND (1) * 2 * P
330 Q = INT (A * R1)
340 W = INT (B * R1)
350 PRINT "RADIANER FRA X-AKSEN =";A
360 PRINT "      FRA Z-AKSEN =";B
370 PRINT
380 P1 = 100000 * RND (1) + RND (1)
390 X = SIN (B) * COS (A) * P1
400 Y = SIN (B) * SIN (A) * P1
410 Z = COS (B) * P1
420 PRINT "MÅLET ER SIDST IAGTTAGET PÅ FØLGENDE KOORDINATER:"
430 PRINT
```

```

440 PRINT " X-KOORDINAT =" ; X
450 PRINT " Y-KOORDINAT =" ; Y
460 PRINT " Z-KOORDINAT =" ; Z
470 R = R + 1
480 IF R > 5 THEN GOTO 590
490 ON R GOTO 500,520,540,560,580
500 P3 = INT (P1 * .05) * 20
510 GOTO 590
520 P3 = INT (P1 * .1) * 10
530 GOTO 590
540 P3 = INT (P1 * .5) * 2
550 GOTO 590
560 P3 = INT (P1)
570 GOTO 590
580 P3 = P1
590 PRINT "ANSLRET AFSTAND =" ; P3
600 PRINT
610 PRINT "INDTAST VINKELAFVIGELSE FRA X ";
620 INPUT A1
630 PRINT " VINKELAFVIGELSE FRA Z ";
640 INPUT B1
650 PRINT " AFSTAND ";
660 INPUT P2
670 PRINT
680 IF P2 < 20 THEN PRINT "DU HAR SPRÆNGT DIG SELV I LUFTEN!!": GOTO 870
690 A1 = A1 / R1
700 B1 = B1 / R1
710 PRINT "RADIANER FRA X-AKSEN =" ; A1
720 PRINT " FRA Z-AKSEN =" ; B1
730 PRINT
740 X1 = P2 * SIN (B1) * COS (A1)
750 Y1 = P2 * SIN (B1) * SIN (A1)
760 Z1 = P2 * COS (B1)
770 D = ((X1 - X) ^ 2 + (Y1 - Y) ^ 2 + (Z1 - Z) ^ 2) ^ (1 / 2)
780 IF D > 20 THEN GOTO 940
790 PRINT
800 PRINT "* * * * * P L E T S K U D * * * * *"
810 PRINT "***** MÅLET ER DESTRUERET *****"
820 PRINT
830 PRINT "SKUDET EKSPLODEREDE"; D ; " KILOMETER"
840 PRINT "FRA MÅLET."
850 PRINT
860 PRINT "DIN MISSION FULDFØRT PR"; R ; "SKUD."
870 R = 0
880 PRINT
890 PRINT
900 PRINT
910 PRINT "NÆSTE MÅL....."
920 PRINT
930 GOTO 310
940 X2 = X1 - X
950 Y2 = Y1 - Y
960 Z2 = Z1 - Z
970 IF X2 < 0 THEN GOTO 1000
980 PRINT "SKUDET RAMTE"; X2 ; "KM. FORAN MÅLET."
990 GOTO 1010
1000 PRINT "SKUDET RAMTE"; (- 1) * X2 ; "KM. BAGVED MÅLET."
1010 IF Y2 < 0 THEN GOTO 1040
1020 PRINT "SKUDET RAMTE"; Y2 ; "KM. TIL VENSTRE FOR MÅLET."
1030 GOTO 1050
1040 PRINT "SKUDET RAMTE"; (- 1) * Y2 ; "KM. TIL HØJRE FOR MÅLET."
1050 IF Z2 < 0 THEN GOTO 1080
1060 PRINT "SKUDET RAMTE"; Z2 ; "KM. OVER MÅLET."
1070 GOTO 1090
1080 PRINT "SKUDET RAMTE"; (- 1) * Z2 ; "KM. UNDER MÅLET."
1090 PRINT
1100 PRINT "SKUDETS ANSLREDE POSITION:"
1110 PRINT " X-KOORDINAT =" ; X1
1120 PRINT " Y-KOORDINAT =" ; Y1
1130 PRINT " Z-KOORDINAT =" ; Z1
1140 PRINT "AFSTAND FRA MÅLET =" ; D
1150 PRINT
1160 PRINT
1170 GOTO 470
1180 END

```

Månelanding

Dette spil, der findes i mange forskellige versioner, har lige så mange forskellige navne (Raket, Lunar, L.E.M. og Apollo). Det er uden tvivl det mest populære computerspil. Det findes i versioner, der starter mellem 500 meter og 200 kilometer over Månens overflade eller en anden måne eller planets overflade. I nogle af spillene kan man styre fartøjet med stabiliseringsraketter og/eller med bremsraketter.

De tre versioner, som findes her, repræsenterer de mest populære varianter af spillet.

I de fleste versioner ligger fristelsen i at begynde at bremse for tidligt, så man ikke har brændstof nok til at klare den sidste del af turen. Dette er selvfølgelig katastrofalt (hvilket du nok selv skal finde ud af, når du selv forsøger at lande på Månen!).

Månelanding er oprindeligt skrevet i FOCAL af Jim Storer, da han studerede ved Lexington High School, og er oversat til BASIC af David Ahl. Raket er skrevet af Eric Peters fra DEC, mens L.E.M. er skrevet af William Labaree II fra Alexandria, Virginia.

I Månelanding indstiller du bremsraketternes forbrændings-hastighed (hvor mange kilo brændstof der skal forbruges pr. sekund). Dette gøres hvert tiende sekund, og du skal forsøge at foretage en blød landing på Månen. Et forbrug på 100 kg/s svarer til at bremse helt, mens 0 kg/s resulterer i frit fald. Bremsraketterne starter først ved et forbrug på 4 kg/s, så brug IKKE hastigheder mellem 1 og 3 kg/s. Hvis du ønsker, at landingen skal blive lidt sværere, men også mere virkelighedstro, kan du sætte den tilgængelige brændstofmængde (N) til 8.000 kg, og vægten på månelandingsfartøjet (M) til 16.250 kg i programlinierne 390 og 400.

Visse versioner af BASIC tillader ikke de serieberegninger, som findes i linierne 1010 og 1020 (når du nærmer dig måneoverfladen bliver disse værdier meget små). Hvis disse linier ikke virker, så skift beregningerne i linie 1010 ud med følgende forenkede formel:

$$-Q*(1+Q*(1/2+Q*(1/3+Q*(1/4+Q/5))))$$

Den anden beregning bør du selv kunne finde ud af.

MORRISTOWN CREATIVE COMPUTING NEW JERSEY

MÅNELANDING

I DETTE SPIL SIMULERER DATAMATEN LANDINGEN AF ET APOLLO-MÅNELANDINGSMODUL.

DATAMATEN OMBORD PÅ MÅNELANDINGSFARTØJET ER UDE AF DRIFT (FORDI DEN BLEV FREMSTILLET AF XEROX), SÅ DU ER NØDT TIL AT STYRE MODULET MANUELT.

DU SKAL SÆTTE BREMSERAKETTERNES PARAMETRE TIL EN VÆRDI MELLEM 0 (= FRIT FALD) OG 100 (= MAKSIMUM VÆRDI) KILO BRÆNDSTOF PR. SEKUND.

NYE VÆRDIER INDTASTES HVER 10. SEKUND.

FARTØJET VEJER 16.500 KILO,
BRÆNDSTOFFET VEJER 8.250 KILO.

***** H E L D O G L Y K K E *****

SEK.	KM	KM/T	KG BRÆNDSTOF	BRÆNDHAST.
0	192	5760	8250	?0
10	175.919	5817.59	8250	?0
20	159.679	5875.19	8250	?0
30	143.279	5932.79	8250	?0
40	126.719	5990.39	8250	?0
50	110	6047.99	8250	?0
60	93.12	6105.59	8250	?0
70	76.08	6163.19	8250	?100
80	59.77	5572.58	7250	?100
90	45.161	4938.73	6250	?100
100	32.379	4255.44	5250	?100
110	21.572	3515.11	4250	?100
120	12.912	2708.21	3250	?100
130	6.599	1822.62	2250	?100
140	2.874	842.55	1250	?70
150	1.541	106.58	550	?10
160	1.329	45.69	450	?0
170	1.122	103.29	450	?10
180	.922	41.03	350	?0
190	.728	98.63	350	?0
200	.374	156.23	350	?0

PÅ MÅNEN EFTER: 207.56 SEKUNDER
LANDINGSHASTIGHED: 199.82 KM/T

DESVÆRRE, DU OVERLEVEDE DET IKKE. DU KIKSEDE!
JA FAKTISK SÅ LYKKEDES DET DIG AT SKABE ET NYT MÅNEKRATER!
DET ER 10.37 METER DYBT!

VIL DU SPILLE IGEN ('J' ELLER 'N') ?

```

100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY      "
110 PRINT "-----"
120 PRINT "              MÅNELANDING"
130 PRINT "-----"
140 PRINT
150 PRINT
160 PRINT "I DETTE SPIL SIMULERER DATAMATEN LANDINGEN AF ET APOLLO-"
170 PRINT "MÅNELANDINGSMODUL."
180 PRINT
190 PRINT "DATAMATEN OMBORD PÅ MÅNELANDINGSFARTØJET ER UDE AF DRIFT"
200 PRINT "(FORDI DEN BLEV FREMSTILLET AF XEROX), SÅ DU ER NØDT TIL"
210 PRINT "AT STYRE MODULET MANUELT."
220 PRINT
230 PRINT "DU SKAL SÆTTE BREMSERAKETTERNES PARAMETRE TIL EN VÆRDI"
240 PRINT "MELLEML 0 (= FRIT FALD) OG 100 (= MAKSIMUM VÆRDI) KILO"
250 PRINT "BRÆNDSTOF PR. SEKUND."
260 PRINT
270 PRINT "NYE VÆRDIER INDTASTES HVER 10. SEKUND."
280 PRINT
290 PRINT "      FARTØJET VEJER 16.500 KILO,"
300 PRINT "BRÆNDSTOFFET VEJER 8.250 KILO."
310 PRINT
320 PRINT "***** H E L D   O G   L Y K K E   *****"
330 L = 0
340 PRINT
350 PRINT "SEK.", "KM", "KM/T", "KG BRÆNDSTOF", "BRÆNDHAST."
360 PRINT
370 A = 120
380 V = 1
390 M = 33000
400 N = 16500
410 G = .001
420 Z = 1.8
430 PRINT INT (L), INT (A * 1.6 * 1000) / 1000, INT (3600 * 100 * 1.6 *
V) / 100, INT ((M * .5) - (N * .5)),
440 INPUT K
450 K = K * 2
460 T = 10
470 IF M - N < .001 THEN 580
480 IF T < .001 THEN 430
490 S = T
500 IF M > = N + S * K THEN 520
510 S = (M - N) / K
520 GOSUB 1000
530 IF I < = 0 THEN 860
540 IF V < = 0 THEN 560
550 IF J < 0 THEN 920
560 GOSUB 800
570 GOTO 470
580 PRINT
590 PRINT "BRÆNDSTOF SLUPPET OP EFTER"; INT (L * 100) / 100;"SEKUNDER"
600 S = (- V + SQR (V * V + 2 * A * G)) / G
610 PRINT
620 V = V + G * S
630 L = L + S
640 W = 3600 * V
650 PRINT
660 PRINT "      PÅ MÅNEN EFTER"; INT (L * 100) / 100;"SEKUNDER"
670 PRINT "LANDINGSHASTIGHED"; INT (W * 1.6 * 100) / 100;"KM/T"
680 PRINT
690 IF W < = 1.2 THEN PRINT "PERFEKT LANDING"; GOTO 1040
700 IF W < = 10 THEN PRINT "GOD LANDING (KUNNE VÆRE BEDRE)"; GOTO 1040
710 IF W > 60 THEN 760
720 PRINT "SKADE PÅ FARTØJET.....DU ER STRANDET PÅ MÅNEN. DU,KAN"
730 PRINT "IKKE KOMME HERFRA FØR ET REDNINGSFARTØJ NÅR FREM....."
740 PRINT "JEG HÅBER DU HAR NOK ILT!!!"
750 GOTO 1040
760 PRINT "DESVÆRRE, DU OVERLEVEDE DET IKKE. DU KIKSEDE!"
770 PRINT "JA FAKTISK SÅ LYKKEDES DET DIG AT SKABE ET NYT MÅNEKRATER!"
780 PRINT "DET ER"; INT (W * .277 * .3 * 100) / 100;"METER DYBT!"
790 GOTO 1040
800 L = L + S
810 T = T - S
820 M = M - S * K
830 A = I
840 V = J
850 RETURN
860 IF S < .005 THEN 640
870 D = V + SQR (V * V + 2 * A * (G - Z * K / M))
880 S = 2 * A / D
890 GOSUB 1000
900 GOSUB 800
910 GOTO 860
920 W = (1 - M * G / (Z * K)) / 2
930 S = M * V / (Z * K * (W + SQR (W * W + V / Z))) + .05
940 GOSUB 1000
950 IF I < = 0 THEN 860
960 GOSUB 800
970 IF J > 0 THEN 470
980 IF V > 0 THEN 920
990 GOTO 470
1000 Q = S * K / M
1010 J = V + G * S + Z * (- Q - Q * Q / 2 - Q ^ 3 / 3 - Q ^ 4 / 4 - Q ^ 5
/ 5)
1020 I = A - G * S * S / 2 - V * S + Z * S * (Q / 2 + Q ^ 2 / 6 + Q ^ 3 /
12 + Q ^ 4 / 20 + Q ^ 5 / 30)
1030 RETURN
1040 PRINT
1050 PRINT
1060 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1070 INPUT AS
1080 PRINT
1090 IF LEFT$(AS,1) = "J" THEN 290
1100 END

```



```

T,X,F ?50,50,90
-----
TID: 1550      HØJ: 23159.0476      AFS: -4194932.31

LOD: -310.784426      VAN: 232.757828      BRÆ: 127.000007

T,X,F ?50,50,90
-----
TID: 1600      HØJ: 5635.0199      AFS: -4180432.7

LOD: -390.110712      VAN: 352.732969      BRÆ: 102.000007

T,X,F ?10,0,0
-----
TID: 1610      HØJ: 1654.71263      AFS: -4176908.82

LOD: -405.96275      VAN: 353.550403      BRÆ: 102.000007

T,X,F ?10,100,0
-----
TID: 1614.5      HØJ: -143.162801      AFS: -4175317.69

LOD: -393.082658      VAN: 353.921054      BRÆ: 97.5000066

KRASZZZZH !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
DIT NEDSLAG HAR DANNET ET NYT MÅNEKRATER!
DET ER 143.162801 METER DYBT.
VED NEDSLAGET HAVDE DU EN HASTIGHED PÅ 1904.17231 KILOMETER/TIME

ØNSKER DU AT PRØVE IGEN (SVAR 'J' ELLER 'N') ?N

ÆRGELIGT NOK, FOLKENE BAG APOLLO-PROJEKTET KAN IKKE LIDE
AT SKULLE MISTE GODE ASTRONAUTER.

100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY
110 PRINT "-----"
120 PRINT "      L.E.M."
130 PRINT "-----"
140 REM LEM er et interaktivt spil, der simulerer en månelanding
150 REM der minder om månelandingerne i Apollo programmer.
160 REM Der er ingen tilfældigheder/held i dette spil
170 Z$ = "START"
180 B1 = 1
190 M = 17.95
200 F1 = 5.25
210 N = 7.5
220 RO = 926
230 VO = 1.29
240 T = 0
250 HO = 60
260 R = RO + HO
270 A = - 3.425
280 R1 = 0
290 A1 = 8.84361E - 04
300 R3 = 0
310 A3 = 0
320 M1 = 7.45
330 MO = M1
340 B = 750
350 T1 = 0
360 F = 0
370 P = 0
380 N = 1
390 M2 = 0
400 S = 0
410 C = 0
420 IF LEFT$ (Z$,1) = "J" THEN 2490
430 PRINT
440 PRINT "MÅNELANDINGS SIMULATOR"
450 PRINT
460 PRINT "HAR DU FLØJET EN APOLLO/LEM MISSION FØR "
470 PRINT "(INDTAST 'J' ELLER 'N') ";
480 INPUT QS
490 PRINT
500 IF LEFT$ (QS,1) = "J" THEN 540
510 IF LEFT$ (QS,1) = "N" THEN 570
520 PRINT "SVAR BLOT PÅ SPØRGSMÅLET, ";
530 GOTO 470
540 PRINT
550 PRINT "VÆLG MÅLEENHEDSNUMMER ";
560 GOTO 610
570 PRINT
580 PRINT "HVILKET MÅLESYSTEM FORETRÆKKER DU?"
590 PRINT "1 = DET METRISKE      0 = DET ENGELSK"

```

```

600 PRINT "INDTAST DET ØNSKEDE NUMMER ";
610 INPUT K
620 PRINT
630 IF K = 0 THEN 720
640 IF K = 1 THEN 660
650 GOTO 600
660 Z = 1852.8
670 M$ = "METER"
680 G3 = 3.6
690 N$ = " KILOMETER"
700 G5 = 1000
710 GOTO 770
720 Z = 6080
730 M$ = "FOD"
740 G3 = .592
750 N$ = "SØMIL"
760 G5 = Z
770 IF B1 = 3 THEN 1500
780 IF LEFT$ (QS,1) = "J" THEN 1130
790 PRINT
800 PRINT "DU ER PÅ EN MÅNELANDINGSMISSION. DU ER PILOT PÅ 'LUNAR"
810 PRINT "EXCURSION MODULE' (LEM). DU SKAL GIVE VISSER KOMMANDOER TIL"
820 PRINT "MODULETS NAVIGATIONSSYSTEM."
830 PRINT
840 PRINT "DATAMATEN OMBORD PÅ MODULET VIL LØBENDE GIVE DIG DE INFOR-"
850 PRINT "MATIONER DU SKAL BRUGE FOR AT LANDE MODULET PÅ MÅNEN."
860 PRINT
870 PRINT "FLYVEVINKLEN DEFINERES SÅLEDES:"
880 PRINT "+/- 180 GRADER = VINKELRET BORT FRA MÅNEOVERFLADEN."
890 PRINT " - 90 GRADER = TANGENT I SAMME RETNING SOM BANERETNINGEN"
900 PRINT " + 90 GRADER = TANGENT I MODSAT RETNING AF BANERETNINGEN"
910 PRINT "      0 GRADER = DIREKTE MOD MÅNEN."
920 PRINT
930 PRINT "      <-----BANERETNING"
940 PRINT
950 PRINT "      -180,+180"
960 PRINT "      ""
970 PRINT "      !"
980 PRINT "      -90 <-----+-----> +90
990 PRINT "      !"
1000 PRINT "      v"
1010 PRINT "      0"
1020 PRINT
1030 PRINT "-----MÅNENS OVERFLADE"
1040 PRINT
1050 PRINT
1060 PRINT "ALLE VINKLER FRA -180 TIL +180 KAN INDFASTES."
1070 PRINT
1080 PRINT "NÅR VINKLEN ÆNDRES, FORBRUGES BRÆNSTOF."
1090 PRINT "DRIVKRAFT TIL RÅDIGHED: FRA 0 TIL 100 PROCENT."
1100 PRINT
1110 PRINT "INDTASTNING AF NEGATIV DRIVKRAFT ELLER FLYVETID ER FORBUDT."
1120 PRINT
1130 PRINT
1140 PRINT "      INDTAST: TIDSINTERVAL I SEKUNDER.....(T)"
1150 PRINT "      PROCENT AF MAKSIMAL DRIVKRAFT.....(%)
1160 PRINT "      FLYVEVINKEL I GRADER.....(F)"
1170 PRINT
1180 IF LEFT$ (QS,1) = "J" THEN 1230
1190 PRINT "FOR EKSEMPEL:"
1200 PRINT "T,X,F ?10,65,-60"
1210 PRINT "(INDTAST '0,0,0' HVIS DU ØNSKER AT AFBRYDE DIN MISSION)"
1220 PRINT
1230 PRINT "UDSKRIFT: TOTALT BRUGT TID I SEKUNDER"
1240 PRINT "      HØJDE OVER MÅNEN I ";M$
1250 PRINT "      AFSTAND FRA LANDINGSPLADSEN I ";M$
1260 PRINT "      LODRET HASTIGHED I ";M$;"SEKUND"
1270 PRINT "      VANDRET HASTIGHED I ";M$;"SEKUND"
1280 PRINT "      BRÆNDELSENHEDER TILBAGE"
1290 PRINT
1300 GOTO 1500
1310 PRINT
1320 PRINT "T,X,F ";
1330 INPUT T1,F,P
1340 F = F / 100
1350 IF T1 < 0 THEN 2000
1360 IF T1 = 0 THEN 2370
1370 IF ABS (F - .05) > 1 THEN 2070
1380 IF ABS (F - .05) < .05 THEN 2070
1390 IF ABS (P) > 180 THEN 2030
1400 N = 20
1410 IF T1 < 400 THEN 1430
1420 N = T1 / 20
1430 T1 = T1 / N
1440 P = P * 3.14159 / 180
1450 S = SIN (P)
1460 C = COS (P)
1470 M2 = MO * T1 * F / B

```

```

1480 R3 = -.5 * R0 * ((VO / R) ^ 2) + R * A1 * A1
1490 A3 = - 2 * R1 * A1 / R
1500 FOR I = 1 TO N
1510 IF M1 = 0 THEN 1590
1520 M1 = M1 - M2
1530 IF M1 > 0 THEN 1610
1540 F = F * (1 + M1 / M2)
1550 M2 = M1 + M2
1560 PRINT "DU HAR IKKE MERE BRÆNDSTOF."
1570 M1 = 0
1580 GOTO 1610
1590 F = 0
1600 M2 = 0
1610 M = M - .5 * M2
1620 R4 = R3
1630 R3 = -.5 * R0 * ((VO / R) ^ 2) + R * A1 * A1
1640 R2 = (3 * R3 - R4) / 2 + .00526 * F1 * F * C / M
1650 A4 = A3
1660 A3 = - 2 * R1 * A1 / R
1670 A2 = (3 * A3 - A4) / 2 + .0056 * F1 * F * S / (M * R)
1680 X = R1 * T1 + .5 * R2 * T1 * T1
1690 R = R + X
1700 HO = HO + X
1710 R1 = R1 + R2 * T1
1720 A = A + A1 * T1 + .5 * A2 * T1 * T1
1730 A1 = A1 + A2 * T1
1740 M = M - .5 * M2
1750 T = T + T1
1760 IF HO < 3.287828E - 04 THEN 1780
1770 NEXT I
1780 H = HO * Z
1790 H1 = R1 * Z
1800 D = R0 * A * Z
1810 D1 = R * A1 * Z
1820 T2 = M1 * B / M0
1830 PRINT "-----"
1840 PRINT "TID:","T;" HØJ:","H;" AFS:","D
1850 PRINT
1860 PRINT "LOD:","H1;" VAN:","D1;" BRÆ:","T2
1870 PRINT
1880 IF HQ < 3.287828E - 04 THEN 1950
1890 IF R0 * A > 164.4736 THEN 2290
1900 IF M1 > 0 THEN 1320
1910 T1 = 20
1920 F = 0
1930 P = 0
1940 GOTO 1400
1950 IF R1 < - 8.21957E - 04 THEN 2220
1960 IF ABS (R * A1) > 4.931742E - 04 THEN 2220
1970 IF HO < - 3.287828E - 04 THEN 2220
1980 IF ABS (D) > 10 * Z THEN 2330
1990 GOTO 2170
2000 PRINT
2010 PRINT "DETTE MÅNEFARTØJ KAN IKKE OPHÆVE TYNGDEKRAFTEN!!"
2020 GOTO 1310
2030 PRINT
2040 PRINT "HVIS DU ØNSKER AT SNURRE RUNDT, SÅ GÅ UDEN FOR MODULET OG"
2050 PRINT "TAG DIG EN SVINGOM."
2060 GOTO 1310
2070 PRINT
2080 PRINT "HELT UMULIG DRIVKRAFT-VÆRDI ";
2090 IF F < 0 THEN 2150
2100 IF F - .05 < .05 THEN 2130
2110 PRINT "- FOR STOR"
2120 GOTO 1310
2130 PRINT "- FOR LILLE"
2140 GOTO 1310
2150 PRINT "- NEGATIV"
2160 GOTO 1310
2170 PRINT
2180 PRINT "DU ER LANDET PÅ MÅNEBASEN I STILHEDENS HAV. TILLYKKE!!!"
2190 PRINT "MODULET BLEV IKKE SKADET, OG DU KAN FORTSÆTTE MED AT UD-"
2200 PRINT "FORSKE MÅNEOVERFLADEN."
2210 GOTO 2390
2220 PRINT
2230 PRINT "KRASZZZZZ!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!"
2240 PRINT "DIT NEDSLAG HAR DANNET ET NYT MÅNEKRATER!"
2250 PRINT "DET ER "; ABS (H);M$;" DYBT."
2260 X1 = SQR (D1 * D1 + H1 * H1) * G3
2270 PRINT "VED NEDSLAGET HAVDE DU EN HASTIGHED PÅ";X1;N$;" / TIME"
2280 GOTO 2390
2290 PRINT
2300 PRINT "DU ER FORSVUNDET I DET UENDELIGE RUM....."
2310 PRINT "DU VIL ALDRIG BLIVE FUNDET MERE....."
2320 GOTO 2390
2330 PRINT "DU ER LANDET SIKKERT - "
2340 PRINT
2350 PRINT "MEN DU ER"; ABS (D / G5);N$;" FRA LANDINGSPLADSEN"
2360 GOTO 2390
2370 PRINT
2380 PRINT "MISSIONEN AFBRUDT."
2390 PRINT
2400 PRINT "ØNSKER DU AT PRØVE IGEN (SVAR 'J' ELLER 'N') ";
2410 INPUT Z$
2420 IF LEFT$ (Z$,1) = "J" THEN 190
2430 IF LEFT$ (Z$,1) = "N" THEN 2450
2440 GOTO 2400
2450 PRINT
2460 PRINT "ÆRGELIGT NOK, FOLKENE BAG APOLLO-PROJEKTET KAN IKKE LIDE"
2470 PRINT "AT SKULLE MISTE GODE ASTRONAUTER."
2480 GOTO 2640
2490 PRINT
2500 PRINT "OK, VIL DU HAVE ALLE INSTRUKTIONERNE, ELLER KUN INDTAST-"
2510 PRINT "NINGS- OG UDSKRIFTS-BESKRIVELSERNE "
2520 PRINT "1 = ALLE INSTRUKTIONER"
2530 PRINT "2 = INDTAST/UDSKRIVNINGS INSTRUKTIONER"
2540 PRINT "3 = INGEN INSTRUKTIONER"
2550 PRINT
2560 PRINT "HVAD VÆLGER DU ";
2570 INPUT B1
2580 Q$ = "N"
2590 IF B1 = 1 THEN 570
2600 Q$ = "J"
2610 IF B1 = 2 THEN 540
2620 IF B1 = 3 THEN 540
2630 GOTO 2520
2640 END

```

Månelandingsraket

I denne version starter du 150 meter over måneoverfladen. Du styrer forbrændingshastigheden på rakterne, der affyres hvert sekund. Hver brændstofenhed mindsker faldhastigheden med 3.0 m/s. Den maksimale bremsekraft med rakterne er 10 m/s/s.

***** IKKE MERE BRÆNDSTOF *****

15 1.5 5 0 I*

***** OVERFLADE KONTAKT *****

LANDING EFTER 15.2649111 SEKUNDER.
LANDINGSHASTIGHED = 6.32455532 METER/SEKUND.
0 BRÆNDELSENHEDER TILBAGE.

***** DESVÆRRE, DU KIKSEDE!!!!

DER VIL BLIVE SENDT KONDOLENCEBREVE TIL DIN FAMILIE.

VIL DU SPILLE IGEN ('J' ELLER 'N') ?N

DATA-KONTROL FRAKOBLET

MORRISTOWN CREATIVE COMPUTING NEW JERSEY

RAKET

MÅNELANDINGS SIMULATION

ØNSKES INSTRUKTIONER ('J' ELLER 'N') ?J

DU SKAL LANDE PÅ MÅNEN, OG DU SKAL NAVIGERE MÅNEMODULET SIKKERT NED PÅ MÅNEN, FRA EN HØJDE PÅ 500 METER OVER LANDINGSTEDET. DU STARTER MED EN HASTIGHED NED MOD MÅNEOVERFLADEN PÅ 50 METER/SEKUNDET. DU HAR 120 ENHEDER BRÆNDSTOF TIL RÆDIGHED.

HER FØLGER DE REGLER, SOM MODULET ER UNDERLAGT:

- (1) FOR HVERT SEKUND BLIVER HØJDEN OVER MÅNEOVERFLADEN, HASTIGHEDEN OG BRÆNDSTOFMÆNGDE UDSEKREVET I EN RAPPORT.
- (2) NÅR RAPPORTEN ER UDSEKREVET, VIL DER PÅ SKÆRMEN SKRIVES ET '?'. INDTAST DET ANTAL ENHEDER BRÆNDSTOF SOM DU VIL BRUGE I DET NÆSTE SEKUND. HVER ENHED BRÆNDSTOF VIL MINDSKE DIN NEDSTIGNING MED 0,3 METER/SEKUNDET.
- (3) DEN STØRSTE MULIGE BREMSEKRAFT SOM DINE RAKETTER KAN UDVIKLE ER 10 METER/SEKUND/SEKUND, ELLER 30 BRÆNDSTOF-ENHEDER PR. SEKUND.
- (4) NÅR DU KOMMER I KONTAKT MED MÅNEOVERFLADEN, SÅ VIL RAKETTERNE AUTOMATISK BLIVE SLÅET FRA, OG DER VIL BLIVE UDSEKREVET EN RAPPORT OVER DIN LANDINGSHASTIGHED OG TILBAGEVÆRENDE BRÆNDSTOF.
- (5) HVIS DU LØBER TØR FOR BRÆNDSTOF, SÅ VIL '?' IKKE LÆNGERE KOMME TIL SYNE, MEN DER VIL STADIG BLIVE UDSEKREVET EN RAPPORT HVERT SEKUND, INDTIL DU RAMMER MÅNEOVERFLADEN.

VI BEGYNDER LANDINGEN NU!.....

H E L D O G L Y K K E

SEK	METER	HASTIGHED	BRÆNDSTOF	AFSTANDSKURVE
0	500	50	120	I *
?0				
1	447.5	55	120	I *
?0				
2	390	60	120	I *
?0				
3	327.5	65	120	I *
?0				
4	260	70	120	I *
?10				
5	192.5	65	110	I *
?5				
6	127.5	65	105	I *
?25				
7	72.5	45	80	I *
?25				
8	37.5	25	55	I *
?25				
9	22.5	5	30	I*
?7				
10	18.5	3	23	I*
?7				
11	16.5	1	16	I*
?4				
12	15	2	12	I*
?3				
13	12	4	9	I*
?3				
14	7	6	6	I*
?6				


```

100 PRINT "      MORRISTOWN      CREATIVE COMPUTING      NEW JERSEY      "
110 PRINT "-----"
120 PRINT "                      RAKET"
130 PRINT "-----"
140 PRINT
150 PRINT "MÅNELANDINGS SIMULATION"
160 PRINT "-----"
170 PRINT
180 PRINT "ØNSKES INSTRUKTIONER ('J' ELLER 'N') ";
190 INPUT AS
200 IF LEFT$(AS,1) = "N" THEN 460
210 PRINT
220 PRINT "DU SKAL LANDE PÅ MÅNEN, OG DU SKAL NAVIGERE MÅNEMODULET"
230 PRINT "SIKKERT NED PÅ MÅNEN, FRA EN HØJDE PÅ 500 METER OVER"
240 PRINT "LANDINGSSTEDET. DU STARTER MED EN HASTIGHED NED MOD"
250 PRINT "MÅNEOVERFLADEN PÅ 50 METER/SEKUNDET. DU HAR 120 ENHEDER"
260 PRINT "BRÆNDSTOF TIL RÆDIGHED."
270 PRINT
280 PRINT "HER FØLGER DE REGLER, SOM MODULET ER UNDERLAGT:"
290 PRINT "(1) FOR HVERT SEKUND BLIVER HØJDEN OVER MÅNEOVERFLADEN,"
300 PRINT "HASTIGHEDEN OG BRÆNDSTOFMÆNGDE UDSKREVET I EN RAPPORT."
310 PRINT "(2) NÅR RAPPORTEN ER UDSKREVET, VIL DER PÅ SKÆRMEN SKRIVES"
320 PRINT "ET '?'. INDTAST DET ANTAL ENHEDER BRÆNDSTOF SOM DU VIL"
330 PRINT "BRUGE I DET NÆSTE SEKUND. HVER ENHED BRÆNDSTOF VIL"
340 PRINT "MINDSKE DIN NEDSTIGNING MED 0,3 METER/SEKUNDET."
350 PRINT "(3) DEN STØRSTE MULIGE BREMSEKRAFT SOM DINE RAKETTER KAN"
360 PRINT "UDVIKLE ER 10 METER/SEKUND/SEKUND, ELLER 30 BRÆNDSTOF-"
370 PRINT "ENHEDER PR. SEKUND."
380 PRINT "(4) NÅR DU KOMMER I KONTAKT MED MÅNEOVERFLADEN, SÅ VIL"
390 PRINT "RAKETTERNE AUTOMATISK BLIVE SLÆT FRA, OG DER VIL BLIVE"
400 PRINT "UDSKREVET EN RAPPORT OVER DIN LANDINGSFASTIGHED OG"
410 PRINT "TILBAGEVÆRENDE BRÆNDSTOF."
420 PRINT "(5) HVIS DU LØBER TØR FOR BRÆNDSTOF, SÅ VIL '?' IKKE LÆNGERE"
430 PRINT "KOMME TIL SYNE, MEN DER VIL STADIG BLIVE UDSKREVET EN"
440 PRINT "RAPPORT HVERT SEKUND, INDTIL DU RAMMER MÅNEOVERFLADEN."
450 PRINT
460 PRINT "VI BEGYNDER LANDINGEN NU!....."
470 PRINT
480 PRINT "H E L D   O G   L Y K K E"
490 PRINT
500 PRINT
510 PRINT "SEK METER      HASTIGHED      BRÆNDSTOF      AFSTANDSKURVE"
520 PRINT
530 T = 0
540 H = 500
550 V = 50
560 F = 120
570 PRINT T; TAB( 4);H; TAB( 12);V; TAB( 20);
580 PRINT F; TAB( 29);"I"; TAB( H / 12 + 29);"*"
590 INPUT B
600 IF B < 0 THEN 760
610 IF B > 30 THEN B = 30
620 IF B > F THEN B = F
630 V1 = V - B + 5
640 F = F - B
650 H = H - .5 * (V + V1)
660 IF H < = 0 THEN 780
670 T = T + 1
680 V = V1
690 IF F > 0 THEN 570
700 IF B = 0 THEN 740
710 PRINT
720 PRINT "***** IKKE MERE BRÆNDSTOF *****"
730 PRINT
740 PRINT T; TAB( 4);H; TAB( 12);V; TAB( 20);
750 PRINT F; TAB( 29);"I"; TAB( H / 12 + 29);"*"
760 B = 0
770 GOTO 630
780 PRINT
790 PRINT "***** OVERFLADE KONTAKT *****"
800 PRINT
810 H = H + .5 * (V + V1)
820 IF B = 5 THEN 850
830 D = ( - V + SQR (V * V + H * (10 - 2 * B))) / (5 - B)
840 GOTO 860
850 D = H / V
860 V1 = V + (5 - B) * D
870 PRINT "LANDING EFTER";T + D;"SEKUNDER."
880 PRINT "LANDINGSFASTIGHED =" ;V1;"METER/SEKUND."
890 PRINT F;"BRÆNDSLENHEDER TILBAGE."
900 IF V1 < > 0 THEN 930
910 PRINT "TILLYKKE!!! EN PERFEKT LANDING."
920 PRINT "DIT KØREKORT VIL BLIVE FORNYET.....SENERE."
930 IF ABS (V1) < 2 THEN 980
940 PRINT
950 PRINT "***** DESVÆRRE, DU KIKSEDE!!!"
960 PRINT
970 PRINT "DER VIL BLIVE SENDT KONDOLANCEBREVE TIL DIN FAMILIE."
980 PRINT
990 PRINT
1000 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1010 INPUT AS
1020 IF LEFT$(AS,1) = "J" THEN 460
1030 PRINT
1040 PRINT "DATA-KONTROL FRAKOBLET"
1050 PRINT
1060 END

```

Navn

Navn er et morsomt lille program, der kan få noget til at ske mellem et generet menneske og en datamat. Den sorteringsalgoritme, der anvendes, er vanvittig ineffektiv. Som enhver læser af Creative Computing vil vide, er denne den mest langsomme sorteringsmetode, der findes. Men programmet er morsomt, og det er det, der tæller her.

Navn blev oprindeligt skrevet af Geoffrey Chase fra Abbey, Portsmouth, Rhode Island.

```
MORRISTOWN/CREATIVE COMPUTING/N. JERSEY
```

```
-----  
                        NAVN  
-----
```

```
GODDAV!
```

```
MIT NAVN ER KARSTEN KOMPUTER.
```

```
HVAD HEDDER DU ?  
(FORNAVN OG EFTERNAVN) ?CHRISTIAN BORUP
```

```
TAK!
```

```
PUROB NAITSIHHC.
```

```
HOVSA!! JEG KOM TIL AT SKRIVE DIT NAVN  
BAGFRA. EN DATAMAT DER ER SÅ KLOG SOM  
JEG BURDE IKKE GØRE SÅ PINLIGE FEJL.
```

```
JEG LAGDE NETOP MÆRKE TIL, AT BOGSTA-  
VERNE KOM I FORKERT RÆKKEFØLGE.
```

```
LAD MIG LIGE ORDNE DEM IGEN, SÅDAN :
```

```
ABCHIIOPRRSTU
```

```
SYNES DU IKKE AT DET SER BEDRE UD  
(SVAR 'JA' ELLER 'NEJ') ?N
```

```
DET ER JEG KED AF AT HØRE.
```

```
DET VAR HYGGELIGT OG EN STOR OPLEVEL-  
SE AT MØDE DIG, CHRISTIAN BORUP.  
PAS PÅ DIG SELV.....FARVEL.
```

```
100 PRINT "MORRISTOWN/CREATIVE COMPUTING/N. JERSEY"  
110 PRINT "-----"  
120 PRINT "                        NAVN"  
130 PRINT "-----"  
140 PRINT  
150 DIM B$(40)  
160 PRINT "GODDAV!"  
170 PRINT  
180 PRINT "MIT NAVN ER KARSTEN KOMPUTER."  
190 PRINT  
200 PRINT "HVAD HEDDER DU ?"  
210 PRINT "(FORNAVN OG EFTERNAVN) ";  
220 INPUT A$  
230 L = LEN (A$)  
240 PRINT  
250 PRINT "TAK!"  
260 PRINT  
270 FOR I = 1 TO L  
280 B$(I) = MID$(A$,I,1)  
290 NEXT I  
300 FOR I = L TO 1 STEP - 1  
310 PRINT B$(I);  
320 NEXT I  
330 PRINT "."  
340 PRINT  
350 PRINT "HOVSA!! JEG KOM TIL AT SKRIVE DIT NAVN"  
360 PRINT "BAGFRA. EN DATAMAT DER ER SÅ KLOG SOM"  
370 PRINT "JEG BURDE IKKE GØRE SÅ PINLIGE FEJL."  
380 PRINT  
390 PRINT "JEG LAGDE NETOP MÆRKE TIL, AT BOGSTA-"  
400 PRINT "VERNE KOM I FORKERT RÆKKEFØLGE."  
410 PRINT  
420 PRINT "LAD MIG LIGE ORDNE DEM IGEN, SÅDAN : "  
430 PRINT  
440 FOR J = 2 TO L  
450 I = J - 1  
460 T$ = B$(J)  
470 IF T$ > B$(I) THEN 510  
480 B$(I + 1) = B$(I)  
490 I = I - 1  
500 IF I > 0 THEN 470  
510 B$(I + 1) = T$  
520 NEXT J  
530 FOR I = 1 TO L  
540 PRINT B$(I);  
550 NEXT I  
560 PRINT  
570 PRINT  
580 PRINT "SYNES DU IKKE AT DET SER BEDRE UD "  
590 PRINT "(SVAR 'JA' ELLER 'NEJ') ";  
600 INPUT D$  
610 IF LEFT$(D$,1) = "J" THEN 650  
620 PRINT  
630 PRINT "DET ER JEG KED AF AT HØRE."  
640 GOTO 670  
650 PRINT  
660 PRINT "JEG VIDSTE DU VILLE GIVE MIG RET!"  
670 PRINT  
680 PRINT "DET VAR HYGGELIGT OG EN STOR OPLEVEL-"  
690 PRINT "SE AT MØDE DIG, ";A$;"."  
700 PRINT "PAS PÅ DIG SELV.....FARVEL."  
710 END
```

Appendix

Basic-test

Dette program bør indtastes og køres før bogens andre programmer.

Ved hjælp af BASIC-TEST vil du hurtigt være i stand til at undersøge hvilke af bogens BASIC-instruktioner, der fejlfortolkes af den BASIC din datamat bruger. Indtast programmet omhyggeligt og kør det. Hvis din computer kan køre programmet og giver de samme resultater (bortset fra RND-funktionen – selvfølgelig) som nedenstående kørsel, så vil du ikke få problemer med bogens programmer. Husk dog, at de værdier som funktionerne beregner heller ikke behøver at passe på de sidste decimaler. Hvis BASIC-TEST standser midt i kørslen eller giver fejlmeddelelse, så find ud af, hvilken programlinie der ikke udføres som forventet. Arbejd med problemet og prøv forskellige muligheder, indtil din version af BASIC-TEST udskriver de samme resultater som prøvekørslen. På den måde kan du være sikker på, at problemet er løst tilfredsstillende, og at den samme løsning vil kunne anvendes i bogens andre programmer.

Programmet er skrevet af Christian Borup.

```
-----
                        BASIC-TEST
-----

*** DETTE ER ET BASIC TEST PROGRAM ***

3.333
-3.333

*** INDBYGGEDE FUNKTIONER ***

-.190240727
-.981737473
.193779633
1.27931201
28.0222826
1.2038728
1.82565057
5.555      0      5.555
-6          0      5
-1          0      1

OBS!
NEDENSTRENDE VÆRDIER KAN AFVIGE PR DIN
COMPUTER, DE SKAL BLOT VÆRE STØRRE END
0 OG MINDRE END 1

.93753437   .0544707574
.246017374   .220233661
.975331284   .253108841
.869225068   .533446289
.892159924   .225467183

***** ** *

*** HOP OG LØKKER ***

GOTO UDFØRT
GOSUB UDFØRT
IF-THEN UDFØRT
IF-THEN GOSUB UDFØRT
ON-GOTO UDFØRT
ON-GOSUB UDFØRT

-2          -2
-2          -1
-2          0
-2          1
-2          2
0           0
0           1
0           2
2           2

*** DATA-SÆTNINGER ***
```

```
100 PRINT
110 PRINT "-----"
120 PRINT "          BASIC-TEST"
130 PRINT "-----"
140 REM DETTE ER EN REMARK-SÆTNING
150 PRINT
160 A = 0:B = 2
170 C = 3.333:C1 = - 3.333
180 D = 5.555:D1 = - 5.555
190 K$ = "COMPUTER SPIL"
200 DIM A(5),B(2,2),A$(10)
210 PRINT "*** DETTE ER ET BASIC TEST PROGRAM ***"
220 PRINT
230 PRINT C
240 PRINT C1
250 PRINT
260 PRINT "*** INDBYGGEDE FUNKTIONER ***"
270 PRINT
280 PRINT SIN (C)
290 PRINT COS (C)
300 PRINT TAN (C)
310 PRINT ATN (C)
320 PRINT EXP (C)
330 PRINT LOG (C)
340 PRINT SQR (C)
350 PRINT ABS (D1), ABS (A), ABS (D)
360 PRINT INT (D1), INT (A), INT (D)
370 PRINT SGN (D1), SGN (A), SGN (D)
380 PRINT
390 PRINT "OBS!"
400 PRINT "NEDENSTRENDE VÆRDIER KAN AFVIGE PR DIN"
410 PRINT "COMPUTER, DE SKAL BLOT VÆRE STØRRE END"
420 PRINT "0 OG MINDRE END 1"
430 PRINT
440 FOR I = 1 TO 5
450 PRINT RND (1), RND (1)
460 NEXT I
470 PRINT
480 PRINT "*****"; TAB( 12);"*";
490 PRINT TAB( 13);"*"; TAB( 15);"*"
500 PRINT
510 PRINT
520 PRINT "*** HOP OG LØKKER ***"
530 PRINT
540 GOTO 1370
550 GOSUB 1390
560 IF C = 3.333 THEN 1410
570 IF C = 3.333 THEN GOSUB 1430
580 IF A < > 0 THEN 1450: PRINT "DENNE BASICS
IF-THEN PROCEDURE FORKERT": GOTO 580
590 ON B GOTO 1470,1450,1470
600 ON B GOSUB 1490,1510,1490
610 PRINT
620 FOR I = - 2 TO 2 STEP 2
630 FOR J = I TO 2
640 PRINT I,
650 PRINT J
660 NEXT J
670 NEXT I
680 PRINT
690 PRINT "**** DATA-SÆTNINGER ****"
700 PRINT
710 DATA 5,4,3,2,1
720 DATA "E","D","C","B","A"
730 DATA "5","4","3","2","1"
740 FOR I = 1 TO 5
750 READ A(I)
760 PRINT A(I);
770 NEXT I
780 PRINT
790 FOR I = 1 TO 10
800 READ A$(I)
810 PRINT A$(I);
820 NEXT I
830 PRINT
840 PRINT
850 RESTORE
860 FOR J = 1 TO 5
870 READ M
880 NEXT J
890 PRINT M
900 PRINT
910 PRINT "*** OPERATORER ***"
920 PRINT
```

```

930 PRINT (C + C) * D1 / B - C1 ^ 2
940 PRINT C + C * D1 / B - C1 ^ 2
950 PRINT ABS (1 < > 1), ABS (2 < 1)
960 PRINT ABS (1 = 1), ABS (2 > 1)
970 PRINT ABS ((1 > = 1) AND (2 > = 3))
980 PRINT ABS ((1 > = 1) OR (2 > = 3))
990 PRINT
1000 PRINT "*** BRUGER-DEFINERED FUNKTIONER ***"
1010 PRINT
1020 DEF FN R(X) = X * 10 + X / (X / X)
1030 X = - 4
1040 PRINT FN R(X), FN R(C1)
1050 PRINT
1060 PRINT "*** STRENGE ***"
1070 PRINT
1080 FOR I = 5 TO 1 STEP - 1
1090 PRINT ASC (A$(I)), CHR$ (70 - I)
1100 NEXT I
1110 PRINT
1120 FOR I = 10 TO 6 STEP - 1
1130 PRINT STR$ (I - 5), VAL (A$(I))
1140 NEXT I
1150 PRINT
1160 PRINT LEN (K$)
1170 PRINT
1180 PRINT LEFT$ (K$,3);"*";
1190 PRINT MID$ (K$,4,7);"*";
1200 PRINT RIGHT$ (K$,3)
1210 PRINT
1220 A$ = A$(8) + "-" + A$(9) + "-" + A$(10) + "-0"
1230 PRINT A$
1240 PRINT
1250 PRINT "*** DENNE LØKKE RINGER KLOKKEN ***"
1260 FOR I = 1 TO 5
1270 PRINT CHR$ (7);
1280 NEXT I
1290 PRINT
1300 PRINT "VIL DU SPILLE IGEN ('J' ELLER 'N') ";
1310 INPUT A$
1320 PRINT
1330 RESTORE
1340 IF LEFT$ (A$,1) = "J" THEN 210
1350 GOTO 1360
1360 END
1370 PRINT "GOTO UDFØRT"
1380 GOTO 550
1390 PRINT "GOSUB UDFØRT"
1400 RETURN
1410 PRINT "IF-THEN UDFØRT"
1420 GOTO 570
1430 PRINT "IF-THEN GOSUB UDFØRT"
1440 RETURN
1450 PRINT "ON-GOTO UDFØRT"
1460 GOTO 600
1470 PRINT "FEJL I ON-GOTO"
1480 GOTO 600
1490 PRINT "FEJL I ON-GOSUB"
1500 RETURN
1510 PRINT "ON-GOSUB UDFØRT"
1520 RETURN
1530 END

```

Fortegnelse over spillene opdelt i kategorier

Morskab

FRASEGENERATOR · 69, bind 1
GODDAV · 70, bind 1
NAVN · 72, bind 2
POESI · 21, bind 3
RUSSISK ROULETTE · 30, bind 3
SAKS-STEN-PAPIR · 31, bind 3
UGEDAG · 72, bind 3

Undervisning

AKTIER · 12, bind 1
BORGERKRIG · 44, bind 1
BYTTEPENGE · 53, bind 1
DYRERIGET · 63, bind 1
HAMMURABI · 9, bind 2
HÆNGT MAND · 18, bind 2
HØVDING · 22, bind 2
KAST · 32, bind 2
KEMIKER · 35, bind 2
KONGE · 36, bind 2
LITTERATUR QUIZ · 52, bind 2
MATEMATISK TERNING · 62, bind 2
PELSHANDLER · 14, bind 3
SYNONYM · 57, bind 3
TOG · 62, bind 3

Billeder og kurver

BANNER · 28, bind 1
BILLE · 33, bind 1
HOP · 16, bind 2
KALENDER · 27, bind 2
KANIN · 31, bind 2
LABYRINT · 47, bind 2
LIV · 53, bind 2
LIV FOR TO · 56, bind 2
LOVE · 58, bind 2
PLOT I 3-D · 20, bind 3
RUDERE · 29, bind 3
SINUSKURVE · 32, bind 3

Gæt tal eller bogstaver

BOGSTAV · 36, bind 1
GÆT · 8, bind 2
HØJT-LAVT · 21, bind 2
OMRING · 10, bind 3
STJERNER · 46, bind 3
TAL · 61, bind 3

Fjern objekter

BUNKE · 52, bind 1
LIGE VINDER · 49, bind 2
LIGE VINDER, VERSION 2 · 50, bind 2
NIM · 8, bind 3
TÆNDSTIKKER · 68, bind 3

Koordinatsystemer

BOMBARDEMENT · 39, bind 1
HURKLE · 17, bind 2
PIZZA · 18, bind 3
SLAGSKIB · 34, bind 3
SMUGORM · 39, bind 3
SÆNKE SLAGSKIBE · 58, bind 3
UBÅDSJAGT · 70, bind 3

Logik

AWARI · 22, bind 1
BAGELS · 24, bind 1
BAGLÆNS · 26, bind 1
BONDESPIL · 41, bind 1
CIFFER · 54, bind 1
DAM FOR EN · 59, bind 1
DRONNINGESKAK · 61, bind 1
FLIP-FLOP · 68, bind 1
HAPS · 11, bind 2
KRYDS OG BOLLE · 40, bind 2
KRYDS OG BOLLE, VERSION 2 · 41, bind 2
KRYDS OG BOLLE I 3-D · 42, bind 2
KUBUS · 45, bind 2
MASTER MIND · 59, bind 2
NICOMACHUS · 7, bind 3
ORD · 13, bind 3
SOLITAIRE · 41, bind 3
TÅRNET I HANOI · 69, bind 3

Rumfart

MÅL · 63, bind 2
MÅNELANDING · 65, bind 2
MÅNELANDING, L.E.M. · 67, bind 2
MÅNELANDING, RAKET · 70, bind 2
ORBIT · 11, bind 3
SPLAT · 44, bind 3
SUPER STAR TREK INSTRUKTIONER · 47, bind 3
SUPER STAR TREK X · 50, bind 3

Sport

AMERIKANSK FODBOLD · 15, bind 1
AMERIKANSK FODBOLD, VERSION 2 · 19, bind 1
BASKETBALL · 30, bind 1
BOKSNING · 37, bind 1
BOWLING · 49, bind 1
GOLF · 72, bind 1
ISHOCKEY · 23, bind 2
PILESPIL · 16, bind 3
SLALOM · 37, bind 3
TYREFÆGTNING · 64, bind 3

Kasino og hasard

CRAPS · 56, bind 1
ENOGTYVE · 65, bind 1
HESTEVÆDDELØB · 14, bind 2
POKER · 22, bind 3
ROULETTE · 26, bind 3
TERNING · 62, bind 3
TYVEKNÆGT · 66, bind 3

Kort- og brætspil

DAM · 57, bind 1
GOMOKO · 7, bind 2
HASARD · 13, bind 2
KRIG · 39, bind 2

Krigsspil

KAMP · 29, bind 2
KAST BOMBERNE · 33, bind 2
SKYTTE · 33, bind 3

Borgens Computerbøger

David Ahl

Basic computerspil

Bind 1-3, 80 sider pr. bd., ill.

Klassisk samling af spil til computere. Spillene er skrevet i BASIC og vil kunne bruges på de fleste computere.

Barker, Jacobi og Zimmermann

Mikrodatamaten som tegneredskab

347 sider, ill.

Om at fremstille tegninger ved hjælp af mikrocomputere. Bogen er også håndbog i brugen af MONSTER-programmet, som gør datamaten til en tegnemaskine.

X.T. Bui

Bedre beslutninger med BASIC

234 sider, ill.

Programsamling til brug for beslutningstagere. Behandler brugen af PERT, prognosemodeller, nulpunktsanalyse, investeringsmodeller m.m.

Oluf Danielsen

EDB i hverdagen

42 sider, ill.

Debatbog til brug i folkeskolens samtidsorientering.

Andreas Dripke

Maskinkodeprogrammering med 6502

183 sider

Bogen, som henvender sig til begyndere, gennemgår instruktionssættet for 6502-processoren og forklarer brugen af en assembler.

Walter A. Ettl

Wordstar på en nem måde

192 sider

Indføring i brugen af det meget udbredte tekstbehandlingsprogram. Indeholder eksempler på færdige opstillinger af breve m.m. Velegnet som både opslagsbog og begyndervejledning.

Jørgen Feder

Pascal – også for begyndere

ca. 385 sider

Den mest omfattende bog på dansk om programmeringssproget Pascal. Fungerer både som indføring og håndbog. Dækker de meget udbredte Poly-Pascal og Turbo-Pascal. Fyldig behandling af brugen af pointere.

Annie Fox og David Fox

Lænestols-BASIC

304 sider, ill.

Helt elementær, humoristisk indføring i programmeringssproget BASIC. Kræver ikke adgang til en computer.

Eugene Galanter

Computeren – hverdagsredskab i hjem og skole

240 sider, ill.

Lettilgængelig bog om børn og computere. Henvender sig til forældre og pædagoger.

Goldschlager og Lister

Datalogi og datamaskiner

320 sider

Dybtdgående gennemgang af moderne databehandlingsmetoder med hovedvægten på algoritmebegrebet og principperne for hvordan opgaver omformes til algoritmer.

Tord-Jöran Hallberg

Sådan virker mikrodatamaten

100 sider, ill.

Gennemgår en tænkt mikrodatamat MM og forklarer i detaljer hvordan den programmeres i maskinkode.

Fietz, Kitza og Mantz

Håndbog i MSDOS/PCDOS

ca. 180 sider

Bogen falder i to dele: Først fortælles om hvad operativsystemet er, hvilke opgaver det løser og hvordan man arbejder med det. Bogens anden del er et referenceafsnit, hvor hver enkelt kommando gennemgås. Bogen indeholder et appendiks om nogle af de nye kommandoer i version 3.0 og senere.

Ken Knecht

Microsoft BASIC

261 sider

Populær og omhyggelig gennemgang af denne meget udbredte version af BASIC med mange programeksempler, herunder læsning og skrivning i filer.

Jørn Lorentzen og Henrik Nellager

Maskinkode med Amstrad

340 sider

Om maskinkodeprogrammering på Amstrad CPC464, 664 og 6128. Forfatterne starter med Z80 processorens instruktionssæt og behandler derefter Amstrad-computernes særlige egenskaber m.h.t. grafik, farver og lyd. Stort afsnit om firmwarerutiner, der kan bruges som byggesten i egne programmer.

Jørn Lorentzen og Henrik Nellager

Maskinkode med ZX81

240 sider

Populær begynderbog med mange eksempler, herunder komplet Space Invader spil.

Murtha og Waite

CP/M for begyndere

208 sider

Om CP/M vers. 2.2. Beskriver operativsystemets virkemåde og opgaver. Læseren lærer assemblerprogrammering med DDT og ASM og bogen giver nyttige tips om brugen af STAT, SYSGEN, PIP og andre hjælpeprogrammer.

Glenford J. Myers

Kunsten at teste edb-programmer

246 sider

Fortæller om fejlfinding og afprøvning af edb-programmer. Opstiller et sæt regler og principper, som man kan gå frem efter, når programmer skal fejlrettes.

Erwin Neutzsky-Wulff

Mikrodatamaten

Programmering og anvendelse – en bog om ZX81 BASIC

280 sider

Underholdende og lærerig bog om programmering af hjemme-computeren ZX81.

Erwin Neutzsky-Wulff

Programmering med Commodore BASIC

220 sider, ill.

Om programmering af hjemmecomputeren VIC20. Mange program-eksempler. Kan også bruges med PET og de lidt større Commodore maskiner.

Erwin Neutzsky-Wulff

BASIC med Commodore 64

218 sider, ill.

Bogen er skrevet specielt til Commodore 64 hjemmecomputeren. Indeholder mange morsomme spil og lege som kan testes ind i computeren. Ideel som BASIC-lærebog til selvstudium.

Erwin Neutzsky-Wulff

Amstrad BASIC

200 sider

Bogen er skrevet til brug med Amstrad CPC464, 664 og 6128. Ideel som BASIC-lærebog til selvstudium. Udforsker computerens muligheder og egenskaber. Indeholder mange underholdende spil og lege, der er lige klar til at blive tastet ind i computeren.

Lon Poole og Mary Borchers

En samling nyttige BASIC-programmer

339 sider, ill.

Dækker et bredt spektrum af almindeligt forekommende beregnings-problemer inden for rentesregning, geometri, statistik m.m.

Lon Poole

IBM PC – den personlige computer

480 sider, ill.

Populær bog med hovedvægten på BASIC-programmering med IBM PC. Desuden en fyldig vejledning i brugen af operativsystemet PCDOS. Bogen kan også bruges sammen med IBM PC/XT og AT og med IBM-kompatible computere.

Inge-Lise Salomon

Elementær datalære for HF og tilsvarende undervisningstrin

304 sider, ill.

Fortæller om edb-maskiner, anvendelse af edb, mennesker og edb, samt programmering i COMAL. En grundig bog, der er velegnet til aften-skole og selvstudium, samt undervisning i gymnasiet.

Alan Simpson

Lotus 1-2-3

Det elektroniske kolonneark

332 sider, ill.

Vejledning i brugen af Lotus 1-2-3 regnearket. De forskellige faciliteter gennemgås, herunder brugen af grafik og database, samt brugen af makroer, der gør 1-2-3 til et meget kraftfuldt værktøj.

Ian Sinclair

ZX Spectrum og hvordan man får glæde af den

180 sider

Populær gennemgang af ZX Spectrum hjemmecomputeren.

Jesper Skavin

Maskinkodeprogrammering med Z80

224 sider

Instruktionssættet for Z80 processoren gennemgås og instruktionernes virkemåde forklares.

Ian Stewart og Robin Jones

Lær BASIC-programmering med ZX Spectrum

250 sider

Letforståelig bog om BASIC-programmering på ZX Spectrum hjemme-computeren. Mange færdige programeksempler.

Ian Stewart og Robin Jones

Maskinkode med Commodore 64

200 sider, ill.

Bogen fortæller om hvordan man fra BASIC kan lave maskinkode-programmer. Mange underholdende og lærerige programeksempler.

Ludwigs, Poppensieker og Surowiecki

UNIX for begyndere

ca. 225 sider

Gennemgår UNIX-operativsystemet, der vinder stigende udbredelse i flerbrugersystemer.

Niels Søndergaard

Lav dine egne computereventyr med BASIC

200 sider, ill.

Første bog på dansk om en ny og underholdende anvendelse af computere. Computereventyr er tekstspil, hvor computeren på skærmen forklarer hvad der sker, og spørger dig hvad du vil gøre. Du befinder dig måske i en ufremkommelig jungle, hvor du skal finde en skat og forsvare dig mod alskens farer.

Carl Townsend

Programmering med dBASE II

250 sider

Bogen fortæller detaljeret hvordan man selv kan bruge dBASE II til opbygning af databaser til mange forskellige formål. Talrige eksempler.

Her er en samling af de bedste spil til computere.

Hvert spil er omtalt med en indledning, en listning af programmet lige klar til at taste ind, og et prøvespil der viser, hvordan spillet forløber.

Alle spillene er skrevet i Microsoft BASIC, som i forskellige udgaver findes på flertallet af microcomputere.

Bogen hjælper dig med at tilpasse spillene, hvis du har en computer med en anden BASIC.

Bogen henvender sig til alle brugere af mikrocomputere.

Bind 2

BORGEN

ISBN 87-418-7685-7

